True*Grid*® Examples Manual

Version 2.1

XYZ Scientific Applications, Inc.

October 30, 2001
Rachel Urbanski

Table of Contents

Acknowledgments

I would like to thank the following people for contributing models to this manual:

Mike Burger
Alois Denek
Trent Eggleston
Tracy Glover
Jeff Painter
Robert Rainsberger
Doug Stillman

Introduction

This manual is designed to give **True*Grid***® users examples of how to create certain meshes. From these examples it is hoped that the user will be able to learn how to create their own model to fit their specific meshing needs. It is necessary to note that although these examples produce the desired mesh, they are not the only methods that can be used to create that model. We strongly encourage you to not only learn how to design through the methods shown, but to also figure out your own unique style in order to successfully design your own meshes.

In Chapters I - IX the meshes are set up in increasing difficulty. These examples are only intended to give you the commands to generate that mesh. They contain no comments. In Chapter X you will find examples that have been heavily commented. Along with these comments, this chapter contains many diagrams from the computational window. The Recipes chapter (Chapter XI) is intended to show sections of input to do specific tasks that require multiple commands in **True*Grid***®. It shows how these commands work together. It also serves as a recipe for similar types of problems. This chapter is not intended as a full input to **True*Grid***®. It is accompanied with explanations for you to see how an experienced **True*Grid***® user might go about solving the problem.

This manual used a few conventions worth mentioning. Colors are used in this manual the way they are used in **True*Grid***®. Green is used for the mesh, red for surfaces, and yellow for curves. Regions that are highlighted are in the appropriate color. Not all occurrences of commands have been placed in the index. In particular, the block command is not found in the index at all. Text which is input to **True*Grid***® appears in the Courier font. Bold face font is used primarily to distinguish keyword commands.

We hope that you find these examples useful when creating your own mesh. In your use of **True*Grid***®, if you develop examples that you would like to see in this manual, feel free to e-mail them to us at info@truegrid.com or fax to (925) 373-6326. Bug reports will also be appreciated.

I. Simple Cubes

1. Simple cube with one element

```
block 1 2; 1 2; 1 2; 1 2; 1 2; 1 2;
```



2. Simple cube with many elements

```
block 1 21; 1 21;1 21; 1 2;1 2;1 2;
```



3. Simple cube with varying mesh sizes

```
block 1 6 11 16 21;1 6 11 16 21; 1 6 11 16 21;
       1 2 4 8 16;  1 2 4 8 16;   1 2 4 8 16;
```



4. Simple cube with sections deleted

```
block 1 6 11 16 21;1 6 11 16 21;1 6 11 16 21;1 2 4 8 16;1 2 4 8
16 ;1 2 4 8 16;
dei 1 2 0 4 5; 1 2 0 4 5; 1 2 0 4 5;
```



II. Simple Cylinders

1. Simple solid cylinder

```
block 1 3 5 7;1 3 5 7;1 3 5 7 9 11 13 15;
1 2 3 4;1 2 3 4;1 2 3 4 5 6 7 8;
sd 1 cyli 2.5 2.5 2.5 0 0 1 2
sfi -1 -4; -1 -4; 1 8;sd 1
```



2. Simple cylinder with hole

```
block 1 3 5 7; 1 3 5 7;1 3 5 7 9 11 13 15; 1 2 3 4;1 2 3 4;1 2 3
4 5 6 7 8;
sd 1 cyli 2.5 2.5 2.5 0 0 1 2
sd 2 cyli 2.5 2.5 2.5 0 0 1 1
sfi -1 -4; -1 -4; 1 8;sd 1
sfi -2 -3; -2 -3; 1 8;sd 2
dei 2 3; 2 3; 1 8;
mseq i 2 2 2;
mseq j 2 2 2;
```



3. Hollow Shell Cylinder

Method 1

```
cylinder -1;1 19;1 5;1.5;0 360;0 2;
```

Method 2

```
block -1 -5;-1 -5;1 5;-1 1;-1 1;0
2;
sd 1 cyli 0 0 0 0 0 1 1.5
sfi -1 -2; -1 -2; -1 -2;sd 1
```



4. Cylinder with Stiffeners

```
cylinder 1 -3;1 19;-1 -5;1.2 1.5;0 360;0 2;
```



5. Cylinder with spherical cap

```
block -1 5 -9;-1 5 -9;-1 5 -9;-1 0 1;-1 0 1;-3 0 1;
sfi -1 -3; -1 -3;1 2; cyli 0 0 0 0 0 1 5
sfi -1 -3;-1 -3;2 -3;sp 0 0 0 5
de 1 0 0 2 0 0
dei 2 3; 1 3; -1;
```



6. Solid cylinder with spherical end and liner

```
block 1 3 5 7 9 11 13;1 3 5 7 9 11 13;1 7 9 11 13;-2.5 -2.5 -2.5
0 2.5 2.5 2.5;-2.5 -2.5 -2.5 0 2.5 2.5 2.5;-7.5 0 2.5 2.5 2.5;
dei 1 3 0 5 7;;3 5;
dei ;1 3 0 5 7;3 5;
sfi -2 -6;-2 -6;2 -4; sp 0 0 0 5
sfi -1 -7;-1 -7;2 -5; sp 0 0 0 5.5
sfi -2 -6;-2 -6;1 2; cy 0 0 0 0 0 1 5
sfi -1 -7;-1 -7;1 2;cy 0 0 0 0 0 1
5.5
de 1 0 0 4 0 0
```



7. Telescoping cylinder

Method 1

```
block 1 3 5 7 9 11 13 15 17 19 21;1 3 5 7 9 11 13 15 17 19 21; 1
3 5 7 9 11 13 15 17; 1 2 3 4 5 6 7 8 9 10 11;1 2 3 4 5 6 7 8 9 10
11;1 2 3 4 5 6 7 8 9;
sd 1 cyli 6 6 6 0 0 1 6
sd 2 cyli 6 6 6 0 0 1 5
sd 3 cyli 6 6 6 0 0 1 4
sd 4 cyli 6 6 6 0 0 1 3
sd 5 cyli 6 6 6 0 0 1 2
sfi -1 -11; -1 -11; 1 9;sd 1
sfi -2 -10; -2 -10; 1 9;sd 2
sfi -3 -9; -3 -9; 1 9;sd 3
sfi -4 -8; -4 -8; 1 9;sd 4
sfi -5 -7; -5 -7; 1 9;sd 5
dei 5 7; 5 7; 1 9;
dei 1 11; 1 4 0 8 11; 7 9;
dei 1 4 0 8 11; 4 8; 7 9;
dei 1 11; 1 3 0 9 11; 5 7;
dei 1 3 0 9 11; 3 9; 5 7;
dei 1 2 0 10 11; 1 11; 3 5;
dei 2 10; 1 2 0 10 11; 3 5;
```



Method 2

```
cylinder 1 3 5 7 9;1 41;1 3 5 7 9;1 2 3 4 5;0 360;0 1 2 3 4;
dei 4 5;; 2 5;
dei 3 5;; 3 5;
dei 2 5;; 4 5;
```



8. Thick cylindrical shell with spherical cap

```
block 1 3 5 7 9;1 3 5 7 9;1 5 7 9;-1 -1 0 1 1;
-1 -1 0 1 1;-2 0 1 1;
dei 1 2 0 4 5;1 2 0 4 5;;
dei 1 2 0 4 5;;3 4;
dei ;1 2 0 4 5;3 4;
sfi -1 -5; -1 -5;2 -4; sp 0 0 0 5
sfi -2 -4;-2 -4;2 -3; sp 0 0 0 3
sfi -1 -5;-1 -5;1 2; cy 0 0 0 0 0 1 5
sfi -2 -4;-2 -4;1 2; cy 0 0 0 0 0 1 3
de 1 0 0 3 0 0
de 2 2 1 4 4 3
```



III. Simple Plates

1. Simple circular plate

```
block 1 5;1 5;-1;-1 1; -1 1;0;
curd 1 arc3 whole rt 1.5 0 0 rt 0 1.5 0 rt -1.5 0 0 ;
curs 1 2 1 2 2 1 1
curs 1 1 1 1 2 1 1
curs 1 1 1 2 1 1 1
curs 2 1 1 2 2 1 1
```



2. Circular plate with hole

Method 1

```
cylinder 1 3;1 21;-1;1 1.5;0 360;0;
```

Method 2

```
block 1 5 11 15;1 5 11 15;-1;-1 -1 1 1;-1 -1 1 1;0;
dei 1 2 0 3 4;1 2 0 3 4;-1;
curd 1 arc3 whole rt 1.5 0 0 rt 0 1.5 0 rt -1.5 0 0;
cur 4 2 1 4 3 1 1
cur 2 4 1 3 4 1 1
cur 1 2 1 1 3 1 1
cur 2 1 1 3 1 1 1
de 2 2 0 3 3 0;
curd 2 arc3 whole rt 1 0 0 rt 0 1 0 rt -1
0 0;
cur 3 2 1 3 3 1 2
cur 2 2 1 3 2 1 2
cur 2 2 1 2 3 1 2
cur 2 3 1 3 3 1 2
```



3. Zoned circular plate

```
block 1 3 7 9;1 3 7 9;-1;-0.5 -0.5 0.5 0.5;-0.5 -0.5 0.5 0.5;0;
dei 1 2 0 3 4;1 2 0 3 4;-1;
curd 1 arc3 whole rt 1 0 0 rt 0 1 0 rt -1 0 0 ;
curs 1 4 1 4 4 1 1
curs 2 1 1 3 1 1 1
curs 4 2 1 4 3 1 1
curs 1 2 1 1 3 1 1
```



4. Half circular plate

```
block 1 3 5 7 9;1 3 5 7 9;-1;-.5 -.5 0 .5 .5;-.5 -.5 0 .5 .5;0;
dei 1 2 0 4 5;1 2 0 4 5;-1;
curd 1 arc3 whole rt 1 0 0 rt 0 1 0 rt -1 0 0;
curs 2 1 1 4 1 1 1
curs 5 2 1 5 4 1 1
curs 2 5 1 4 5 1 1
curs 1 2 1 1 4 1 1
de 1 0 0 3 0 0
```



5. Circular plate with offset hole

```
block 1 5 11 15;1 5 11 15;-1;-1 -1 1 1;-1 -1 1 1;0;
dei 1 2 0 3 4;1 2 0 3 4;-1;
curd 1 arc3 whole rt 1.5 0 0 rt 0 1.5 0 rt -1.5 0 0;
curs 1 2 1 1 3 1 1
curs 2 4 1 3 4 1 1
curs 2 1 1 3 1 1 1
curs 4 2 1 4 3 1 1
curd 2 arc3 whole rt .25 .5 0 rt 0 .75 0 rt -.25 .5 0 ;
curs 3 2 1 3 3 1 2
curs 2 2 1 3 2 1 2
curs 2 2 1 2 3 1 2
curs 2 3 1 3 3 1 2
dei 2 3; 2 3; -1;
tfi 2 3;1 2; ;
tfi 2 3;1 2; ;
tfi 3 4;2 3; ;
tfi 2 3;3 4; ;
tfi 1 2;2 3; ;
```



6. Circular plate with outer liner

```
block 1 3 5 11 13 15;1 3 5 11 13 15;-1;-.5 -.5 -.5 .5 .5 .5;-.5
-.5 -.5 .5 .5 .5;0;
dei 1 3 0 4 6;1 3 0 4 6;-1;
curd 1 arc3 whole rt 1.1 0 0 rt 0 1.1 0
rt -1.1 0 0;
curs 6 3 1 6 4 1 1
curs 3 6 1 4 6 1 1
curs 1 3 1 1 4 1 1
curs 3 1 1 4 1 1 1
curd 2 arc3 whole rt 1 0 0 rt 0 1 0
rt -1 0 0;
curs 3 2 1 4 2 1 2
curs 5 3 1 5 4 1 2
curs 3 5 1 4 5 1 2
curs 2 3 1 2 4 1 2
```



7. Right triangular plate

```
block 1 5 9;1 5 9;-1;0 .3 1;0 .3 1;0;
de 2 2 0 3 3 0
pa 2 3 0 12 .5 .5
pa 3 2 0 12 .5 .5
```



8. Square plate with hole

Method 1

```
block 1 5 9 13;1 5 9 13;-1;-2 -2 2 2;-2 -2 2 2;0;
dei 1 2 0 3 4;1 2 0 3 4;-1;
de 2 2 0 3 3 0
curd 1 arc3 whole rt 1 0 0 rt 0 1 0 rt -1 0 0;
curs 3 2 1 3 3 1 1
curs 2 3 1 3 3 1 1
curs 2 2 1 2 3 1 1
curs 2 2 1 3 2 1 1
```

Method 2

```
block 1 5 9 13; 1 5 9 13; -1; 0 1 2                         3;0
1 2 3;0;
de 2 2 0 3 3 0
sd 1 cy 1.5 1.5 0 0 0 1 .75
sfi 2 3; 2 3; -1; sd 1
```



IV. Spheres

1. Simple sphere

Method 1

```
block 1 3 5 7 9;1 3 5 7 9;1 3 5 7 9;-2.5 -2.5 0 2.5 2.5;-2.5 -2.5
0 2.5 2.5;-2.5 -2.5 0 2.5 2.5;
dei 1 2 0 4 5; 1 2 0 4 5;;
dei 1 2 0 4 5;;1 2 0 4 5;
dei ;1 2 0 4 5;1 2 0 4 5;
sfi -1 -5;-1 -5;-1 -5; sp 0 0 0 5
```





Method 2

```
block 1 11; 1 11; 1 11; -1 1; -1 1; -1 1;
sfi -1 -2; -1 -2; -1 -2; sp 0 0 0 1
```





2. Solid sphere with liner

```
block 1 3 5 7 9 11 13;1 3 5 7 9 11 13;1 3 5 7 9 11 13;-2.5 -2.5
-2.5 0 2.5 2.5 2.5;-2.5 -2.5 -2.5 0 2.5 2.5 2.5;-2.5 -2.5 -2.5 0
2.5 2.5 2.5;
dei 1 3 0 5 7;1 3 0 5 7;;
dei 1 3 0 5 7;;1 3 0 5 7;
dei;1 3 0 5 7;1 3 0 5 7;
sfi -2 -6;-2 -6;-2 -6;sp 0 0 0 5
sfi -1 -7;-1 -7;-1 -7;sp 0 0 0 5.5
de 1 0 0 4 0 0
```



3. Spherical shell

```
block -1 5 -9;-1 5 -9;-1 5 -9;-1 0 1;-1 0 1;-1 0 1;
sfi -1 -3 ;-1 -3 ;-1 -3 ;sp 0 0 0 5
de 1 0 0 2 0 0
```



4. Thick spherical shell

```
block 1 3 5 7 9;1 3 5 7 9;1 3 5 7 9;-1 -1 0 1 1;-1 -1 0 1 1;-1 -1
0 1 1;
dei 1 2 0 4 5; 1 2 0 4 5;;
dei 1 2 0 4 5;;1 2 0 4 5;
dei ;1 2 0 4 5;1 2 0 4 5;
sfi -1 -5;-1 -5;-1 -5;sp 0 0 0 5
sfi -2 -4;-2 -4;-2 -4;sp 0 0 0 3
de 2 2 2 4 4 4
de 1 0 0 3 0 0
```



5. Simple solid sphere section

```
merge
cylinder 1 4;1 37;1 21;1.8 2;0 180;-1.8 1.8;
sd 1 sp 0 0 0 1.8
sd 2 sp 0 0 0 2
sfi -1;;;sd 1
sfi -2;;;sd 2
sd 3 cy 0 0 0 0 0 1 .4
sfi ;; -1 0 -2;sd 3
```



6. Stair stepped sphere

```
merge
cylinder 1 3 5 7 9 11;1 37;1 3 5 7 9 11 13 15 17 19 21;1 1.1 1.2
1.3 1.4 1.5;0 180;-1 -.8 -.6 -.4 -.2 0 .2 .4 .6 .8 1;
sfi -1; 1 2; 1 11;sp 0 0 0 1
sfi -2; 1 2; 1 11;sp 0 0 0 1.1
sfi -3; 1 2; 1 11;sp 0 0 0 1.2
sfi -4; 1 2; 1 11;sp 0 0 0 1.3
sfi -5; 1 2; 1 11;sp 0 0 0 1.4
sfi -6; 1 2; 1 11;sp 0 0 0 1.5
sfi 1 6; 1 2; -1 0 -11;cone 0 0 0 0 0 1 0 15
sfi 1 6; 1 2; -2 0 -10;cone 0 0 0 0 0 1 0 30
sfi 1 6; 1 2; -3 0 -9;cone 0 0 0 0 0 1 0 45
sfi 1 6; 1 2; -4 0 -8;cone 0 0 0 0 0 1 0 60
sfi 1 6; 1 2; -5 0 -7;cone 0 0 0 0 0 1 0 75
dei 5 6; 1 2; 2 3 0 9 10;
dei 4 6; 1 2; 3 4 0 8 9;
dei 3 6; 1 2; 4 5 0 7 8;
dei 2 6; 1 2; 5 7;
```



V. Cylindrical Objects

1. Bead

```
sd 1 plan 6 0 0 1 0 0
sd 2 plan 8 0 0 1 0 0
sd 3 plan 0 6 0 0 1 0
sd 4 plan 0 8 0 0 1 0
sd 5 cy 7 7 7 0 0 1 7
sd 6 sp 7 7 2 7
sd 7 sp 7 7 12 7
block 1 6 7 9 10 15;1 6 7 9 10 15;1 6 10 14 18 23;
      5 5 6 8 9 9;5 5 6 8 9 9;-2 -2 2 12 16 16;
dei 1 2 0 5 6;1 2 0 5 6;;
dei 1 2 0 5 6;;1 2 0 5 6;
dei ;1 2 0 5 6;1 2 0 5 6;
dei 3 4;3 4;;
sfi -3;3 4;;sd 1
sfi -4;3 4;;sd 2
sfi 3 4;-3;;sd 3
sfi 3 4;-4;;sd 4
sfi -1 -6;-1 -6;3 4;sd 5
sfi -1 -6;-1 -6;-1 3;sd 6
sfi -1 -6;-1 -6;4 -6;sd 7
endpart
merge
```



2. Fiber optic cable

```
parameter d 1
  r [(sqrt(2)-1)*%d];
sd 1 cy 0 [%d+%r] 0 0 0 1 %d
sd 2 cy 0 0 0 0 0 1 %r
block 1 2 5 6;1 2 5 6;1 12;
  [-%d/2] [-%d/2] [%d/2] [%d/2]
  [%r+%d/2] [%r+%d/2] [%r+3*%d/2] [%r+3*%d/2]
  0 10
dei 1 2 0 3 4; 1 2 0 3 4;;
sfi -1 -4; -1 -4;;sd 1
t1=sqrt(x*x+y*y)
t2=atan(x/y)
x=t1*cos(t2+z*9)
y=t1*sin(t2+z*9)
lct 3 rz 90;repeat 3;lrep 0:3;
endpart
block 1 2 5 6;1 2 5 6;1 12;
  [-%r/2] [-%r/2] [%r/2] [%r/2]
  [-%r/2] [-%r/2] [%r/2] [%r/2]
  0 10
dei 1 2 0 3 4; 1 2 0 3 4;;
sfi -1 -4; -1 -4;;sd 2
endpart
merge
```



3. Anode







```
parameter
md 20
rho1 .1
rho2 .125
rho3 .15
xc 1
zc 1
z1 0
z2 .25
z3 .375
z4 .75
z5 .855
z7 1.125;
parameter
pi 3.14159
z6 [%z7-2*%rho2*%xc/%zc]
ij1 [max(2,nint((%rho2-%rho1)*%md))]
ij2 [max(1,nint(.25*%pi*%rho2*%md))]
ij3 [max(2,nint((%rho3-%rho2)*%md))]
k1 [max(1,nint((%z2-%z1)*%md))]
k2 [max(1,nint((%z3-%z2)*%md))]
k3 [max(1,nint((%z4-%z3)*%md))]
k4 [max(1,nint((.5*(%z6+%z7)-%z4)*%md))]
epsin  [%rho1/sqrt(2)]
delin  [(%rho2-%epsin)*%xc/%zc]
epsout [%rho2/sqrt(2)]
delout [(%rho2-%epsout)*%xc/%zc];
sd 1 cy   0 0 0 0 0 1 %rho1
sd 2 cy   0 0 0 0 0 1 %rho2
sd 3 cy   0 0 0 0 0 1 %rho3
sd 4 plan %rho2 0 %z7 [-%xc] 0 %zc
block
1 [1+%ij3] [1+%ij3+%ij1] [1+%ij3+%ij1+%ij2] [1+%ij3+%ij1+2*%ij2]
[1+%ij3+2*%ij1+2*%ij2] [1+2*%ij3+2*%ij1+2*%ij2];
1 [1+%ij3] [1+%ij3+%ij1] [1+%ij3+%ij1+%ij2] [1+%ij3+%ij1+2*%ij2]
[1+%ij3+2*%ij1+2*%ij2] [1+2*%ij3+2*%ij1+2*%ij2];
1 [1+%k1] [1+%k1+%k2] [1+%k1+%k2+%k3] [1+%k1+%k2+%k3+%k4];
[-2*%rho2] [-2*%rho2] [-2*%rho1] 0 [2*%rho1] [2*%rho2] [2*%rho2]
[-2*%rho2] [-2*%rho2] [-2*%rho1] 0 [2*%rho1] [2*%rho2] [2*%rho2]
%z1 %z2 %z3 %z4 %z7
dei 1 2 0 6 7;;1 2 0 3 5;
dei ;1 2 0 6 7;1 2 0 3 5;
dei 1 3 0 5 7;1 3 0 5 7;;
pb 2 3 0 2 3 0 xy [-%epsout] [-%epsout]
```

```
pb 3 2 0 3 2 0 xy [-%epsout] [-%epsout]
pb 6 3 0 6 3 0 xy   %epsout  [-%epsout]
pb 5 2 0 5 2 0 xy   %epsout  [-%epsout]
pb 2 5 0 2 5 0 xy [-%epsout]   %epsout
pb 3 6 0 3 6 0 xy [-%epsout]   %epsout
pb 6 5 0 6 5 0 xy   %epsout    %epsout
pb 5 6 0 5 6 0 xy   %epsout    %epsout
pb 1 3 2 1 3 3 xy [-%epsout] [-%epsout]
pb 3 1 2 3 1 3 xy [-%epsout] [-%epsout]
pb 1 5 2 1 5 3 xy [-%epsout]   %epsout
pb 3 7 2 3 7 3 xy [-%epsout]   %epsout
pb 7 3 2 7 3 3 xy   %epsout  [-%epsout]
pb 5 1 2 5 1 3 xy   %epsout  [-%epsout]
pb 7 5 2 7 5 3 xy   %epsout    %epsout
pb 5 7 2 5 7 3 xy   %epsout    %epsout
de 3 3 1 5 5 4
pb 3 3 0 3 3 0 xy [-%epsin ] [-%epsin]
pb 5 3 0 5 3 0 xy   %epsin   [-%epsin]
pb 3 5 0 3 5 0 xy [-%epsin]    %epsin
pb 5 5 0 5 5 0 xy   %epsin     %epsin
pa 2 3 5 z [%z6+%delout]
pa 3 2 5 z [%z6+%delout]
pa 5 2 5 z [%z7-%delout]
pa 6 3 5 z [%z7-%delout]
pa 3 6 5 z [%z6+%delout]
pa 2 5 5 z [%z6+%delout]
pa 6 5 5 z [%z7-%delout]
pa 5 6 5 z [%z7-%delout]
pa 3 3 5 z [%z6+%delin]
pa 5 3 5 z [%z7-%delin]
pa 3 5 5 z [%z6+%delin]
pa 5 5 5 z [%z7-%delin]
pb 4 0 5 4 0 5 z [.5*(%z6+%z7)]
pa 4 4 4 z %z5
sf  0 0 5 0 0 5  sd 4
sfi -3 -5;-3 -5;;sd 1
sfi -2 -6;-2 -6;;sd 2
sfi -1 -7;-1 -7;;sd 3
endpart
stp .001
merge
```

4. Curved pipe with attachment

```
sd 1 ts 0 0 0 0 0 1 12 0 2
sd 2 cy 0 0 0 1 1 0 1
curd 1 arc3 seqnc rt 1.3414213e+01 -5.8635385e-07 1.4142135e+00
                  rt 9.9687033e+00  8.9758606e+00 1.4142135e+00
                  rt 0.0000000e+00  1.3414213e+01 1.4142135e+00 ;
curd 2 arc3 seqnc rt 0.0000000e+00  1.3414213e+01 -1.4142135e+00
                  rt 8.9758606e+00  9.9687033e+00 -1.4142138e+00
                  rt 1.3414213e+01 -5.8635385e-07 -1.4142135e+00;
curd 3 arc3 seqnc rt 1.0585787e+01 -4.6271944e-07 1.4142135e+00
                  rt 7.8667731e+00  7.0832734e+00 1.4142135e+00
                  rt 0.0000000e+00  1.0585787e+01 1.4142135e+00 ;
curd 4 arc3 seqnc rt 1.0724564e+01 -4.6878557e-07 -1.3905760e+00
                  rt 7.8667731e+00  7.0832734e+00 -1.4142135e+00
                  rt 0.0000000e+00  1.0585787e+01 -1.4142135e+00;
block 1 21 26 46;1 11 31;1 4 9 12;0 0 0 0 10 14 24 -2 -1 1 2
tr 4 1 1 4 3 4 rz -90;
tr 3 1 1 3 3 4 rz -48;
tr 2 1 1 2 3 4 rz -42;
sfi 1 4; -1 -2; -1 -4;sd 1
sfi -2 -3; 2 3; -2 -3;sd 2
tfi 1 2 0 3 4; -1 0 -2; 1 4;
curs 1 2 4 4 2 4 1
curs 1 2 1 4 2 1 2
curs 1 1 4 4 1 4 3
curs 1 1 1 4 1 1 4
mti 1 2 0 3 4; 2 3;; 1
mti 2 3; 2 3; 1 2 0 3 4; 1
mt 1 1 1 4 2 4 4
mt 2 2 2 3 3 3 4
ol 2 3 2 3 3 3
ili -1 0 -4; 1 2;;
endpart
merge
```



5. Intersecting pipes

```
parameter d 3
          i1 [  1+%d*2]
          i2 [%i1+%d*4]
          i3 [  1+%d*4]
          i4 [  1+%d*2];
sd 1 cy 0 0 0 1 0 0 1.8
sd 2 cy 0 0 0 1 0 0 2
sd 3 cy 0 0 0 0 0 1 .5
sd 4 cy 0 0 0 0 0 1 .4
sd 5 plan 0 0 0 0 1 0
sd 6 plan 0 0 0 1 0 0
sd 7 plan 1.5 0 0 1 0 0
sd 8 plan 0 0 0 0 -1 1
sd 9 plan 0 0 2.5 0 0 1
gct 3 rxz; ryz;rxz ryz;
lev 1 grep 0 1 2 3;;
pslv 1
block 1 %i1 %i2;1 %i1 %i2;1 %i4;
      0 .5 1.5 0 .5 1.5 1.8 2
dei 1 2;1 2; ;
sfi 1 -2;1 -2; ;sd 3
sfi ; ;-1;sd 1
sfi ; ;-2;sd 2
sfi ;-1; ;sd 5
sfi -1; ; ;sd 6
sfi -3; ; ;sd 7
sfi ;-3; ;sd 8
endpart
block 1 %i3;1 %i4;1 %i1
%i2;0 .5 .4 .5 1.8 2 3
sfi ;-1; ;sd 4
sfi ;-2; ;sd 3
sfi ; ;-1;sd 1
sfi ; ;-2;sd 2
sfi -2; ; ;sd 5
sfi -1; ; ;sd 6
sfi ; ;-3;sd 9
endpart
pplv
merge
```



6. Two intersecting pipes

```
sd 1 cy 0 0 0 0 0 1 1
sd 2 cy 0 0 0 0 1 0 1
sd 3 plan 0 0 0 0 1 0
sd 4 plan 0 0 0 0 1 1
sd 5 plan 0 0 0 0 1 -1
block 1 3 5 7;1 3 5 7;1 3;-.3 -.3 .3 .3 -.3 -.3 .3 .3 1 2
dei 1 2 0 3 4; 1 2 0 3 4;;
sfi -1 -4; -1 -4;;sd 1
insprt 1 4 2 1
sfi ; -3;;sd 3
sfi -1 0 -4;; -1;sd 2
sfi 2 3; -1 0 -5; -1;sd 2
sfi ; 1 3; -1;sd 4
sfi ; 3 5; -1;sd 5
sd 6 plan .3 -.3 2 0 1 0
sd 7 plan -.3 .3 2 0 1 0
sd 8 plan .3 -.3 2 1 0 0
sd 9 plan -.3 .3 2 1 0 0
sfi -2; 2 4;;sd 9
sfi -3; 2 4;;sd 8
sfi 2 3; -2;;sd 6
sfi 2 3; -4;;sd 7
mseq i 4 6 4
mseq j 4 3 3 4
mseq k 6
lct 3 rx 90;rx 180;rx 270;
lrep 0 1 2 3 ;
endpart
merge
```



7. Turbine casing

```
sd 1 cy 0 0 0 0 0 1 60
sd 2 cy 0 0 0 0 0 1 64
sd 3 cy 0 0 15 0 1 0 10
sd 4 cy 0 0 15 0 1 0 8
sd 5 cy 52 0 55 0 1 0 10
sd 6 cy 52 0 55 0 1 0 8
sd 7 cy 45 0 55 -.25 1 0 14.5
cylinder 1 4;1 16 31 131 135 155 159 186;1 6 10 23 27 38 42 47;
         60 64 45 75 105 310 310 350 350 405 0 10 20 41.5 41.5 70
         70 75
dei ; 4 5 0 6 7; 1 5 0 6 8;
dei ; 1 5 0 6 8; 4 5 0 6 7;
dei ; 5 6; 5 6;
dei ; 2 3; 2 3;
pb 1 6 5 1 6 5 yz 348 51
pb 1 6 6 1 6 6 yz 348 59
pb 1 5 5 1 5 5 yz 320 50
pb 1 5 6 1 5 6 yz 320 60
pb 1 6 8 1 7 8 y 345
pb 1 4 8 1 5 8 y 313
pb 1 7 6 1 7 6 yz 348 66
pb 1 6 7 1 6 7 yz 348 66
pb 1 7 7 1 7 7 yz 348 66
pb 1 6 4 1 6 4 yz 348 44
pb 1 7 5 1 7 5 yz 348 44
pb 1 7 4 1 7 4 yz 348 44
pb 1 4 6 1 4 6 yz 317 63
pb 1 5 7 1 5 7 yz 317 63
pb 1 4 7 1 4 7 yz 317 63
pb 1 5 4 1 5 4 yz 317 47
pb 1 4 5 1 4 5 yz 317 47
pb 1 4 4 1 4 4 yz 317 47
pb 2 6 4 2 6 4 yz 338 50
pb 2 7 5 2 7 5 yz 338 50
pb 2 7 4 2 7 4 yz 338 50
pb 2 6 7 2 6 7 yz 338 60
pb 2 7 6 2 7 6 yz 338 60
pb 2 7 7 2 7 7 yz 338 60
pb 2 5 7 2 5 7 yz 316 63
```







```
pb 2 4 6 2 4 6 yz 316 63
pb 2 4 7 2 4 7 yz 316 63
pb 2 4 5 2 4 5 yz 316 47
pb 2 5 4 2 5 4 yz 316 47
pb 2 4 4 2 4 4 yz 316 47
pb 2 6 6 2 6 6 yz 337 58
pb 2 6 5 2 6 5 yz 337 52
pb 2 5 5 2 5 5 yz 317 49
pb 2 5 6 2 5 6 yz 317 61
pb 2 6 8 2 7 8 y 345
tfi -1; 6 7; 5 6;
sfi -1;;;sd 1
sfi -2;;;sd 2
sfi ; -2 -3; -2 -3;sd 3
sfi ; -5 -6; -5 -6;sd 6
sfi -2; 5 6; -4 0 -7;sd 5
sfi -2; -4 0 -7; 5 6;sd 5
sfi -2;-4 0 -7;-4 0 -7;sd 5
sfi -1; -4 0 -7; 4 7;sd 7
sfi -1; 4 7; -4 0 -7;sd 7
bb 1 3 2 2 3 3 1;
bb 1 2 2 2 2 3 2;
bb 1 2 2 2 3 2 3;
bb 1 2 3 2 3 3 4;
bb 2 6 5 2 7 6 5;
bb 2 4 5 2 5 6 6;
bb 2 5 6 2 6 7 7;
bb 2 5 4 2 6 5 8;
endpart
block 1 4 19 23;1 4 14;1 4 8 12;
      -9.542 -9.542 9.542 9.542 60 64 82 12.1 12.1 18 18
dei 1 2 0 3 4;; 1 2 0 3 4;
dei 2 3; -2; 2 3;
dei 2 3;; 2 3;
bb 2 1 1 3 2 1 3;
bb 2 1 4 3 2 4 4;
bb 1 1 2 1 2 3 1;
bb 4 1 2 4 2 3 2;
sfi -1 -4;; -1 -4;sd 3
sfi -2 -3;; -2 -3;sd 4
sfi ; -1;;sd 1
sfi ; -2;;sd 2
endpart
block 1 5 25 29;1 11;1 5 16 20;
      45 45 59.6 59.6 -60 -30 47.7 47.7 62.5 62.5
```

```
dei 1 2 0 3 4;; 1 2 0 3 4;
dei 2 3;; 2 3;
pb 1 2 1 1 2 4 y -45
pb 2 2 1 2 2 4 y -42
pb 3 2 1 3 2 4 y -22
pb 4 2 1 4 2 4 y -23
pb 2 1 2 2 2 2 xz 4.67812e+01 4.89367e+01
pb 3 1 2 3 2 2 xz 5.91233e+01 5.13589e+01
pb 3 1 3 3 2 3 xz 5.91233e+01 5.86411e+01
pb 2 1 3 2 2 3 xz 4.67812e+01 6.10633e+01
bb 1 2 2 2 2 3 6;
bb 3 2 2 4 2 3 5;
bb 2 2 1 3 2 2 8;
bb 2 2 3 3 2 4 7;
sfi -2 -3;; -2 -3;sd 6
sfi -1 -4;; -1 -4;sd 5
endpart
merge
stp 0.001
```

VI. Circular Objects

1. Ring with holes



```
sd 1 cy 0 11 0 0 0 1 1
sd 2 cy 0 0 0 0 0 1 9
sd 3 cy 0 0 0 0 0 1 13
sd 4 pl3 rt 0 0 0 cy 1 110 0 rt 0 0 1
sd 5 pl3 rt 0 0 0 cy 1 70 0 rt 0 0 1
block 1 5 9 13;1 3 7 9;1 4;
-1 -1 1 1 10 10 12 12 0 2
de 2 2 0 3 3 0
sfi -2 -3;-2 -3;;sd 1
sf 0 1 0 0 1 0 sd 2
sf 0 4 0 0 4 0 sd 3
sf 1 0 0 1 0 0 sd 4
sf 4 0 0 4 0 0 sd 5
res 1 0 0 1 0 0 j 1
res 4 0 0 4 0 0 j 1
lct 8 rz 40;repe 8;lrep 0 1 2 3 4 5 6 7 8;
endpart
merge
```

2. Shell of a ring

```
ld 1 lp2 1 1; lfil 45 1 2 -45 .25 lp2 1 2;
sd 1 crz 1
merge
cylinder -1; 1 91; 1 3; 1; 0 360; 1 2;
sfi -1;; 1 2;sd 1
mseq k 30
mseq k -10
endpart
accuracy 3
cylinder -1; 1 91; 1 3; 1; 0                                  360;
1 2;
sfi -1;; 1 2;sd 1
mseq k 30
mseq k -10
endpart merge
```



3. Spool

```
para
      len 0.8
      ht 1
      ri 0.6
      rf 1.0
      d1 3
      d2 3
      d3 3
      dr 15;
ld 1
  lp2  0 0 %len 0;
  lfil  0    [%len+%rf] [%len]   90    %rf ;
  lp2  [%len+%rf] [%rf+%ht];
  lfil  90   %len [%rf+%ht+%rf]  180   %rf ;
  lp2  0 [%rf+%ht+%rf];
ld 2
 lstl 1 [-%len-%rf-%ri] 0;
sd 1 crz 2
cylinder -1 [-1-%d2] [-1-%d2-%d1];
          -1 [-1-%dr] [-1-%dr-%dr];
          -1 [-1-%d2] [-1-%d2-%d3] [-1-%d2-%d3-%d2];
          %ri [%ri+%rf] [%ri+%rf+%len];
          0 180 360;
          0 %rf [%rf+%ht] [%rf+%ht+%rf];
dei 1 -2 -3; -1 -2 -3; 1 -2 -3 4;
sfi -1 3;; -1 -4;sd 1
endpart
merge
```



4. Automatic transmission tail casing

```
sd 1 cy [1.15*cos(22.5)] [1.15*sin(22.5)] 0 0 0 1 .05
cylinder -1 3;1 -13 25 -37 49;1 -6 -11 -16 -21 26;1 1.2; 0 90 180
         270 360; 0 .5 1.5 2.5 3.5 4;
dei ;-2 0 -4;1 2 0 5 6;
dei ;1 3;-3 0 -5;
dei ;3 5;-2 0 -4;
pb 1 1 1 2 5 1 x .2
pb 1 1 2 1 5 2 x .3
pb 1 1 3 1 5 3 x .5
pb 1 1 4 1 5 4 x .7
pb 1 1 5 1 5 5 x .9
pb 2 1 2 2 5 2 x .4
pb 2 1 3 2 5 3 x .6
pb 2 1 4 2 5 4 x .8
pb 2 1 5 2 5 5 x 1
endpart
lev 1 levct 1 xsca .2 ysca .2 mz -4.2; grep 0;;
pslv 1
cylinder 1 3 5 7;1 3 5 7;1 4;1 1.1 1.2 1.3;0 15 30 45;4 4.2;
dei 2 3;2 3; ;
sfi -2 -3;-2 -3; ;sd 1
lct 7 rz 45;repe 7 ;;
lrep 0 1 2 3 4 5 6 7;
endpart
pplv
merge
```



5. Chain link



```
parameter ep .01;
sd 1 cy 0 0 0 0 0 1 .1
sd 2 cy 0 0 0 0 0 1 [.2-%ep]
sd 3 plan [%ep-.2] 0 0 1 0 0
sd 4 plan [.2-%ep] 0 0 1 0 0
sd 5 plan 0 -.2 0 0 1 0
gct 3 rxy;mz .03 my -.4 ;mz .03  my -.4 rxy;
lev 1 levct 4 my -.8;repeat 4; grep 0;;
pslv 1
cylinder 1 2;1 3 10 12 15;1 2;.1 .2 -45 0 180 225
315 .1 .13
sfi -1;;;sd 1
sfi -2; 2 3;;sd 2
sfi -2; 1 2;;sd 4
sfi -2; -1;;sd 5
sfi -2; 3 4;;sd 3
sfi -2;4 5;;sd 5
sfi -2; -5;;sd 4
grep 0 1 2 3;
lct 1 rxz my -.4;lrep 0 1;
endpart
block 1 4;1 3 5;1 2 3 6 7 8;-.1 .1 -.1 0 .1254
-.16 -.13 -.1 .1 .13 .16
sfi -1 -2;-1 -3;;sd 1
lct 2 my -.4;rz 180 my -.8;lrep 1 2;
endpart
pplv
merge
```

6. Lid seal

```
cylinder 1 2 3 0 -5 0 7 8 9;
         1 31;1 2 3 4 5;
         10 11 11.5 0 11.5 0 11.5 13 14
         0 360
         0 1 2 3 4
dei 1 2;; 4 5;
dei 2 3;; 2 5;
dei 7 8;; 2 5;
dei 8 9;; 3 5;
dei -5;; 1 2;
endpart merge
```



7. Gear made with simple sprockets

```
sd 1 cy 0 12 0 0 0 1 .8
sd 2 cy 0 0 0 0 0 1 10
sd 3 cy 0 0 0 0 0 1 12
sd 4 pl3 rt 0 0 0 cy 1 95 0 rt 0 0 1
sd 5 pl3 rt 0 0 0 cy 1 85 0 rt 0 0 1
sd 6 cy 0 6 0 0 0 1 .8
sd 7 cy 0 0 0 0 0 1 4
sd 8 cy 0 0 0 0 0 1 6
sd 9 pl3 rt 0 0 0 cy 1 100 0 rt 0 0 1
sd 10 pl3 rt 0 0 0 cy 1 80  0 rt 0 0 1
gct 4 mx 17.5;mx 17.5 rz 90;mx 17.5 rz 180;mx 17.5 rz 270;
lev 1 grep 1 2 3 4;;
block 1 2 3 4 5;1 3 5;1 3;-1.2 -.6 0 .6 1.2 10 11 12 0 .5
de 2 2 0 4 3 0
sfi -2 -4;-2 -3;;sd 1
sf 0 1 0 0 1 0 sd 2
sf 0 3 0 0 3 0 sd 3
sf 1 0 0 1 0 0 sd 4
sf 5 0 0 5 0 0 sd 5
lct 35 rz 10;repe 35;
lrep  0  1  2  3  4  5  6  7  8  9 10 11 12 13 14 15 16 17 18 19
     20 21 22 23 24 25 26 27 28 29 30 31 32 33 34 35;
endpart
pslv 1
block 1 2 3 4 5;1 3 5;1 3;-1.2 -.5 0 .5 1.2 4 5 6 0 .5
de 2 2 0 4 3 0
sfi -2 -4;-2 -3;;sd 6
sf 0 1 0 0 1 0 sd 7
sf 0 3 0 0 3 0 sd 8
sf 1 0 0 1 0 0 sd 9
sf 5 0 0 5 0 0 sd 10
lct 17 rz 20;repe 17;
lrep  0  1  2  3  4  5  6  7  8  9 10 11
12 13 14 15 16 17;
endpart
pplv
merge
```



VII. Miscellaneous Objects

1. Dodecahedron membrane

```
parameter a -116.565;
lev 1 levct 1 rxy rz 36 mz [-2*(cos(36)+1)*sin(116.565)-2];
grep 0;;
gct  5 mx [-2*cos(36)] ry %a mx [2*cos(36)];
       mx [-2*cos(36)] ry %a mx [2*cos(36)] rz 72;
       mx [-2*cos(36)] ry %a mx [2*cos(36)] rz 144;
       mx [-2*cos(36)] ry %a mx [2*cos(36)] rz 216;
       mx [-2*cos(36)] ry %a mx [2*cos(36)] rz 288;
pslv 1
block 1 2 3;1 2 3;-1;0 1 2 0 1 2 0
tr 1 1 1 3 3 1 rz -45;
dei 2 3; 2 3; -1;
pb 1 3 1 1 3 1 xy [2*cos(36)] [2*sin(36)]
pb 3 1 1 3 1 1 xy [2*cos(36)] [-2*sin(36)]
pb 2 3 1 2 3 1 xy [2*cos(36)] 0
pb 3 2 1 3 2 1 xy [2*cos(36)] 0
lin 1 1 1 1 3 1
lin 1 1 1 3 1 1
pb 2 2 1 2 2 1 x 1.078689e+00
lct 4 rz 72;repeat 4;
lrep 0 1 2 3 4;
grep 0 1 2 3 4 5;
endpart
pplv
merge
stp 0.0001
```



2. Pyramid

```
sd 1 plan 1 .70710671 2.1213205 0 1 0
sd 2 cy 2 3 2.1213205 0 1 0 1
sd 3 pl3 rt 2 3 2.1213205
         rt 3 .70710671 2.1213205
         rt 2 .70709592 3.1213160
sd 4 pl3 rt 2 3 2.1213205
         rt 3 .70710671 2.1213205
         rt 2 .70709592 1.1213250
sd 5 pl3 rt 2 3 2.1213205
         rt 2 .70709354 1.1213237
         rt 1 .70710671 2.1213205
sd 6 pl3 rt 2 3 2.1213205
         rt 1 .70710671 2.1213205
         rt 2 .70709354 3.1213176
sd 7 plan 0 -1 0 0 1 0
sd 9 plan 0 .7 0 0 1 0
sd 10 plan 2 3 2.1213205 1 0 1
sd 11 plan 2 3 2.1213205 -1 0 1
block 1 2 3;1 2;1 2;1 2 3 1 2 1 2
tr 1 1 1 3 2 2 rx 45;
pb 1 1 2 3 1 2 y 7.0710671e-01
mb 1 1 2 1 1 2 x .5
mb 3 1 2 3 1 2 x -.5
mb 1 1 2 3 1 2 y -.75
mb 2 1 2 2 1 2 y .25
sfi -2;-2;-2;sd 1
sfi -2;-1;-1;sd 1
sfi -1 0 -3;-2;-1;sd 1
sfi ;-2;-2;sd 2
sfi ;-1;-1;sd 2
sfi -1;-2;-1;sd 2
sfi 2 3;-2;;sd 3
sfi 2 3;;-1;sd 4
sfi 1 2;;-1;sd 5
sfi 1 2;-2;;sd 6
sfi -1 0 -3;-2;-1;sd 9
sfi -2;-2;-2;sd 9
sfi -2;-1;-1;sd 9
sfi -3;-1;-1;sd 10
sfi -1;-2;-2;sd 10
sfi -3;-2;-2;sd 11
sfi -1;-1;-1;sd 11
endpart
```



```
block 1 2 3 4 5;1 2;1 2 3;1 1.5 2 2.5 3 -1 0 1 2 3
dei 1 2 0 4 5;;1 2;
pb 4 2 2 4 2 2 xy 2.5 -.04289329
pb 3 2 2 3 2 2 xy 2    .2071067
pb 2 2 2 2 2 2 xy 1.5 -.04289329
pb 2 1 2 4 2 2 z 2.1213202
sfi 3 5;;-3;sd 3
sfi 3 -5;;-1 3;sd 4
sfi -1 3;;-1 3;sd 5
sfi 1 3;;-3;sd 6
sfi ;-1;;sd 7
sfi -2 0 -4;-2;-1 0 -3;sd 2
sfi -1 0 -5;-2;-2;sd 2
sfi -1 0 -5;-2;-3;sd 9
sfi -3;-2; -1 0 -3;sd 9
sfi -4;;-1;sd 10
sfi -5;;-2;sd 10
sfi -2;;-3;sd 10
sfi -1;;-2;sd 11
sfi -2;;-1;sd 11
sfi -4;;-3;sd 11
endpart
merge
```

3. Strut

```
sd 1 cone 0 0 0 0 0 1 0      35.311
sd 2 cone 0 0 0 0 0 1 5.667 -35.311
sd 3 cone 0 0 0 0 0 1  .75   35.311
sd 4 cone 0 0 0 0 0 1 6.417 -35.311
sd 5 plan 0 0 0 0 0 1
sd 6 plan 0 0 8 0 0 1
cylinder
1 2 6 7 8 ; 1 3 5 6 8 10 ; 1 3 4 6 7 9 ;
4.25 5 10 10.5 11 0 8 13.5 16.5 22 30 0 2 3 5 6 8
dei 2 3 ; 1 3 0 4 6 ; ;
dei 2 3 ; ; 1 3 0 4 6 ;
dei 3 4 ; 1 2 0 5 6 ; ;
dei 3 4 ; 2 5 ; 1 2 0 5 6 ;
dei 1 2 ; 1 2 0 5 6 ; ;
pb 1 2 1 2 2 6 2 0
pb 1 5 1 2 5 6 2 30
mb 2 3 1 2 3 6 2 -1.3
mb 2 4 1 2 4 6 2  1.3
sf 1 1 1 1 6 2 sd 2
sf 1 1 5 1 6 6 sd 1
sf 2 1 1 2 6 2 sd 4
sf 2 1 5 2 6 6 sd 3
sf 1 1 1 5 6 1 sd 5
sf 1 1 6 5 6 6 sd 6
endpart
merge
```



4. Rear hub

```
para
  r 7.0
  s 1
  t 1
r0  [%r+0.5*cos(  0)]
r1  [%r+0.5*cos( 10)]
r2  [%r+0.5*cos( 20)]
r3  [%r+0.5*cos( 30)]
r4  [%r+0.5*cos( 40)]
r5  [%r+0.5*cos( 50)]
r6  [%r+0.5*cos( 60)]
r7  [%r+0.5*cos( 70)]
r8  [%r+0.5*cos( 80)]
r9  [%r+0.5*cos( 90)]
r10 [%r+0.5*cos(100)]
r11 [%r+0.5*cos(110)]
r12 [%r+0.5*cos(120)]
r13 [%r+0.5*cos(130)]
r14 [%r+0.5*cos(140)]
r15 [%r+0.5*cos(150)]
r16 [%r+0.5*cos(160)]
r17 [%r+0.5*cos(170)]
r18 [%r+0.5*cos(180)];
ld 1 lp
[%r0 *sin(  0)] [%r0 *cos(  0)]
[%r1 *sin( -1)] [%r1 *cos( -1)]
[%r2 *sin( -2)] [%r2 *cos( -2)]
[%r3 *sin( -3)] [%r3 *cos( -3)]
[%r4 *sin( -4)] [%r4 *cos( -4)]
[%r5 *sin( -5)] [%r5 *cos( -5)]
[%r6 *sin( -6)] [%r6 *cos( -6)]
[%r7 *sin( -7)] [%r7 *cos( -7)]
[%r8 *sin( -8)] [%r8 *cos( -8)]
[%r9 *sin( -9)] [%r9 *cos( -9)]
[%r10*sin(-10)] [%r10*cos(-10)]
[%r11*sin(-11)] [%r11*cos(-11)]
[%r12*sin(-12)] [%r12*cos(-12)]
[%r13*sin(-13)] [%r13*cos(-13)]
[%r14*sin(-14)] [%r14*cos(-14)]
[%r15*sin(-15)] [%r15*cos(-15)]
[%r16*sin(-16)] [%r16*cos(-16)]
[%r17*sin(-17)] [%r17*cos(-17)]
[%r18*sin(-18)] [%r18*cos(-18)]
sd 1 cy 0 0 0 0 0 1 6.625
sd 2 cy 0 0 0 0 0 1 3.23
```







```
sd 3 cy 0.0 5.56 0.0 0 0 1 0.507
sd 4 cy 0.0 3.5 0.0 0 0 1 0.2735
sd 5 cy 0 3.5 0 0 0 1 0.6349
sd 6 cy -1.15 3.6 0 0 0 1 0.562
sd 7 ts 0 0 5.81 0 0 1 1.2 0 3.14
sd 8 cy 0 0 0 0 0 1 2.375
sd 9 ts 0 0 5.81 0 0 1 1.59 0 2.0
sd 10 ts 0 0 5.81 0 0 1 2.15 0 1.44
sd 11 cp 1 rx -90 ;
sd 12 plan 0 0 0 1 0 0
sd 13 plan 0 0 0 [cos(18)] [sin(18)] 0
sd 14 plan 0 0 6.44 0 0 1
sd 15 plan 0 0 7.94 0 0 1
sd 16 cy 0 0 0 0 0 1 4.34
gct 1 ryz ; lev 1 grep 0 1 ; ; pslv 1
para
i2 [  1+max(%t,2*%s)]
i3 [%i2+max(%t,  %s)]
i4 [%i3+max(%t,2*%s)]
i5 [%i4+max(%t,3*%s)]
i6 [%i5+max(%t,3*%s)]
i7 [%i6+max(%t,4*%s)]
i8 [%i7+max(%t,5*%s)]
j2 [  1+max(%t,2*%s)]
j3 [%j2+max(%t,6*%s)]
k2 [  1+max(%t,2*%s)]
k3 [%k2+max(%t,2*%s)]
k4 [%k3+max(%t,3*%s)]
k5 [%k4+max(%t,3*%s)]
k6 [%k5+max(%t,3*%s)] ;
cylinder
1 %i2 %i3 %i4 %i5 %i6 %i7 %i8 ;
1 %j2 %j3 ;
1 %k2 %k3 %k4 %k5 %k6 ;
2.59 3.096 3.257 3.8873 4.34 5.053 6.067 6.625
90.0 94.5 108.0
5.81 6.44 7.25 7.94 8.9 10.0
pb 6 2 2 7 2 4 y 96.0
pa 4 0 6 x 3.5383
pa 4 0 3 x 3.4
mb 7 2 1 7 2 6 xy .6 .6
mb 4 1 3 4 3 3 yz .6 -.6
de 5 0 1 8 0 2
de 1 0 1 4 0 3
de 1 0 4 2 0 6
de 2 0 5 3 0 6
```

```
de 4 0 4 8 0 6
de 6 1 2 7 2 4
sfi -6 -7 ; 1 -2 ; 2 4 ;sd 3
sfi 1 -4 ; ; 1 -3 ; sd 10
sfi -8;1 3;1 6;sd 1
sfi  ;-1; ;sd 12
sfi  ;-3; ;sd 13
sfi 4 8; ;-4;sd 15
sfi 5 8; ;-2;sd 14
sfi -5; ;1 2;sd 16
res 8 1 2 8 3 4 j 1
res 7 1 2 8 1 4 k 1
res 5 3 2 8 3 2 i 1
res 4 3 4 8 3 4 i 1
res 1 1 3 4 3 3 i 1
res 4 1 1 4 3 3 k 1
res 5 2 2 8 3 4 k 1
res 5 1 2 6 1 4 k 1
relaxi 1 8;-3;1 6;5 0.001 1
relaxi 1 5;-1;1 4;5 0.001 1
relaxi 4 8; ;-4;5 0.001 1
relaxi 5 8; ;-2;5 0.001 1
lct 10 rz 0 ; rz 36 ; rz 72 ; rz 108 ; rz 144 ;
ryz rz 36 ; ryz rz 72 ; ryz rz 108 ; ryz rz 144 ; ryz rz 180 ;
  lrep 1 2 3 4 5 6 7 8 9 10 ;
endpart
para i2 [   1+max(%t, 2*%s)]
     i3 [%i2+max(%t, 2*%s)]
     i4 [%i3+max(%t,   %s)]
     i5 [%i4+max(%t, 2*%s)]
     j2 [   1+max(%t, 4*%s)]
     j3 [%j2+max(%t, 4*%s)]
     j4 [%j3+max(%t, 2*%s)]
     k2 [   1+max(%t,10*%s)]
     k3 [%k2+max(%t, 2*%s)]
     k4 [%k3+max(%t, 3*%s)]
     k5 [%k4+max(%t, 7*%s)] ;
cylinder
1 %i2 %i3 %i4 %i5 ;
1 %j2 %j3 %j4;
1 %k2 %k3 %k4 %k5 ;
2.375 2.865 3.2265 3.7734 4.1349
90.0 96.75 103.5 112.5
0. 2.75 3.444 3.949 5.81
de 1 0 1 2 0 3
de 1 0 4 3 0 5
```

```
de 3 1 1 4 2 2
ma 5 2 4 y 2.5
ma 5 3 4 y 1.0
mb 3 2 1 3 2 2 xy .1 5
mb 4 2 1 4 2 2 xy -.1 5
sf 5 1 1 5 2 4 sd 5
sf 5 2 1 5 3 3 sd 6
sf 5 3 1 5 4 3 sd 2
sfi -3 -4 ; 1 -2 ; 1 2 ;sd 4
sf 1 0 3 1 0 4 sd 8
sfi 1 -3 ; ; -4 5 ; sd 9
sf 5 3 3 5 4 5 sd 7
sf 5 0 5 5 0 5 sd 7
lin 3 4 4 5 4 4
lin 2 4 1 5 4 1
lin 2 4 1 5 4 4
lin 2 2 1 5 4 4
lin 2 3 1 5 3 1
lin 2 2 1 5 3 1
lin 2 3 1 5 4 1
res 3 0 4 5 0 5 i 1.0
res 2 0 3 5 0 3 i 1.0
res 2 3 1 5 4 2 i 1.0
res 3 1 5 5 4 5 j 1
res 3 1 4 5 4 4 j 1
res 1 1 3 1 4 4 j 1
res 2 1 1 2 4 4 j 1
relaxi ;-1;2 5;5 0.001 1
relaxi ;-4; ;5 0.001 1
lct 8 rz 0 ; rz 45 ; rz 90 ; rz 135 ;
ryz rz 45 ; ryz rz 90 ; ryz rz 135 ; ryz rz 180 ;
  lrep 1 2 3 4 5 6 7 8 ;
endpart
pplv
merge
```

5. Finite element model of concentric spheres with cylindrical holes

```
parameter a .176776695;
sd 1 sp 0 0 0 .45 sd 2 cy 0 0 0 1 0 0 .25
sd 3 cy 0 0 0 0 1 0 .25 sd 4 cy 0 0 0 0 0 1 .25
sd 5 plan 0 0 0 1 0 0 sd 6 plan %a 0 0 1 0 0
sd 7 plan [-%a] 0 0 1 0 0 sd 8 plan 0 %a 0 0 1 0
sd 9 plan 0 0 %a 0 0 1
gct 4 ry 90;rz 90;csca 2 ry 45 rx 45;ryz;
lev 1 grep 0 3;;
pslv 1
block 1 6 11;1 6;1 6;[-%a] 0 %a %a .45 %a .45
sf 1 0 0 1 0 0 sd 7 sf 2 0 0 2 0 0 sd 5
sf 3 0 0 3 0 0 sd 6 sf 0 1 0 0 1 0 sd 4
sf 0 2 0 0 2 0 sd 1 sf 0 0 1 0 0 1 sd 3
sf 0 0 2 0 0 2 sd 1 mate 1
lct 3 rx 90;rx 180;rx 270;lrep 0 1 2 3;repe 0 1 2;
endpart
block 1 6;1 6;1 6;.17677 .45 .17677 .45
.17677 .45
sf 1 0 0 1 0 0 sd 6 sf 0 1 0 0 1 0 sd 8
sf 0 0 1 0 0 1 sd 9 sf 2 0 0 2 0 0 sd 1
sf 0 2 0 0 2 0 sd 1 sf 0 0 2 0 0 2 sd 1
mate 2 lrep 0 1 2 3;repe 0 4;
endpart
pplv
block 1 8;1 8;1 8;-.2 .2 -.2 .2 -.2 .2
sfi -1 -2;-1 -2;-1 -2;sp 0 0 0 .2
mate 3
endpart
merge
```







6. Two spheres intersecting

```
sd 1 sphe 0 0 0 3
sd 2 sphe 0 0 5.8 3
curd 1 twsurf 1 2
 -5.4983681e-01 -9.4525093e-01 3.0064039e+00
 -9.2336166e-01 -4.8572582e-01 2.9872661e+00
 -1.0761518e+00 1.0624153e-01 3.0016773e+00
 -7.6276845e-01 8.7529075e-01 3.0337660e+00
 -1.9322823e-01 1.1674269e+00 3.0432305e+00
 4.1287965e-01 1.3024358e+00 3.1291969e+00
 9.1572303e-01 9.8007196e-01 3.1165490e+00
 1.2294044e+00 4.8923206e-01 3.1075633e+00
 1.1044415e+00 -1.6500537e-01 3.0155823e+00
 9.9821901e-01 -7.3316890e-01 3.0675979e+00
 5.0750321e-01 -1.1908860e+00 3.0936687e+00
 -2.8473176e-02 -1.2566353e+00 3.0760219e+00
 -5.4983681e-01  -9.4525093e-01 3.0064039e+00;;
curd 2 cpcds 1 ; mz 10;
curd 3 cpcds 1 ; mz -10;
sd 3 rule3d 2 3;
sd 4 plan 0 0 0 1 1 0
sd 5 plan 0 0 0 1 -1 0
sd 6 plan 0 0 2.9 0 0 1
sd 7 plan 2.9 0 0 1 0 0
block 1 5 9 13;
      1 5 9 13;
      1 5 9 13;
      -1 -1 1 1;
      -1 -1 1 1;
      -1 -1 1 1;
dei 1 2 0 3 4; 1 2 0 3 4; 1 2 0 3
4;
dei 2 3; 1 2 0 3 4; 1 2 0 3 4;
dei 1 2 0 3 4; 2 3; 1 2 0 3 4;
dei 1 2 0 3 4; 1 2 0 3 4; 2 3;
sfi -1 -4; -1 -4; -1 -4;sd 1
mt 2 2 1 3 3 4 0
bb 3 2 1 3 3 4 1;
bb 2 3 1 3 3 4 2;
bb 2 2 1 2 3 4 3;
bb 2 2 1 3 2 4 4;
bb 4 2 2 4 3 3 5;
bb 2 4 2 3 4 3 6;
bb 1 2 2 1 3 3 7;
```



```
bb 2 1 2 3 1 3 8;
lct 1 mz 5.8;
lrep 0 1;
endpart
cylinder 1 5;
         1 5 9 13 17;
         1 13;
         .5 2;
         -45 45 135 225 315;
         -2 2;
pb 2 1 1 2 5 2 x 2
pb 1 1 1 1 5 2 x .5
bb 2 1 1 2 2 2 1;
bb 2 2 1 2 3 2 2;
bb 2 3 1 2 4 2 3;
bb 2 4 1 2 5 2 4;
sfi ;; -1 0 -2;sd 1
sfi -1;;;sd 3
bb 1 1 2 2 2 2 21;
bb 1 1 1 2 2 1 22 mz 5.8;
bb 1 1 1 2 2 1 23 mz -5.8;
lct 1 mz 5.8;
lrep 0 1;
endpart merge
block 1 5;1 5;1 5;0 0;0 0;0 0;
pb 1 2 1 1 2 1 xyz 1.732050e+00 -1.732051e+00 1.732051e+00
pb 1 2 2 1 2 2 xyz 1.732051e+00 1.732051e+00 1.732051e+00
pb 1 1 1 1 1 1 xyz 5.434830e-01 -5.427358e-01 2.900004e+00
pb 1 1 2 1 1 2 xyz 5.432075e-01 5.429946e-01 2.900007e+00
pb 2 1 1 2 1 1 xyz 1.732050e+00 -1.732051e+00 4.067949e+00
pb 2 1 2 2 1 2 xyz 1.732051e+00 1.732051e+00 4.067949e+00
pb 2 2 2 2 2 2 xyz 2.99250 .413241 2.78001
pb 2 2 1 2 2 1 xyz 2.97117 -.890524e-01 2.77913
pb 2 2 1 2 2 1 xyz 3.14060 -1.92010 2.72391
pb 2 2 2 2 2 2 xyz 2.81902 2.24970 2.80606
bb 1 1 1 1 2 2 21;
bb 1 1 1 2 1 2 22;
sfi ;; -2;sd 5
sfi ;; -1;sd 4
sfi -2; -2;;sd 6
sfi -2; -2;;sd 7
pb 2 2 1 2 2 2 x [sqrt(2)*2.9]
lct 3 rz 90; rz 180; rz 270;
lrep 0 1 2 3;
bb 1 2 1 2 2 2 31;
```

```
bb 2 1 1 2 2 2 32;
endpart
merge
```

7. Mushroom with bricks and shells in one part

```
sd 1 cy 0 0 0 0 0 1 3
sd 2 sp 0 0 0 6
sd 3 plan 0 0 .1 0 0 1
sd 4 cy 0 0 0 0 0 1 1
sd 5 plan 0 0 -2 0 0 1
block 1 8 13 17 22 29;
      1 8 13 17 22 29;
      1 8 0 -10;
      -1 -1 -1 1 1 1
      -1 -1 -1 1 1 1
      0 2 0 2
dei 1 3 0 4 6; 1 3 0 4 6;;
dei 1 2 0 5 6;; 1 2;
dei ; 1 2 0 5 6; 1 2;
dei 2 5; 2 5; -4;
sfi -2 -5; -2 -5;;sd 1
sfi ;; -2 0 -4;sd 2
sfi -1 0 -6;; -4;sd 3
sfi ; -1 0 -6; -4;sd 3
sfi -3 0 -4; -3 0 -4;;sd                                   4
sfi -2 -5; -2 -5; -4;sd 1
sfi ;; -1;sd 5
endpart
merge
```



8. Rim

```
ld 1 lp2 5.5 12 .5 12;lad .5 11.5 45 ;
ld 2 lep .5 .5 .5 11.5 135 180 0
     lp2 0 9.198;lad .5 9.198 22.142 ;
ld 3 lp2 .03687 9.015;lad .5 9.198 22.142
     lp2 2.65 6.35;lar 2.79 6 -.5;
ld 4 lstl 3 0 0 ; lscz 4 -1 lt 4 0 12
ld 5 lstl 2 0 0 ; lscz 5 -1 lt 5 0 12;
ld 6 lstl 1 0 0 ; lscz 6 -1 lt 6 0 12
ld 7 lp2 5.5 11;lep .5 .5 1.5 10.5 90 135 0 ;
ld 8 lep .5 .5 1.5 10.5 135 180 0
     lp2 1 9.19838;lad 1.5 9.19838 22.142 ;
ld 9 lp2 1.0369 9.01;lar 1.146 8.854 .5
     lp2 3.65256 6.347329;lar 3.79289 6 -.5;
ld 10 lstl 9 0 0;lscz 10 -1 lt 10 0 12
ld 11 lstl 8 0 0;lscz 11 -1 lt 11 0 12
ld 12 lstl 7 0 0;lscz 12 -1 lt 12 0 12
ld 13 lep .5 .5 5.5 11.5 -90 90 0 ;
ld 14 lep .5 .5 5.5 .5 -90 90 0 ;
ld 15 lp2 50 0; lfil 90 -50 30 21.8 10
      lfil 21.8 -50 -40 -90 10 ;
      lfil -90 50 -70 -16.7 10
      lfil -16.7 50 0 -90 10
      lp2 50 0; ;
ld 21 lp2 6 11.5; lint 1  7 .5;
ld 22 lint 2  8 .5;
ld 23 lint 3  9 .5;
ld 24 lint 4 10 .5;
ld 25 lint 5 11 .5;
ld 26 lp2 6 .5;lint 6 12 .5;
curd  1 ld2d3d  1 rt y 0 0 ;
curd  2 ld2d3d  2 rt y 0 0 ;
curd  3 ld2d3d  3 rt y 0 0 ;
curd  4 ld2d3d  4 rt y 0 0 ;
curd  5 ld2d3d  5 rt y 0 0 ;
curd  6 ld2d3d  6 rt y 0 0 ;
curd  7 ld2d3d  7 rt y 0 0 ;
curd  8 ld2d3d  8 rt y 0 0 ;
curd  9 ld2d3d  9 rt y 0 0 ;
curd 10 ld2d3d 10 rt y 0 0 ;
curd 11 ld2d3d 11 rt y 0 0 ;
curd 12 ld2d3d 12 rt y 0 0 ;
curd 13 ld2d3d 13 rt y 0 0 ;
curd 14 ld2d3d 14 rt y 0 0 ;
```



```
curd 15 ld2d3d 15 rt z 0 0 rx 180;
sd  1 swept 15 r  1 0  1 1 ;rx 90;
sd  2 swept 15 r  2 0  2 1 ;rx 90;
sd  3 swept 15 r  3 0  3 1 ;rx 90;
sd  4 swept 15 r  4 0  4 1 ;rx 90;
sd  5 swept 15 r  5 0  5 1 ;rx 90;
sd  6 swept 15 r  6 0  6 1 ;rx 90;
sd  7 swept 15 r  7 0  7 1 ;rx 90;
sd  8 swept 15 r  8 0  8 1 ;rx 90;
sd  9 swept 15 r  9 0  9 1 ;rx 90;
sd 10 swept 15 r 10 0 10 1 ;rx 90;
sd 11 swept 15 r 11 0 11 1 ;rx 90;
sd 12 swept 15 r 12 0 12 1 ;rx 90;
sd 13 swept 15 r 13 0 13 1 ;rx 90;
sd 14 swept 15 r 14 0 14 1 ;rx 90;
sd 21 swept 15 r 21 0 21 1 ;rx 90;
sd 22 swept 15 r 22 0 22 1 ;rx 90;
sd 23 swept 15 r 23 0 23 1 ;rx 90;
sd 24 swept 15 r 24 0 24 1 ;rx 90;
sd 25 swept 15 r 25 0 25 1 ;rx 90;
sd 26 swept 15 r 26 0 26 1 ;rx 90;
curd 16 sdedge  1.1;
curd 17 sdedge  2.1;
curd 18 sdedge  3.1;
curd 19 sdedge  4.3;
curd 20 sdedge  5.3;
curd 21 sdedge  6.3;
curd 22 sdedge  7.1;
curd 23 sdedge  8.1;
curd 24 sdedge  9.1;
curd 25 sdedge 10.3;
curd 26 sdedge 11.3;
curd 27 sdedge 12.3;
curd 28 sdedge 14.1;
curd 29 sdedge 14.3;
curd 31 sdedge 21.1;
curd 32 sdedge 22.1;
curd 33 sdedge 23.1;
curd 34 sdedge 23.3;
curd 35 sdedge 24.1;
curd 36 sdedge 25.1;
curd 37 sdedge 26.1;
block 1 2 0 -3 0 4 5;
      1 26 33 40 77 81 86 110 114 118 157 163 168 193;
      1 4 6 9 12 14 17;
```

```
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
pb 1 1 7 1 0 7 xz 5.5       12
pb 1 1 6 1 0 6 xz 0.1059953  11.807832
pb 1 1 5 1 0 5 xz 0.0332094  9.0279236
pb 1 1 4 1 0 4 xz 2.7928922  6
pb 1 1 3 1 0 3 xz 0.0397480  2.9882596
pb 1 1 2 1 0 2 xz 0.1589992  0.1343248
pb 1 1 1 1 0 1 xz 5.5       0
pb 2 1 7 6 0 7 xz 5.5       11.5
pb 2 1 6 6 0 6 xz [(1.1226459+0.1059953)/2]
[(10.828031+11.807832)/2]
pb 2 1 5 6 0 5 xz [(1.0468456+0.0332094)/2]
[(8.9957981+9.0279236)/2]
pb 2 1 4 6 0 4 xz [(3.7928166+2.7928922)/2]  [(5.9912739+6)/2]
pb 2 1 3 6 0 3 xz [(1.0397480+0.0397480)/2]
[(2.9882603+2.9882596)/2]
pb 2 1 2 6 0 2 xz [(1.1343218+0.1589992)/2]
[(1.1590023+0.1343248)/2]
pb 2 1 1 6 0 1 xz 5.5 .5
pb 7 1 7 7 0 7 xz 5.5 11
pb 7 1 6 7 0 6 xz 1.1226459  10.828031
pb 7 1 5 7 0 5 xz 1.0468456  8.9957981
pb 7 1 4 7 0 4 xz 3.7928166  5.9912739
pb 7 1 3 7 0 3 xz 1.0397480  2.9882603
pb 7 1 2 7 0 2 xz 1.1343218  1.1590023
pb 7 1 1 7 0 1 xz 5.5 1
mb 1  1 1 7  1 7 xy 5.0000000e+01 0.0000000e+00
mb 1  2 1 7  2 7 xy 5.0000000e+01 5.6559502e+01
tr 1  3 1 7  3 7 rz [111.8/2] v 4.6383636e+01 6.4256866e+01 0;
tr 1  4 1 7  4 7 rz 111.8 v 3.7295105e+01 6.6186745e+01 0;
tr 1  5 1 7  5 7 rz 111.8 v -4.3207474e+01 4.2030010e+01 0;
tr 1  6 1 7  6 7 rz [(111.8+180)/2] v -4.7292480e+01
3.9400856e+01 0;
tr 1  7 1 7  7 7 rz 180 v -4.9998459e+01 3.2733604e+01 0;
tr 1  8 1 7  8 7 rz 180 v -4.9998459e+01 -2.3404497e+01 0;
tr 1  9 1 7  9 7 rz [(180+253.3)/2] v -4.8377472e+01
-2.8690073e+01 0;
tr 1 10 1 7 10 7 rz 253.3 v -4.3875618e+01 -3.2447933e+01 0;
tr 1 11 1 7 11 7 rz 253.3 v 3.6780914e+01 -6.4694916e+01 0;
tr 1 12 1 7 12 7 rz [(253.3+360)/2] v 4.6781811e+01
-6.2576065e+01 0;
mb 1 13 1 7 13 7 xy 4.9998451e+01 -5.5402985e+01
mb 1 14 1 7 14 7 xy 5.0000000e+01 0.0000000e+00
curs 1 1 7 1 14 7 16
curs 1 1 6 1 14 6 17
```

```
curs 1 1 5 1 14 5 18
curs 1 1 4 1 14 4 19
curs 1 1 3 1 14 3 20
curs 1 1 2 1 14 2 21
curs 1 1 1 1 14 1 28
curs 2 1 7 2 14 7 31
curs 2 1 6 2 14 6 32
curs 2 1 5 2 14 5 33
curs 2 1 4 2 14 4 34
curs 2 1 3 2 14 3 35
curs 2 1 2 2 14 2 36
curs 2 1 1 2 14 1 37
curs 4 1 7 4 14 7 31
curs 4 1 6 4 14 6 32
curs 4 1 5 4 14 5 33
curs 4 1 4 4 14 4 34
curs 4 1 3 4 14 3 35
curs 4 1 2 4 14 2 36
curs 4 1 1 4 14 1 37
curs 6 1 7 6 14 7 31
curs 6 1 6 6 14 6 32
curs 6 1 5 6 14 5 33
curs 6 1 4 6 14 4 34
curs 6 1 3 6 14 3 35
curs 6 1 2 6 14 2 36
curs 6 1 1 6 14 1 37
curs 7 1 7 7 14 7 22
curs 7 1 6 7 14 6 23
curs 7 1 5 7 14 5 24
curs 7 1 4 7 14 4 25
curs 7 1 3 7 14 3 26
curs 7 1 2 7 14 2 27
curs 7 1 1 7 14 1 29
sfi -1;;6 7;sd  1
sfi -1;;5 6;sd  2
sfi -1;;4 5;sd  3
sfi -1;;3 4;sd  4
sfi -1;;2 3;sd  5
sfi -1;;1 2;sd  6
sfi -2 0 -4 0 -6;;6 7;sd 21
sfi -2 0 -4 0 -6;;5 6;sd 22
sfi -2 0 -4 0 -6;;4 5;sd 23
sfi -2 0 -4 0 -6;;3 4;sd 24
sfi -2 0 -4 0 -6;;2 3;sd 25
sfi -2 0 -4 0 -6;;1 2;sd 26
```

```
sfi -7;;6 7;sd  7
sfi -7;;5 6;sd  8
sfi -7;;4 5;sd  9
sfi -7;;3 4;sd 10
sfi -7;;2 3;sd 11
sfi -7;;1 2;sd 12
sfi ;;-7;sd 13
sfi ;;-1;sd 14
```









9. Speaker - woofer with ceramic magnet,2 spiders, voice coil form, cone, frame

```
para cd 0.05
     te [%cd*3.14159/180.0*0.05]
     ra 5.0
     rad [(%ra+1.0)*3.14159*7/54.0]
     ra1 4.2
     rad1 [(%ra-1.0)*3.14159*7/54.0]
     cv 0.02
     ef [atan2(%cv,-0.55)/360.0]
     ht 5
     cu 100
     cvv 10
     cv2 7
     off 0.2;
sd 1 sp 0 0 0 .75
sd 2 plan 0 0 0 0 0 1
sd 3 cy 0 0 0 0 0 1 .76
sd 4 cy 0 0 0 0 0 1 .74
sd 5 er [%ra*sqrt(3)/2] [%ra/2] 0 [sqrt(3)] 1 0 %rad 1
sd 6 er [(%ra1-.1)*sqrt(3)/2] [(%ra1-.1)/2] 0 [sqrt(3)] 1 0 %rad1
.8
cylinder 1 11 21 54 86 94;1 2 8 9;-1;
  .8 1.6 1.7 4. 6 6.5 0 [20/3] [160/3] 60 0
 de 4 2 0 5 3 0
 sfi -5;2 3;-1; sd 5
 sfi -4;2 3;-1; sd 6
 z=atan2(%cv,1.65-x)/360*(1-sign(1,x-2.2))+ &
   (2.6/(4.3*4.3)*(x-2.2)*(x-2.2)+%ef)*(1+sign(1,x-2.2))-2*%ef-.2
 lct 6 rz 0;rz 60;rz 120;rz 180;rz 240;rz 300; lrep 1 2 3 4 5 6;
dei 3 6;; -1;
res 1 1 1 1 4 1 j 1
res 6 1 1 6 4 1 j 1
bb 3 1 1 3 2 1 1;
bb 3 2 1 3 3 1 2;
bb 3 3 1 3 4 1 3;
endpart
cylinder -1 2 29 34;1 49;1 -6 7;.75 .775 1.5 2 0 360 -2 -.2 0
 z=z+.4*sin(5.5*x*480)*(1-sign(1,x-1.5))*atan(%cu*(1.5-x))/180* &
 atan(%cv2*(x-.75)*(x-.75))/180*(1-sign(1,.75-x))
mseq k 10 2
mseq k 4 -1
insprt 1 6 1 9
dei 2 4;; 1 2;
pb 1 1 2 4 2 2 z -1.180942e+00
```

```
pb 4 1 3 4 2 3 xz 1.700000e+00 -3.095492e-01
mseq i 0 90 -3
insprt 1 2 2 15
insprt 1 2 3 10
insprt 1 2 4 10
insprt 1 2 5 11
insprt 1 2 6 11
insprt 1 2 7 11
insprt 1 2 8 11
insprt 1 2 9 11
insprt 1 2 10 11
insprt 1 2 11 10
mseq i 0 -13 -9 -7 -8 -8 -8 -7 -7 -7 -6 -4 0
drs 4 1 1 5 2 4 i 1.5 1.5
drs 5 1 1 6 2 4 i 1.5 1.5
drs 6 1 1 7 2 4 i 1.5 1.5
drs 7 1 1 8 2 4 i 1.5 1.5
drs 8 1 1 9 2 4 i 1.5 1.5
drs 9 1 1 10 2 4 i 1.5 1.5
drs 10 1 1 11 2 4 i 1.5 1.5
drs 11 1 1 12 2 4 i 1.5 1.5
res 12 1 1 13 2 4 i 3
endpart
cylinder 1 11 21 54 86 94;1 2 8 9;-1;
  .8 1.6 1.7 4. 6 6.5 0 [20/3] [160/3] 60 0
 de 4 2 0 5 3 0
 sfi -5;2 3;-1; sd 5
 sfi -4;2 3;-1; sd 6
 z=atan2(%cv,1.65-x)/360*(1-sign(1,x-2.2))+ &
   (2.6/(4.3*4.3)*(x-2.2)*(x-2.2)+%ef)*(1+sign(1,x-2.2))-2*%ef-.2
 lct 6 rz 0;rz 60;rz 120;rz 180;rz 240;rz 300; lrep 1 2 3 4 5 6;
dei 1 3 0 4 6;; -1;
mseq j 2 12 2
trbb 3 1 1 3 2 1 1;
trbb 3 2 1 3 3 1 2;
trbb 3 3 1 3 4 1 3;
bb 4 1 1 4 2 1 4;
bb 4 3 1 4 4 1 5;
tfi 3 4; 1 4; -1;
relax 3 1 1 4 4 1 10 0 1
endpart
cylinder 1 11 21 54 86 94;1 2 8 9;-1;
  .8 1.6 1.7 4. 6 6.5 0 [20/3] [160/3] 60 0
 de 4 2 0 5 3 0
 sfi -5;2 3;-1; sd 5
```

```
 sfi -4;2 3;-1; sd 6
 z=atan2(%cv,1.65-x)/360*(1-sign(1,x-2.2))+ &
   (2.6/(4.3*4.3)*(x-2.2)*(x-2.2)+%ef)*(1+sign(1,x-2.2))-2*%ef-.2
 lct 6 rz 0;rz 60;rz 120;rz 180;rz 240;rz 300; lrep 1 2 3 4 5 6;
dei 1 4;; -1;
mseq j 8 48 8
trbb 4 1 1 4 2 1 4;
trbb 4 3 1 4 4 1 5;
endpart
cylinder 1 11 21;1 73;-1;6.5 7.2 7.45 0 60 4.95
 x=x-(z-4.7)/.3*%cd*180/(3.14159*(%cd*%cd+(x-7.2)*(x-7.2)))*%te
 z=z+.5*atan2(%cd,7.2-x)/180
 lct 6 rz 0;rz 60;rz 120;rz 180;rz 240;rz 300; lrep 1 2 3 4 5 6;
endpart
cylinder 1 77 87;1 217 433;-1;6 6.5 7 0 180 360 %ht
 z=z+.4*sin(3.5*x*720)*(1-sign(1,x-6.5))*atan(%cu*(6.5-x))/180* &
 atan(%cvv*(x-6))/90
bb 1 1 1 1 2 1 8;
bb 1 2 1 1 3 1 9;
insprt 1 2 1 7
insprt 1 2 2 10
insprt 1 2 3 11
insprt 1 2 4 10
insprt 1 2 5 11
insprt 1 2 6 10
insprt 1 2 7 11
mseq i -4 -5 -5 -5 -5 -5 -5 -2 -4
drs 2 1 1 3 3 1 i 1.15 1.15
drs 3 1 1 4 3 1 i 1.15 1.15
drs 3 1 1 4 3 1 i 1.15 1.15
drs 4 1 1 5 3 1 i 1.15 1.15
drs 5 1 1 6 3 1 i 1.15 1.15
drs 6 1 1 7 3 1 i 1.15 1.15
drs 7 1 1 8 3 1 i 1.15 1.15
res 8 1 1 9 3 1 i 1.15
res 1 1 1 2 3 1 i .65
pb 10 1 1 10 3 1 xz 6.9877639e+00 4.9868231e+00
endpart
cylinder 1 11;1 25 49;-1;.75 2.76 0 180 360 0
z=z+(x-.75)*%ht/5.25
bb 2 1 1 2 2 1 6;
bb 2 2 1 2 3 1 7;
endpart
cylinder 1 2 17 18;1 73 145;-1;2.76 2.95 5.81 6 0 180 360 0
dom 1 1 1 3 3 1
```

```
z=z+(x-.75)*%ht/5.25
pb 4 1 1 4 3 1 z %ht
trbb 1 1 1 1 2 1 6;
trbb 1 2 1 1 3 1 7;
trbb 4 1 1 4 2 1 8;
trbb 4 2 1 4 3 1 9;
endpart
block 1 9 21 29; 1 9 21 29; -1; -.5 -.5 .5 .5 -.5 -.5 .5 .5 .75
 dei 1 2 0 3 4;1 2 0 3 4;;
 sf 1 1 0 4 4 0 sd 1
 sfi -1 0 -4;;-1; sd 2
 sfi ;-1 0 -4;-1; sd 2
endpart
cylinder 1 2 23 26; 1 49; 1 4 6 9 10; 0.76 0.78 2.7 3.0 0 360
   [-2-%off] [-1.4-%off] [-1.1-%off] [-.6-%off] [-.5-%off]
 de 1 1 4 2 2 5
 de 3 1 3 4 2 5
 de 3 1 1 4 2 2
endpart
block 1 5 17 21;1 5 17 21;1 3 4 14;-.3 -.3 .3 .3 -.3 -.3 .3 .3
  [-2-%off] [-1.8-%off-.01] [-1.8-%off+.01] [-.6-%off]
 dei 1 2 0 3 4;1 2 0 3 4;;
 sfi -1 -4;-1 -4;1 2; sd 3
 sfi -1 -4;-1 -4;3 4; sd 4
endpart
merge
```

10. Piston and rod

```
parameter m 4 n 2 t 2 u 1;
curd 1 csp3 00 14.1825 -2.5 -4 15.24 -1.5 -4 15.24 1.5 -4 14.1825
2.5 -4 ;;;
curd 2 csp3 00 14.1825 -2.5  4 15.24 -1.5  4 15.24 1.5  4 14.1825
2.5  4 ;;;
sd 1 cy 0 0 0 1 0 0 20
sd 2 cy 0 0 0 1 0 0 18
sd 3 cy 21 0 0 0 0 1 8
sd 4 cy 2 0 0 0 1 1 9
sd 5 cy 2 0 0 0 -1 1 9
sd 6 cy 21 0 0 0 0 1 2.5
sd 7 cy 0 0 0 1 0 0 16
sd 8 cy -35 0 0 0 0 1 8.5
sd 9 cy 21 0 0 0 0 1 4
sd 10 rule3d 1 2 ;
sd 11 cy 14.5 10 0 0 0 1 7
sd 12 cy -22.5 -4 0 0 0 1 9.5
sd 13 cy 35.5 10.5 0 1 0 0 1.5
sd 14 cy 0 0 0 1 0 0 14
sd 15 cy 0 0 0 1 0 0 16
sd 18 cy -35 0 0 0 0 1 7.5
sd 19 plan -38.5 0 0 1 0 0
sd 21 plan 22.5 2.5 0 0 1 0
sd 23 cy 22.5 9.5 0 1 0 0 4
sd 25 cy 22.5 -10.5 0 1 0 0 1.5
sd 26 cy 14.5 -10 0 0 0 1 7
sd 27 cy -22.5 4 0 0 0 1 9.5
sd 28 cy 22.5 -9.5 0 1 0 0 4
sd 29 plan 22.5 -2.5 0 0 1 0
cylinder 1 [1+%u];
         1 [1+%u*90];
         1 [1+%u];
         18.1 20.4
         -210 149
         36.1 36.9
lct 1 ry 90;lrep 1;
endpart
cylinder 1 [1+%u];
         1 [1+%u*90];
         1 [1+%u];
         18.1 20.4
         -140 219
         34.1 34.9
```





```
lct 1 ry 90;lrep 1;
endpart
cylinder
  -1         [-(1+%u   )];
  -1         [-(1+%u   )] [-(1+%u*2 )] [-(1+%u*3 )] [-(1+%u*4 )]
[-(1+%u*5 )] [-(1+%u*6 )] [-(1+%u*7 )] [-(1+%u*8 )] [-(1+%u*9 )]
[-(1+%u*10)] [-(1+%u*11)] [-(1+%u*12)] [-(1+%u*13)] [-(1+%u*14)]
[-(1+%u*15)] [-(1+%u*16)] [-(1+%u*17)] [-(1+%u*18)] [-(1+%u*19)]
[-(1+%u*20)] [-(1+%u*21)] [-(1+%u*22)] [-(1+%u*23)] [-(1+%u*24)]
[-(1+%u*25)] [-(1+%u*26)] [-(1+%u*27)] [-(1+%u*28)] [-(1+%u*29)]
[-(1+%u*30)] [-(1+%u*31)] [-(1+%u*32)] [-(1+%u*33)] [-(1+%u*34)]
[-(1+%u*35)] [-(1+%u*36)] [-(1+%u*37)] [-(1+%u*38)] [-(1+%u*39)]
[-(1+%u*40)] [-(1+%u*41)] [-(1+%u*42)] [-(1+%u*43)] [-(1+%u*44)]
[-(1+%u*45)] [-(1+%u*46)] [-(1+%u*47)] [-(1+%u*48)] [-(1+%u*49)]
[-(1+%u*50)] [-(1+%u*51)] [-(1+%u*52)] [-(1+%u*53)] [-(1+%u*54)]
[-(1+%u*55)] [-(1+%u*56)] [-(1+%u*57)] [-(1+%u*58)] [-(1+%u*59)]
[-(1+%u*60)] [-(1+%u*61)] [-(1+%u*62)] [-(1+%u*63)] [-(1+%u*64)]
[-(1+%u*65)] [-(1+%u*66)] [-(1+%u*67)] [-(1+%u*68)] [-(1+%u*69)]
[-(1+%u*70)] [-(1+%u*71)] [-(1+%u*72)] [-(1+%u*73)] [-(1+%u*74)]
[-(1+%u*75)] [-(1+%u*76)] [-(1+%u*77)] [-(1+%u*78)] [-(1+%u*79)]
[-(1+%u*80)] [-(1+%u*81)] [-(1+%u*82)] [-(1+%u*83)] [-(1+%u*84)]
[-(1+%u*85)] [-(1+%u*86)] [-(1+%u*87)] [-(1+%u*88)] [-(1+%u*89)]
[-(1+%u*90)];
   1         [1  +%u    ];
18.1 20.4
   1   4   8  12  16  20  24  28  32  36  40  44  48  52  56
  60  64  68  72  76  80  84  88  92  96 100 104 108 112 116
 120 124 128 132 136 140 144 148 152 156 160 164 168 172 176
 180 184 188 192 196 200 204 208 212 216 220 224 228 232 236
 240 244 248 252 256 260 264 268 272 276 280 284 288 292 296
 300 304 308 312 316 320 324 328 332 336 340 344 348 352 356
 359
32.1 32.9
lct 1 ry 90;lrep 1;
endpart
block
      1 [1+3*%n] [1+5*%n] [1+16*%n] [1+17*%n];
      1 [1+%n] [1+2*%n] [1+3*%n] [1+5*%n] [1+5*%n+%m]
        [1+5*%n+3*%m] [1+5*%n+4*%m] [1+7*%n+4*%m]
        [1+8*%n+4*%m] [1+9*%n+4*%m] [1+10*%n+4*%m];
      1 [1+%n] [1+3*%n] [1+4*%n];
      -38.5 -27.5 -22.5 13.5 16
      -12.5 -11.5 -9.5 -8.5 -7.5 -5 5 7.5 8.5 9.5 11.5 12.5
      -4 -2 2 4
dei 1 2;4 9;;
dei 3 5;1 5 0 8 12;;
dei 2 4;6 7;1 2 0 3 4;
```

```
dei 1 3;1 2 0 11 12;1 2 0 3 4;
mb 2  9 0 3 12 0 x -3.5
mb 2  1 0 3  4 0 x -3.5
pb 5  6 0 5  6 0 xy 16 -.143
pb 5  7 0 5  7 0 xy 16  .143
pb 2  6 1 2  6 4 xy -27.5 -4.07223
pb 2  7 1 2  7 4 xy -27.5  4.07223
pb 5  5 0 5  5 0 xy 16.5 -3.733
pb 5  8 0 5  8 0 xy 16.5  3.733
pb 1  2 4 3  2 4 yz -11.9  3.2
pb 1 11 4 3 11 4 yz  11.9  3.2
pb 1  1 3 3  1 3 yz -11.9  3.2
pb 1 12 3 3 12 3 yz  11.9  3.2
pb 1  1 2 3  1 2 yz -11.9 -3.2
pb 1 12 2 3 12 2 yz  11.9 -3.2
pb 1  2 1 3  2 1 yz -11.9 -3.2
pb 1 11 1 3 11 1 yz  11.9 -3.2
pb 4  8 1 4  8 4 xy 14.2  2.5
pb 4  5 1 4  5 4 xy 14.2 -2.5
pb 5  6 1 5  6 4 xy 16.25 -1.875
pb 5  7 1 5  7 4 xy 16.25  1.875
pb 5  5 1 5  5 4 xy 16.5   -3.7
pb 5  8 1 5  8 4 xy 16.5    3.7
mbi -4;-7;;y -3.4
mbi -4;-6;;y  3.4
res 5  5 1 5  8 4 j 1
sfi 1 2;-4 0 -9;;sd 8
sfi -2;4 9;;sd 8
sfi -4;-5;;sd 26
sfi -4;-8;;sd 11
sfi -5;;;sd 9
sfi -2;-6;;sd 27
sfi -2;-7;;sd 12
sfi -1;;;sd 19
sfi -4;-6;;sd 29
sfi -4;-7;;sd 21
sfi -4;6 7;1 4;sd 10
tfi 4 5;5 8;-1 0 -4;
tf 4 5 1 5 8 4
insprt 1 1 1 [2*%n]
dei 1 4;2 3 0 10 11;2 3;
dei 1 2;4 9;1 4;
mb 1 1 1 1 12 4 x -4.5
sfi 1 4;-2 -3;-2 -3;sd 25
sfi 1 4;-10 -11;-2 -3;sd 13
```

```
sfi 1 4;-1 3;-1 -4;sd 28
sfi 1 4;10 -12;-1 -4;sd 23
bb 1 4 1 2 4 4 2;
bb 1 9 1 2 9 4 3;
endpart
block 1 [1+%n] [1+3*%n] [1+6*%n] [1+7*%n] [1+11*%n];
      1 [1+%n] [1+%n+%m] [1+%n+3*%m] [1+%n+4*%m] [1+2*%n+4*%m];
      1 [1+%n] [1+2*%n] [1+4*%n] [1+5*%n] [1+6*%n];
      -27.5 -29.25 -31 -37.25 -38.5 -46.5
      -8.5 -7.5 -5 5 7.5 8.5
      -6 -4 -2 2 4 6
dei 1 2 0 4 6;1 2 0 5 6;;
dei 2 4;2 5;;
dei 5 6;2 5;1 2 0 5 6;
pb 2 1 0 2 1 0 xy -27.5 -7.5
pb 2 6 0 2 6 0 xy -27.5  7.5
pb 4 1 0 4 1 0 x -38.5
pb 4 6 0 4 6 0 x -38.5
mb 1 1 1 2 2 6 xy -1 -1
mb 1 5 1 2 6 6 xy -1  1
pb 5 2 0 5 2 0 y -8.5
pb 5 5 0 5 5 0 y  8.5
pb 4 3 0 5 3 0 xy -40.5 -3.5
pb 4 4 0 5 4 0 xy -40.5  3.5
mb 6 2 0 6 2 0 y  1
mb 6 5 0 6 5 0 y -1
mb 4 2 1 4 2 6 xy -1  .5
mb 4 5 1 4 5 6 xy -1 -.5
pb 2 3 1 2 3 6 xy -28.44 -3.638
pb 2 4 1 2 4 6 xy -28.44  3.638
pb 3 2 1 3 2 6 xy -31.81 -6.79
pb 3 5 1 3 5 6 xy -31.81  6.79
pb 5 3 1 5 3 6 xy -42.73 -3.54
pb 5 4 1 5 4 6 xy -42.73  3.54
pb 4 3 1 4 3 6 xy -41.82 -3.13
pb 4 4 1 4 4 6 xy -41.82  3.13
pb 1 2 1 1 2 6 xy -28.99 -6.01
pb 1 5 1 1 5 6 xy -28.99  6.01
pb 2 1 1 2 1 6 xy -28.99 -6.01
pb 2 6 1 2 6 6 xy -28.99  6.01
pb 2 2 1 2 2 6 xy -29.73 -5.34
pb 2 5 1 2 5 6 xy -29.73  5.34
pb 1 4 1 1 4 6 xy -27.53  4.06
pb 1 3 1 1 3 6 xy -27.53 -4.06
pb 3 6 1 3 6 6 xy -31.38  7.69
```

```
pb 3 1 1 3 1 6 xy -31.38 -7.69
pb 4 1 1 4 1 6 xy -38.5  -7.75
pb 5 2 1 5 2 6 xy -38.5  -7.75
pb 4 6 1 4 6 6 xy -38.5   7.75
pb 5 5 1 5 5 6 xy -38.5   7.75
res 6 2 2 6 5 5 j 1
sfi -1 -5;-1 -6;;sd 8
sfi -2 -4;-2 -5;;sd 18
sfi -4;-1 0 -6;;sd 19
sfi -5;-2 0 -5;;sd 19
sfi -1;-3;;sd 27
sfi -1;-4;;sd 12
insprt 1 2 5 [2*%n]
tfi 5 7;2 5;-2 0 -5;
tfi 5 7;2 5;2 5;
bb 5 2 2 6 2 5 2;
bb 5 5 2 6 5 5 3;
endpart
block -1;
      1 [1+2*%m];
      1 [1+%n] [1+2*%n] [1+4*%n] [1+5*%n] [1+6*%n];
      16.3296
      -1.59 1.59
      -6 -4 -2 2 4 6
sfi -1;;; sd 9
bb 1 1 1 1 2 6 1;
mate 0
endpart
block 1 [1+%m] [1+2*%m] [1+5*%m] [1+6*%m];
      1 [1+%m] [1+4*%m] [1+5*%m];
      1 [1+%n] [1+2*%n] [1+4*%n] [1+5*%n] [1+6*%n];
      25 27.5 27.5 17.5 17.5
      -2.5 -2.5 2.5 2.5
      -6 -4 -2 2 3 6
dei 3 4;2 3;;
dei 1 3 0 4 5;1 2 0 3 4;;
pb 4 4 1 4 4 6 xy 17.264105  2.918027
pb 5 3 1 5 3 6 xy 17.264105  2.918027
pb 4 1 1 4 1 6 xy 17.264105 -2.918027
pb 5 2 1 5 2 6 xy 17.264105 -2.918027
sfi -2 -5;-1 -4;;sd 9
sfi -3 -4;-2 -3;;sd 6
insprt 1 4 2 %m
insprt 1 3 4 %m
trbb 5 3 1 5 4 6 1;
```

```
endpart
block
1 [1+%m] [1+2*%m] [1+5*%m] [1+6*%m] [1+6*%m+%n]
  [1+6*%m+2*%n] [1+6*%m+3*%n] [1+6*%m+4*%n]
  [1+6*%m+5*%n] [1+6*%m+6*%n] [1+7.5*%m+6*%n];
1 [1+%t] [1+2*%t] [1+3*%t+%m] [1+3*%t+4*%m]
  [1+4*%t+5*%m] [1+5*%t+5*%m] [1+6*%t+5*%m];
1 [1+%t] [1+2*%t] [1+3*%t+%m] [1+3*%t+4*%m]
  [1+4*%t+5*%m] [1+5*%t+5*%m] [1+6*%t+5*%m];
2 11 15 27 31 32 33 34 35 36 37 43
-20 -20 -20 -6.5 6.5 20 20 20
-20 -20 -20 -6.5 6.5 20 20 20
de 1 2 0 2 7 0
de 5 2 0 6 3 0
de 5 6 0 6 7 0
de 5 3 3 10 6 6
de 3 4 0 4 5 0
de 1 2 2 5 7 7
dei ;1 3 0 6 8;1 3 0 6 8;
dei 5 7 0 8 9 0 10 11;;1 2 0 7 8;
dei 1 2;;1 4 0 5 8;
dei 5 7 0 8 9 0 10 11;1 2 0 7 8;;
pb 3 4 0 3 4 0 x 13
pb 3 5 0 3 5 0 x 13
pb 4 4 0 4 4 0 x 29
pb 4 5 0 4 5 0 x 29
pb 3 4 0 4 4 0 2 -8
pb 3 5 0 4 5 0 2 8
mb 12 3 3 12 6 6 1 2
sfi ;-3 -6;-3 -6;sd 14
sfi -12;-2 0 -7;2 7;sd 15
sfi -12;2 7;-2 0 -7;sd 15
sfi -3 -4;-4 -5;;sd 3
sfi 1 12;-1 -8;-1 -8;sd 1
sfi 1 11;-2 -7;-2 -7;sd 2
sfi 1 -2;7 8;-5 6;sd 4
sfi 1 -2;-5 6;7 8;sd 4
sfi 1 -2;7 8;3 -4;sd 5
sfi 1 -2;3 -4;7 8;sd 5
sfi 1 -2;1 2;-5 6;sd 5
sfi 1 -2;-5 6;1 2;sd 5
sfi 1 -2;1 2;3 -4;sd 4
sfi 1 -2;3 -4;1 2;sd 4
relaxi 1 5;-1 -2 -7 -8;-1 -2 -7 -8;10 0.001 1
endpart
```

```
block
1 [1+2*%m] [1+5*%m] [1+7*%m];
1 [1+2*%m] [1+5*%m] [1+7*%m];
1 [1+%t] [1+3*%t] 0 [3+3*%t] [3+5*%t] [3+6*%t];
13 13 29 29
-8 -8 8 8
-20 -17 -14 0 14 17 20
de 2 2 0 3 3 0
dei 1 2 0 3 4;1 2 0 3 4;;
pb 4 3 2 4 3 2 xyz 2.709948e+01 5.176531e+00 -1.723960e+01
pb 3 4 2 3 4 2 xyz 2.709948e+01 5.176531e+00 -1.723960e+01
pb 1 3 2 1 3 2 xyz 1.490052e+01 5.176526e+00 -1.723960e+01
pb 2 4 2 2 4 2 xyz 1.490052e+01 5.176526e+00 -1.723960e+01
pb 2 1 2 2 1 2 xyz 1.490052e+01 -5.176526e+00 -1.723960e+01
pb 1 2 2 1 2 2 xyz 1.490052e+01 -5.176526e+00 -1.723960e+01
pb 3 1 2 3 1 2 xyz 2.709948e+01 -5.176531e+00 -1.723960e+01
pb 4 2 2 4 2 2 xyz 2.709948e+01 -5.176531e+00 -1.723960e+01
pb 2 1 6 2 1 6 xyz 1.490052e+01 -5.176526e+00 1.723960e+01
pb 1 2 6 1 2 6 xyz 1.490052e+01 -5.176526e+00 1.723960e+01
pb 1 3 6 1 3 6 xyz 1.490052e+01 5.176526e+00 1.723960e+01
pb 2 4 6 2 4 6 xyz 1.490052e+01 5.176526e+00 1.723960e+01
pb 3 1 6 3 1 6 xyz 2.709948e+01 -5.176531e+00 1.723960e+01
pb 4 2 6 4 2 6 xyz 2.709948e+01 -5.176531e+00 1.723960e+01
pb 4 3 6 4 3 6 xyz 2.709948e+01 5.176531e+00 1.723960e+01
pb 3 4 6 3 4 6 xyz 2.709948e+01 5.176531e+00 1.723960e+01
pb 2 2 1 2 3 7 x [21-1.1]
pb 3 2 1 3 3 7 x [21+1.1]
pb 2 2 1 3 2 7 y -1.1
pb 2 3 1 3 3 7 y  1.1
sfi -2 -3;-2 -3;;sd 6
sfi -1 -4 ; -1 -4 ; ; sd 3
sfi ;;-1 -7;sd 1
sfi ;;-2 -6;sd 2
endpart
block 1 [1+%m] [1+4*%m] [1+5*%m];
      1 [1+%m] [1+4*%m] [1+5*%m];
      1 [1+%t] [1+3*%t] [1+3*%t+2*%m] [1+3*%t+2*%m+%n]
[1+3*%t+2*%m+2*%n]
      [1+3*%t+2*%m+4*%n] [1+3*%t+2*%m+5*%n] [1+3*%t+2*%m+6*%n]
      [1+3*%t+4*%m+6*%n] [1+5*%t+4*%m+6*%n] [1+6*%t+4*%m+6*%n];
      [21-1.1] [21-1.1] [21+1.1] [21+1.1]
      -1.1 -1.1 1.1 1.1
      -20 -17 -14 -6 -4 -2 2 4 6 14 17 20
dei 1 2 0 3 4; 1 2 0 3 4;;
sfi -1 -4; -1 -4;;sd 6
sfi ;; -1 0 -12;sd 1
```

```
sfi ;; -2 0 -11;sd 2
endpart
merge
```

VIII. Brackets

1. L-shaped bracket



```
sd 1 plan 2 9 0 1 1 0
sd 2 plan 2 0 0 1 0 0
sd 3 plan 0 2 0 0 1 0
sd 4 cy 4 4 0 0 0 1 2
sd 5 cy 12.75 0 0 0 1 0 .75
sd 6 plan 0 0 0 1 0 0
sd 7 plan 0 0 0 0 1 0
sd 8 cy 3 3 0 0 0 1 3
sd 9 plan 0 0 -.44 0 0 1
sd 10 plan 0 0 .44 0 0 1
sd 11 plan 4 0 0 1 0 0
sd 12 plan 0 4 0 0 1 0
sd 13 plan 0 10 0 0 1 0
sd 14 plan 14.5 0 0 1 0 0
sd 15 plan 0 0 -3 0 0 1
sd 16 plan 0 0 3 0 0 1
block 1 3 5 8 10;1 3 5 8 10;1 3;
  2 3.4 4.95 7.2 9 2 3.4 4.95 7.2 9 -.44 .44
dei 2 5; 4 5;;
dei 3 5; 3 4;;
dei 4 5; 2 3;;
dei 1 2; 1 2;;
pb 1 2 1 1 2 2 xy 2 2
pb 2 1 1 2 1 2 xy 2 2
pb 2 4 1 3 5 2 xy 2.9 6.7
pb 4 2 1 5 3 2 xy 6.7 2.9
res 3 1 1 5 1 2 i 1
res 1 3 1 1 5 2 j 1
sfi 1 -5; 1 -5;;sd 1
sfi -3 4; -3 4;;sd 1
sfi -1; 3 5;;sd 2
sfi 3 5; -1;;sd 3
sfi -1 3; -1 3;;sd 4
sfi ;;-1;sd 9
sfi ;;-2;sd 10
sfi -3;-1;;sd 11
sfi -1;-3;;sd 12
endpart
block 1 3 5 10 13 15 17;1 3 5 10 11;1 5 6 8 9 13;
  0 2 4 9 12.4 13.05 14.5
```

```
  0 2 4 9 10
  -3 -.67 -.44 .44 .67 3
dei 5 6;; 2 5;
dei 2 7; 2 5;;
dei 1 2; 1 2;;
pb 1 1 1 2 2 6 xy 0 0
sfi 2 -4; 2 -4;;sd 1
sfi -2; 3 5;;sd 2
sfi 3 7; -2;;sd 3
sfi -2 3; -2 3;;sd 4
sfi -5 -6;; -2 -5;sd 5
sfi -1; 3 5;;sd 6
sfi 3 7; -1;;sd 7
sfi -1 3; -1 3;;sd 8
sfi ;; -3;sd 9
sfi ;; -4;sd 10
sfi -3;-2;;sd 11
sfi -2;-3;;sd 12
sfi ;-5;;sd 13
sfi -7;;;sd 14
sfi ;;-1;sd 15
sfi ;;-6;sd 16
endpart
merge
```

2. Bracket with hole



```
sd 1 cy 0 3 0 0 0 1 1
sd 2 cy 0 3 0 0 0 1 1.5
sd 3 plan -1.5 0 0 1 0 0
sd 4 plan 1.5 0 0 1 0 0
block 1 4 9 12;1 5 8 13;1 4;
     -1.5 -1 1 1.5 0 2 3 3 0 2
de 2 2 0 3 4 0
pb 0 4 0 0 4 0 xy 0 5
sfi -2 -3;-2 4; ;sd 1
sfi -1 -4;3 4; ;sd 2
sf 1 1 0 1 3 0 sd 3
sf 4 1 0 4 3 0 sd 4
res 1 1 0 1 3 0 j 1
res 4 1 0 4 3 0 j 1
endpart
merge
```

3. Feed guide

```
sd 1 cy 0. 0. 3.8 1 0 0 1.
sd 2 cy 0. 0. 3.8 1 0 0 1.6
sd 3 cy 0. 0. 5.8 0 1 0 .25
sd 4 cy 0. 0. 5.8 0 1 0 .68
sd 5 cy 0. 0. 3.8 0 1 0 .125
block 1 4 6 8 11 14 17 19; 1 3 5 7 9;1 6 12 14 18 20 22 26 28 31
33 35 37;-2.5 -1.9 -1.7 -1.5 -1. -.6 -.2 0.;-1.5 -1. -.7 -.3 0.;
0. .7 2.2 3.15 3.65 3.8 3.95 4.45 5.4 5.6 5.8 6. 6.2
dei 1 5; ; 2 13; ;
dei 5 6; ; 2  3 0 9 13;
dei 7 8; ; 5 7 0 10 12;
dei ; 1 2; 2 13;
dei ; 3 5; 4 8;
dei ; 4 5; 7 13;
dei 2 4; 3 5; 1 2;
sfi ;-3 5; -4 -8; sd 1
sfi ;-2 5; -3 -9; sd 2
sfi ; -2; 9 13; plane 0 -1.1 0 0 1 0
sfi ; -4; 8 13; plane 0 -.08 0 0 1 0
sfi ; -2; 2 3;  plane 0 -1. 0 0 1 0
sfi ; -3; 8 13; plane 0 -.6 0 0 1 0
sfi ; -3; 3 4 ; plane 0 -.6 0 0 1 0
sfi ; -4; 3 4 ; plane 0 -.3 0 0 1 0
sfi -7 8; ; -10 -12; sd 3
sfi -6 8 ; ; 11 -13; sd 4
sfi -7 8 ; ; -5  -7; sd 5
sfi -6   ; ; 8 11; plane -.68 0 0 1 0 0
sfi -2 -4; -3 4; ; cy -1.7 -.187 0 0 0 1
.125
sfi -2 ; 4 5; ; plane -1.825 0 0 1 0 0
sfi -4 ; 4 5; ; plane -1.575 0 0 1 0 0
relaxi 6 8; -2; 9 13; 50 .001 1
relaxi 6 8; -4; 9 13; 50 .001 1
relaxi -5; 2 5; 3  9; 50 .001 1
relaxi -8; 2 5; 3  9; 50 .001 1
relaxi 5 8; -2; 3  9; 50 .001 1
relaxi 5 8; -3; 4  8; 50 .001 1
relaxi 1 6; 1 5; -1; 50 .001 1
relaxi 1 6; 1 5; -2; 50 .001 1
lct 3 ryz; rxz; ryz rxz;
lrep 0 1 2 3 ;
endpart
merge
```



4. Fork

```
lcd 1 0 0 1 [4000./49] 10 1;
lcd 2 0 0 1 [220./96] 10 1;
nike3d
sid 1 sl fric .3;
partmode i
block 4 2 6 2 4; 4 2 3 3 2 4; 6 2 6 2 3 3 2 4 2;
      0. .222 .347 1.153 1.278 1.5
      0 .4 .575 .875 1.175 1.35 1.75
     -3 -1 -.875 -.475 -.3 0 .3 .475 .875 1.
dei 2 5; ; 2 10;
dei ; 3 5; 5 7;
sfi ; -3 -5; -5 -7; cy 0 .875 0 1 0 0 .375
sfi -2 3; ; -2 3;  cy .347 0 -.875 0 1 0 .125
sfi 4 -5; ; -2 3; cy 1.153 0 -.875 0 1 0 .125
sfi -2; 2 6; 4 8; plane .311 0 0 1 0 0
sfi -5; 2 6; 4 8; plane 1.189 0 0 1 0 0
pb 2 0 10 2 7 10 x .188
pb 5 0 10 5 7 10 x 1.312
res 1 1 1 6 7 2 k .7
relaxi -1; ; ; 20 .001 1
relaxi -6; ; ; 20 .001 1
relaxi ; -1; 1 3; 20 .001 1
relaxi ; -7; 1 3; 20 .001 1
fci -5; 2 6; 4 8; 2 1. 1 0 0
bi ; ; -1; dx 1 dy 1 dz 1;
sii 1 2 0 5 6; -3 -5; -5 -7; 1 m
mate 1
endpart
block 4 6 4; 6;6;
      0 .311 1.189 1.5
      .575 1.175 -.3 .3
sfi ; -1 -2; -1 -2; cy 0 .875 0 1 0 0 .372
sii 1 2 0 3 4; -1 -2; -1 -2; 1 s
fci 2 3;; -1; 1 1. 0 0 1
mate 2
endpart
merge
```



5. Yoke



```
sd 1 cy -9. 0 0 0 1 0 4.125
sd 2 cy -9. 0 0 0 1 0 5.44
sd 3 cy  0.  0 0 0 0 1 3.44
sd 4 cy  0.  0 0 0 0 1 4.625
sd 5 cy  0.  0 0 0 0 1 6
sd 6 cy  0.  0 0 0 0 1 7.25
sd 7  plane 0 0 5     0 0 1
sd 8  plane 0 0 6.38  0 0 1
sd 9  plane 0 0 10    0 0 1
sd 10 plane 0 1.625 0 0 1 0
sd 11 plane 0 3.75  0 0 1 0
sd 12 cy -9 0 0 0 1 0 7.5
sd 13 cy -9 0 0 0 1 0 6.5
sd 14 plane -9 0 0 1 0 1
sd 15 cy -9 0 0 0 1 0 5.45
plane 1 0 0 0 1 0 0 .01 symm;
plane 2 0 0 0 0 1 0 .01 symm;
plane 3 0 0 0 0 0 1 .01 symm;
gct 3 rxy;rxz;ryz;
lev 1 grep 0 1;;
lev 2 grep 0 2;;
lev 3 grep 0 3;;
pslv 1
pslv 2
pslv 3
partmode i
cylinder
2 2 2 8 2 2 2; 12 3 3 ; 6 2 2 2;
3.44 4.625 6. 7.25 11.25 12.5 15.5 16.5
90 150 165 180
0 4. 5. 6.38 10.
dei 5 8; ; 2 5;
dei 1 3; ; 4 5;
dei 4 5; ; 1 2;
dei 4 8; 1 2; ;
dei 6 8; 1 3; ;
dei 2 8; 1 2; 1 3;
dei 4 5; 1 3; 3 5;
sfi -4 -5; 2 4 ; 1 -2; sd 1
sfi 3 -6; 2 4 ; 1 -3;  sd 2
sfi -2; 1 2; 1 3;  sd 4
sfi 2 8; -2; 1 3;  sd 11
sfi 4 8; -3; 3 5;  sd 10
```

```
sfi 6 8; -3; 1 2;  sd 10
sfi -5; ; 2 5;     sd 14
sfi 5 8; ; -2;     sd 14
sfi -8; ; 1 2;     sd 12
sfi -7; ; 1 2;     sd 13
sfi 1 4; ; -5;     sd 9
sfi ; ; -1; plane 0 0 0 0 0 1
sfi -3; 2 4; 1 2; sd 15
pb 5 3 5 5 3 5 xyz 14.4316 173.833 5.29954
pb 5 3 5 5 3 5 xyz 14.3954 173.528 5.30288
pb 5 4 5 5 4 5 xyz 14.3043 180.000 5.30596
pb 5 4 4 5 4 4 xyz 13.5972 180.000 4.59614
pb 5 3 4 5 3 4 xyz -13.5934 1.62487 4.59121
mbi -3;; -1; xz -0.569952 -0.197193e-01
mbi -3;; -1; xz -0.511467 0.564490e-03
mbi -2; 3 4; -1; xz -0.311569 0.222204e-02
mbi -2; 3 4; -1; xz -0.233189 0.180743e-02
mbi -3;; -1; xz -0.501566 0.317976e-01
relaxi 1 4;-4;1 2;10 .001 1.
relaxi 2 4;1 4;-1;10 .001 1.
relaxi 2 4;2 4;-1;10 .001 1.
relaxi 4 5;-4;3 5;10 .001 1.
relaxi 4 5;-3;3 5;10 .001 1.
relaxi 2 4;-2;1 3;10 .001 1.
bi 1 2; ;-4;dz 1 ;
relax 2 2 1 4 2 3 50 0 1
endpart
pplv
pplv
pplv
merge
```

6. Socket bearing







```
ld 1
 lp 0 0 1.5 0;
 lad 1.5 .5 30.
 ltas 1.5 .5 1 0 3. .5
 lar 0 3.5 .5
ld 2
 lp 0 0 0 .4;
sd 1 swept 2 r 1 0 1 1. ;                          ;sd 2 cy -1.5
.5 0 0 0 1 .2
sd 3 cy 0 3. 0 0 0 1 .2
sd 4 plane 0 0 0 1 0 0
sd 5 plane 0 2.85 0 0 1 0
sd 6 plane 0 3 0 0 1 0
sd 7 plane 0 3.15 0 0 1 0
sd 8 plane 0 3.5 0 0 1 0
sd 9 plane 0 1 0 0 1 0
sd 10 plane 0 1.8 0 0 1 0
sd 11 cy 1.5 .5 0 0 0 1 .2
sd 12 plane 2.1 .5 0 1 .65 0
block
1 4 6 8 14 17 20 23 26 32 34 36 39;
1 4 6 8 12 20 30 32 34 38;
1 4;
-2. -1.7 -1.5 -1.3 -.6 -.3 0 .3 .6 1.3 1.5 1.7 2.
0. .3 .5 .7 1. 1.8 2.9 3. 3.1 3.5
0. .4
dei 1 7 ; ; ;
dei 2 4 0 10 12; 2 4; ;
dei 6 8; 7 9; ;
pb 13 10 2 13 10 2 xyz 0.3295983
3.3759851 0.4000000
pb 13 10 1 13 10 1 xyz 0.3295983
3.3759851 0.0
sfi -2 -4; -2 -4; ;  sd 2
sfi -10 -12; -2 -4; ; sd 11
sfi -6 -8; -7 -9; ; sd 3
sfi 7 -13; -1 -10; ; sd 1
sfi -7; ; ; sd 4
sfi 8 13; -7 ; ; sd 5
sfi 8 13; -8 ; ; sd 6
sfi 8 13; -9 ; ; sd 7
sfi  7 13; -5;  ; sd 9
sfi  7 13; -6;  ;  sd 10
```

```
lini 9 13; -5 0 -6 ; -1;
lini 9 13; -5 0 -6 ; -2;
lini 7 13; -10; -1;
lini 7 13; -10; -2;
lini 8 13; -7 0 -8 0 -9 ; -1;
lini 8 13; -7 0 -8 0 -9 ; -2;
lct 1 ryz;
lrep 0 1;
mate 1
endpart
block
1 3 9 11; 1 9 12 14 16 20 28; 1 3 9 11;
-.5 -.5 .5 .5
-1. 0 .3 .5 .7 1. 1.8
.7 .7 1.7 1.7
dei 1 2 0 3 4; ; 1 2 0 3 4;
dei 2 3; 1 6 ; 2 3;
sfi -1 -4; ; -1 -4; cy 0 0 1.2 0 1 0 .6
sfi -2 -3; ; -2 -3; cy 0 0 1.2 0 1 0 .3
mate 2
endpart
block
1 7 13; 1 4 6 8 12 20;1 4;
-.6 0 .6
0 .3 .5 .7 1. 1.8
.4 1.2
pb 2 1 2 2 6 2 z .6
sfi ; ; -2; cy 0 0 1.2 0 1 0 .6
mate 3
endpart
merge
stp .005
```







7. Post

```
sd 1 cy 0 22.8 75.6 1 0 0 12.605
sd 2 plan 40 57 0 [cos(65)] [sin(65)] 0
sd 3 plan [40+5*cos(65)] [57+5*sin(65)] 0 [cos(65)] [sin(65)] 0
sd 4 plan 40 57 0 1 0 0
sd 5 plan 45 57 0 1 0 0
sd 6 plan 40 -11 0 [-sin(15)] [cos(15)] 0
sd 7 plan [40+5*sin(15)] [-11-5*cos(15)] 0 [-sin(15)] [cos(15)] 0
sd 8 plan [40+185*cos(155)] [57+185*sin(155)] 0
          [-sin(65)] [cos(65)] 0
sd 9 plan [40+78*cos(195)] [-11+78*sin(195)] 0
          [cos(15)] [sin(15)] 0
sd 10 plan [40+73*cos(195)] [-11+73*sin(195)] 0
           [cos(15)] [sin(15)] 0
sd 11 plan 0 0 0 0 0 1
sd 12 plan 0 0 150 0 0 1
sd 13 plan -1.0750001e+01 -2.4598021e+01 1.3813895e+02 0 0 1
sd 14 plan -30 0 0 1 0 0
sd 15 plan [40+40*sin(15)] [-11-40*cos(15)] 0
           [-sin(15)] [cos(15)] 0
sd 16 plan -10.75 0 0 1 0 0
sd 17 pl3 rt -1.0750000e+01  8.0665047e+01 1.0904040e+00
          rt  3.9999626e+01  2.8197687e+01 6.4208916e+01
          rt -1.0750050e+01 -2.4598644e+01 1.3013361e+02
sd 18 pl3 rt -1.0750000e+01  8.0665047e+01 9.0977402e+00
          rt -1.0750058e+01 -2.4598648e+01 1.3814090e+02
          rt  3.9999626e+01  1.6653345e+01 8.6604500e+01
sd 19 pl3 rt -1.0750005e+01  8.0664062e+01 1.2287770e+02
          rt  3.9999626e+01  3.1094267e+01 8.5091270e+01
          rt -1.0750050e+01 -2.4598644e+01 4.0405960e+01
sd 20 pl3 rt -1.0750005e+01  8.0664062e+01 1.1647630e+02
          rt  3.9999626e+01  1.4902173e+01 6.5775673e+01
          rt -1.0750050e+01 -2.4598644e+01 3.4004589e+01
sd 21 plan -1.0749535e+01  8.0664093e+01 1.2287704e+02 0 0 1
sd 22 plan -1.0749534e+01  8.0664093e+01 1.1647564e+02 0 0 1
sd 23 plan -1.0749865e+01  8.0664558e+01 9.0978327e+00 0 0 1
sd 24 plan -1.0749864e+01  8.0664558e+01 1.0904968e+00 0 0 1
sd 25 plan -1.0749969e+01 -2.4598225e+01 3.4004692e+01 0 0 1

block 1 3 9 12 14 26;
 1 6 8 14 16 25 27 33 35;
 1 3 9 15 17 19 28 30 32 38 42;
 44.2 40.5 -10.7 -30.2 -35.4 -129.8
 -72.5 -35 -11.5 11.4 13.6 31.9 32.2 56.8 60.1
```

```
 0 2.8 28.2 65.3 65.7 67.9 82.1 84.2 86.2 105.5 150

dei ;3 5 0 6 8;2 5 0 8 10; dei ;1 3 0 8 9;4 5 0 8 9;
dei 5 6;1 8;1 10; dei 2 5;3 4 0 7 8;6 7;
dei 2 5;5 6;2 4 0 9 10; dei 1 4;1 2;1 10;
dei 4 5;4 7;4 9; dei 3 5;3 4 0 7 8;2 10;
dei ;1 3 0 8 9;3 4 0 9 10; dei 1 3;8 9;2 3;
dei 2 6;3 8;1 2; dei 1 2;1 4 0 7 9;1 2;
dei 3 6;8 9;2 3; dei 1 4 0 5 6;1 2; 10 11;
dei 5 6;2 3;10 11; dei ;3 5 0 6 8;10 11;
dei 2 6;5 6;10 11; dei 2 3;8 9;1 2;

pb 1 5 5 6 5  5 yz 1.7258886e+01 6.6487625e+01
pb 1 5 6 6 5  6 yz 1.3504900e+01 7.0410500e+01
pb 1 5 7 6 5  7 yz 1.4909400e+01 8.2136810e+01
pb 1 5 8 6 5  8 yz 1.8829344e+01 8.5267342e+01
pb 1 6 8 6 6  8 yz 3.0046961e+01 8.2821831e+01
pb 1 6 7 6 6  7 yz 3.2342602e+01 7.9409500e+01
pb 1 6 6 6 6  6 yz 3.0984800e+01 6.9192497e+01
pb 1 6 5 6 6  5 yz 2.7817575e+01 6.6332207e+01
pb 1 2 3 2 2  5 z  4.548142e+01 pb 1 2 6 2 2 6 z 5.326471e+01
pb 3 1 6 5 3  6 z  4.040658e+01 pb 3 1 7 5 3 7 z 1.301319e+02
pb 1 9 8 2 9 10 z  1.036635e+02 pb 1 9 7 2 9 7 z 9.701180e+01
pb 1 9 5 1 9  5 z  3.164070e+01 pb 2 9 5 2 9 5 z 2.890014e+01
pb 1 9 6 2 9  6 z  3.714419e+01 pb 1 2 8 2 2 10 z 1.205030e+02
pb 1 2 7 2 2  7 z  1.122598e+02

res 3 8 1 6 9 11 i 1

sfi 3 4; -4 -7; -4 -9;sd 1 sfi 1 3; -4 -7; -4 5 0 6 7 0 8 -9;sd 1
sfi 3 4; -4 -7; -4 -9;sd 1 sfi 1 3; -4 -7; -4 5 0 6 7 0 8 -9;sd 1
sfi ;-9;;sd 3 sfi ;-8;;sd 2 sfi -2;;;sd 4 sfi -1;;;sd 5
sfi ;-3;;sd 6 sfi ;-2;;sd 7 sfi -6;8 9;;sd 8 sfi -4;1 3;;sd 10
sfi -5; 1 3;;sd 9 sfi ;; -1;sd 11 sfi ;; -11;sd 12
sfi -4; 4 7; 4 9;sd 14 sfi ; -1;;sd 15 sfi -3;;;sd 16
sfi 1 4;-6;3 5;sd 17 sfi 1 3;6 8;-5;sd 17 sfi 3 4; 6 7; -5;sd 17
sfi 1 2;-6;1 3;sd 2 sfi 1 3;3 4;-7;sd 17 sfi 3 4;-4;-7;sd 17
sfi 1 3;3 5;-8;sd 18 sfi 3 4;4 5;-8;sd 18 sfi 1 4; -5; 8 10;sd 18
sfi 1 2;-5;10 11;sd 6 sfi 1 2;-3;-10;sd 18 sfi 1 3;7 8;-6;sd 18
sfi 3 4;-7;-6;sd 18 sfi 1 3;6 8;-8;sd 19 sfi 3 4;6 7;-8;sd 19
sfi 1 4;-6;8 10;sd 19 sfi 1 3;3 4;-6;sd 19 sfi 1 2;-6;10 11;sd 2
sfi 1 2;-8;-10;sd 19 sfi 1 4;-5;3 5;sd 20 sfi 1 3;3 5;-5;
sd 20 sfi 3 4; 4 5; -5;sd 20 sfi 1 3; 7 8; -7;sd 20
sfi 3 4;-7;-7;sd 20 sfi 1 2; -5;1 3;sd 6 sfi 1 2;-3;-3;sd 20
sfi 3 6; 8 9; -8 0 -10;sd 21 sfi 3 6; 8 9; -7;sd 22
```

```
sfi 3 6; 8 9; -6;sd 23 sfi 3 6; 8 9; -2 0 -5;sd 24
sfi 2 3; -3; -3;sd 20 sfi 3 5; 1 3; -3 0 -5;sd 25
sfi 3 5; 1 3; -8 0 -10;sd 13 sfi 1 2; -4; 1 2;sd 7
sfi 1 2; -7; 1 2;sd 3

relaxi -1 0 -2 0 -3 0 -4; 5 6; 5 8;10 0 1

endpart

block 1 3 6 9; 1 3 6 9; 1 3; 0 0 0 0 0 0 0 0 0 0

dei 1 2; 1 2;; dei 3 4; 3 4;;

lini -1; 2 4; -1 0 -2;lini -2; 2 4; -1 0 -2;
lin 2 1 1 4 1 1 lin 2 2 1 4 2 1

pb 3 4 2 3 4 2 xyz 1.462513e+01 6.883212e+01 1.499603e+01
pb 4 3 2 4 3 2 xyz 1.462513e+01 6.883212e+01 1.499603e+01
pb 2 2 2 2 2 2 xyz 4.000000e+01 5.699997e+01 2.817700e+00
pb 2 1 2 2 1 2 xyz 4.500000e+01 5.466843e+01 2.817700e+00
pb 2 1 1 2 1 1 xyz 4.500000e+01 6.018532e+01 2.817700e+00
pb 2 2 1 2 2 1 xyz 4.000000e+01 6.251685e+01 2.817700e+00
pb 1 2 1 1 2 1 xyz 4.000013e+01 6.251621e+01 0.000000e+00
pb 4 2 1 4 2 1 xyz 4.000002e+01 6.251682e+01 2.890014e+01
pb 4 1 1 4 1 1 xyz 4.499963e+01 6.018531e+01 3.164070e+01
pb 4 1 2 4 1 2 xyz 4.500000e+01 5.466825e+01 3.164070e+01
pb 3 1 2 3 1 2 xyz 4.500004e+01 5.466827e+01 1.722881e+01
pb 3 2 2 3 2 2 xyz 4.000005e+01 5.699978e+01 1.585892e+01
pb 2 3 1 2 3 2 yz 6.883246e+01 1.954134e+00
pb 3 3 1 3 3 2 yz 6.641538e+01 8.659911e+00
pb 4 3 1 4 3 1 xyz 1.462508e+01 7.434924e+01 1.499561e+01
pb 3 4 1 3 4 1 xyz 1.462508e+01 7.434924e+01 1.499561e+01
pb 3 3 1 3 3 2 xz 1.943839e+01 8.659911e+00

sfi ;; -2;sd 2 sfi ;; -1;sd 3 sfi 1 4; -2; 1 2;sd 4
sfi 2 4;-1;1 2;sd 5 sfi -1;2 4;1 2;sd 11 sfi 1 2;-4;1 2;sd 16
sfi 2 3;-4;-2;sd 17 sfi -4;2 3;-2;sd 17 sfi -2;-4;1 2;sd 24

endpart

merge

stp 0.001
```

IX. Springs and Screws

1. Spring

```
cylinder 1 3 6 8 ; 1 79 80 81 82 249 250 251 252 331 ; 1 3 6 8 ;
 1.92 1.92 2.08 2.08 0 312 316 320 324 992 996 1000 1004 1320
-.08   -.08 .08 .08
dei 1 2 0 3 4;;1 2 0 3 4;
sfi -1 -4 ; ; -1 -4 ; ts 0 0 0 0 0 1 2 0 .2
dom 1 1 1 4 2 4
 z = z + 4.05
dom 1 2 1 4 2 4
 z = z - 4.05
dom 1 2 1 4 5 4
 z = z + 4.05 +
0.05*(j-79)/3
dom 1 5 1 4 5 4
 z = z - 4.1
dom 1 5 1 4 6 4
 z = z + 0.05*j
dom 1 6 1 4 6 4
 z = z - 12.45
dom 1 6 1 4 9 4
 z = z + 12.45 +
0.05*(j-249)/3
dom 1 9 1 4 9 4
 z = z - 12.5
dom 1 9 1 4 10 4
 z = z + 12.5
endpart
merge
```



2. Drill bit

```
sd 1 cy 0 0 0 0 0 1 1
sd 2 cy 0 0 0 0 0 1 .5
sd 3 cy 0 0 0 0 0 1 .1
sd 4 cy [.75*cos(30)] [.75*sin(30)] 0 0 0 1 .3
sd 5 cy [.75*cos(210)] [.75*sin(210)] 0 0 0 1 .3
sd 6 cone 0 0 0 0 0 1 1 45
ld 1 lp2 .1 .1 .1 -.1 -.1 -.1 -.1 .1 .1 .1;;
sd 7 cp 1 rx 90;
cylinder 1 4 7;1 6 11 16 21;1 101 121;.1 .5 1 0 90 180 270 360
         0 10 12
dei 2 3; 1 2 0 3 4; 1 2;
sfi -1;;;sd 3
sfi -2;;;sd 2
sfi -3;;;sd 1
sfi 2 3; -1 0 -5; 1 2;sd 4
sfi 2 3; -3; 1 2;sd 5
sfi ;; -1;sd 6
mbi ; -1 0 -3 0 -5; -3;y 20
dom 1 1 1 3 5 2
y=y+k*9
dom 1 1 2 3 5 3
y=y+909
dom 1 1 2 3 5 2
y=y-909
update
sfi -1;;;sd 7
merge
```



3. Helix



```
cylinder 1 5;1 226;1 5;1.5 2.5 0 900 -.5 .5
sfi -1 -2;1 2;-1 -2;ts 0 0 0 0 0 1 2 0 1
z=z+0.05*j
endpart
merge
```

4. Screw



```
neutral
sd 1 sp 0 0 0 2
sd 2 plan 0 0 1 0 0 1
sd 3 plan 0 0 0 0 0 1
sd 4 plan 0 -.3 0 0 1 0
sd 5 plan 0 .3 0 0 1 0
block 1 5 13 21 25;
      1 5 11 13 15 21 25;
      1 5 9;
      -.7 -.7 0 .7 .7
      -.5 -.5 -.3 0 .3 .5 .5
       0 .5 .5
dei 1 2 0 4 5;1 7;2 3;
dei 1 5;1 2 0 6 7;2 3;
de 2 3 2 4 5 3
dei 1 2 0 4 5;1 2 0 6 7;1 2;
sfi -1 -5;2 6;1 2;sd 1
sfi 2 4;-1 -7;1 2;sd 1
sfi 2 4;2 3 0 5 6;-3;sd 1
sf 0 0 2 0 0 2 sd 2
sf 0 0 1 0 0 1 sd 3
sfi 2 4;-2 -6;2 3;sd 2
sfi -2 -4;2 3 0 5 6;2 3;sd 2
sf 0 3 0 0 3 0 sd 4
sf 0 5 0 0 5 0 sd 5
z=-.75*z-.2
endpart
cylinder 1 2;
         1 37;
         1 2 3;
         0 .9
         0 360
         -.2 -.1 0
dom 1 1 2 2 2 2
z=-.1+.2*(j-1)/36
dom 1 1 3 2 2 3
z=.4*(j-1)/36
endpart
cylinder 1 2 3;
         1 1081;
         1 2 3;
         0 .9 1.3
         0 10800
```

```
         0 .2 .4
de 2 0 2 3 0 3
z=z+(j-1)*.4/36
dom 3 1 1 3 2 1
z=z+.09
dom 3 1 2 3 2 2
z=z-.09
dom 1 1 1 3 2 3
x=x*min((1+cos(-12.4*z)),1)
endpart
merge
```

X. Wheels

1. Bicycle Wheel - tire, felly, spokes and wheel-nave

```
parameter eps  4
          rad1 300
          rad2 35
          rad5 [%rad1-%rad2]
          sig [%rad2*3/(4*sqrt(2))]
          rad4 [%rad5*cos(80)]
          rad3 20
          rad6 10
          wnl 40
          rad7 [%wnl*(%rad5-%rad3)/(%rad5-%rad4)];
cylinder -1;1 289;1 9;
          [%rad1-%sig-%eps] 0 360 [-%sig] %sig
mate 1 thic 1
sfi -1;;;ts 0 0 0 0 0 1 [%rad1-%eps] 0 %rad2
endpart
cylinder 1 2 6 7; 1 145; 1 2 6 7;
         [%rad1-%sig] [%rad1-%sig] [%rad1+%sig] [%rad1+%sig]
         0 360 [-%sig] [-%sig] %sig %sig
mate 2
dei 1 2 0 3 4;;1 2 0 3 4;
sfi -1 -4;;-1 -4;ts 0 0 0 0 0 1 %rad1 0 %rad2
endpart
cylinder 1 3; 1 37; 1 3;
         %rad6 %rad3 0 360 [-%rad7] %rad7
mate 3
endpart
gct 1 rxy rz 10; lev 1 grep 0 1;; pslv
1
cylinder 1 6; 1 37; -1;
         %rad3 %rad4 0 360 %wnl
pb 1 0 0 1 0 0 z %rad7
mate 3
endpart
beam
cy [%rad5*.5] 10 [%wnl*.5];
cy %rad4  100 %wnl;
cy %rad5  20 0;
cy %rad4 -80 %wnl;
cy %rad5   0 0;
bm 2 3 3 4 1 1 ;
bm 4 5 3 4 1 1;
lct 8 rz 40;repe 8; lrep 0 1 2 3 4 5 6 7 8;
```



```
endpart merge
```

2. Fly wheel

```
sd 1 plan 0 0 0 0 1 0
sd 2 plan 0 .1 0 0 1 0
sd 3 plan 0 0 0 1 -1 0
sd 4 plan 0 .1 0 [-sin(30)] [cos(30)] 0
sd 5 plan 0 -.1 0 [-sin(30)] [cos(30)] 0
sd 6 cy 0 0 0 0 0 1 1
sd 7 cy 0 0 0 0 0 1 1.1
sd 8 cy 0 0 0 0 0 1 3
sd 9 cy 0 0 0 0 0 1 3.1
sd 10 cy 0 0 0 0 0 1 4.1
sd 11 cy 0 0 0 0 0 1 4.2
sd 12 cy 0 0 0 0 0 1 4.3
sd 13 cy 0 0 0 0 0 1 4.4
sd 14 ts 0 0 0 0 0 1 3.6 1.2 .6
sd 15 cy .35 0 0 0 0 1 .05
sd 16 cy [.35*cos(45)] [.35*sin(45)] 0 0 0 1 .05
sd 17 cy 0 0 0 0 0 1 .1
cylinder 1 3 12 14 22 24 26 28;
         1 2 10 12 16;
         1 2 7 8 9 10 11;
         1 1.1 3 3.1 4.1 4.2 4.3 4.4
         0 5 27 33 45
         0 .1 1 1.1 1.2 1.3
1.5
dei 2 3 0 4 5; 2 3 0 4 5; 2 7;
dei 2 8;; 6 7;
dei 2 3 0 4 5; 1 2 0 3 4; 3 7;
dei 5 6; 2 3 0 4 5; 2 5;
dei 5 6; 1 2 0 3 4; 3 5;
dei 6 7; 2 3 0 4 5; 2 4;
dei 7 8;; 5 6;
sfi ; -1;;sd 1
sfi ; -2;;sd 2
sfi ; -5;;sd 3
sfi ; -4;;sd 4
sfi ; -3;;sd 5
sfi -1;;;sd 6
sfi -2;;;sd 7
sfi -3;;;sd 8
sfi -4;;;sd 9
sfi -5;;;sd 10
```



```
sfi -6;;;sd 11
sfi -7;;;sd 12
sfi -8;;;sd 13
sfi 4 5;; -3;sd 14
bb 1 0 0 1 0 0 1;
lct 7 rz 45;repe 7;lrep 0:7;
endpart
cylinder 1 5 7 21;
         1 3 4 13 14 16;
         1 4 11;
         .1 .3 .4 1
         0 0 10 35 45 45
         0 .4 1.4
dei 1 2; 1 2 0 5 6;;
dei 2 3; 1 3 0 4 6;;
mbi 3 4; -5;;y -5
mbi 3 4; -2;;y 5
mbi -3; 3 4;;x -.02
mbi 3 4; -3;;y 5
mbi 3 4; -4;;y -5
mbi -1;; -3;x .1
mbi -2;;;x -.01
mbi -2; -3;;y -2
mbi -2; -4;;y 2
sfi ; -1;;sd 1
sfi ; -6;;sd 3
sfi 1 2; -2;;sd 1
sfi 1 2; -5;;sd 3
sfi -1;; 1 2;sd 17
sfi -4;;;sd 6
sfi -2 -3; 1 -3;;sd 15
sfi -2 -3; -4 6;;sd 16
res 1 2 1 1 5 3 j 1
res 3 4 1 3 6 3 j 1
res 3 1 1 3 3 3 j 1
bb 4 0 0 4 0 0 1;
lct 7 rz 45;repe 7;lrep 0:7;
endpart
merge
```

3. Valve



```
sid 1 tied;
gct 4 rzx ; rz 90; rz 180; rz 270;
lev 1 grep 0 2 3 4;;
pslv 1
lev 2 grep 0 1; ;
pslv 2
cylinder 1 3 7 11;
       1 4 15;
       1 3 6 8;
       .75 1.25 4.625 5.75;
       0. 5. 45.;
       [-7/16] [-3/16] [3/16] [7/16]
dei 1 3;; ;
sfi -3 -4; ; -1 -4; ts 0 0 0 0 0 1 5.25 0 [9/16]
sfi 1 2;; -1; plane 0 0 -.375 0 0 1
sfi 1 2;; -4; plane 0 0  .375 0 0 1
sfi  ; ; -2;  plane 0 0 -.1875 0 0 1
sfi  ; ; -3;  plane 0 0  .1875 0 0 1
sfi ; -2; 2 3; cy 0 .375 0 1 0 0 [3/16]
endpart
block 1 3 7 11; 1 4; 1 4;
      .75 1.25 4.625 5.25;
      0 .5;-.1875 .1875;
dei 1 2 0 3 4; ; ;
sfi ; -2; ; cy 0 .375 0 1 0 0 [3/16]
sfi -2 ; ; ; cy 0 0 0 0 0 1 1.25
sfi -3 ; ; ; ts 0 0 0 0 0 1 5.25 0 [9/16]
sfi  ; ; -1;  plane 0 0 -.1875 0 0 1
sfi  ; ; -2;  plane 0 0  .1875 0 0 1
mate 2
endpart
cylinder 1 3 ; 1 4 8; 1 3 6 8;
         .75 1.25;
         0. 20. 45.;
         [-7/16] [-3/16] [3/16] [7/16]
sfi ;; -1; plane 0 0 -.375 0 0 1
sfi ;; -4; plane 0 0  .375 0 0 1
sfi  ; ; -2;  plane 0 0 -.1875 0 0 1
sfi  ; ; -3;  plane 0 0  .1875 0 0 1
sfi -2; -2; 2 3; cy 0 .375 0 1 0 0 [3/16]
sfi -2 ; ; ; cy 0 0 0 0 0 1 1.25
mate 3
endpart
```

```
pplv
pplv
block 1 2 5 6; 1 2 5 6; 1 19 20 21;
      -.4 -.4 .4 .4;
      -.4 -.4 .4 .4;
      -15 .375 1.125 1.5;
dei 1 2 0 3 4; 1 2 0 3 4; ;
sfi -1 -4; -1 -4; ; cy 0 0 0 0 0 1 .75
sii -1 -4; -1 -4; 2 3; 1 m
mate 4
endpart
cylinder 1 4; 1 5; 1 2;.75 1.25
         -30. 30.;
         .375  1.125;
sfi -2 ; ; ; plane 1.25  0 0 1 0 0
sfi ; -1; ; plane 0 0 0 -.5 -.866 0
sfi ; -2; ; plane 0 0 0 -.5  .866 0
lct 6 rz 60; repe 6; ;
lrep 1 2 3 4 5 6;
sii -1; ; ; 1 s
mate 5
endpart
merge
```

XI. Annotated Examples

1. Fillet Example

Introduction

This example shows how to build a simple model of the trailing section of an impeller blade. The intent is to demonstrate the use of composite surfaces in modeling fillets. The geometry is simply done, where more complex geometry can easily be substituted. A lot of space here is devoted to the geometry, but the details should be skipped. The key to this example is the initialization of the vertices which naturally fall on the fillet surface, and the projection of the mesh onto composite surfaces.

For completeness, the beginning of the blade portion and the trailing edge of the blade have been placed onto cylinders. The initialization of the mesh is critical to getting a good mesh. (And this is typical.)

Fillet surfaces are usually tangent to the surfaces they meet. Placing an edge of the mesh along the curve where the two tangent surfaces meet is numerically very difficult. **True*Grid*®** will search for the intersection of tangent surfaces, but by the nature of the problem, the results may not be as accurate as other types of intersection calculations and it may take a long time for the calculations to complete. There is a special case where two tangent surfaces have bounding edges along the tangent curve. This type of intersecting tangent surfaces does not have to be avoided. But even in this case and in all others, the edge of the mesh is usually best placed down the middle of the fillet instead of along the tangent curve, avoiding all intersection calculations. It usually forms the best quality mesh as well. Alternative methods usually require severe skewing of elements or the use of wedge elements, both of which can have undesirable consequences in the analysis.

2D Geometry

Two dimensional curves are used in this example to create a simple geometry. These curves will be used to build 3D curves, which in turn will be used to form 3D surfaces. These 2D curves should be viewed as cross section curves of the blade. They are numbered in the picture corresponding to the commands that created them in the command file below.



Figure 85 2D Curves for the blade

3D Geometry

The 2D curves are then extruded into 3D curves. These 3D curves are then combined to form ruled surfaces. There are 16 2D curves which produced 16 surfaces depicted below.



Figure 86 16 Ruled surfaces Forming The Blade

Vertex Placement

The placement of the corner nodes (vertices) of the block mesh is key to meshing a fillet. Use the mouse to select the vertex and move it to the desired position along the fillet. Below is a 2D example where the corner vertex is placed on the fillet. (The ***Move Pts.***button, in the Environment Window, can be used with appropriate coordinate constraints **or** simply rotate the geometry so that you can move in the ***Screen Plane***. Don't be too exact when initializing. Let the subsequent projection make the final placement. You can work more quickly with this strategy.)



Figure 87 Before Positioning



Figure 88 After Positioning

Surface Composition



Figure 89 Compose 1 & 2

Two edges of the mesh will be projected onto the fillet. For example, the nearly horizontal edge must be placed partly onto the arc and partly onto a horizontal line.



Figure 90 Compose 2, 3, & 4

Before projecting the mesh to the geometry, define composite surfaces. Surface 17 is composed of 1 and 2, surface 18 of 2, 3, and 4, and surface 19 of 4 and 5. Notice that surfaces 17 and 18 overlap at surface 2. Surfaces 18 and a9 overlap at surface 4. This is needed to avoid intersecting the tangent surfaces.



Figure 91 Compose 4 & 5

Projection Onto Composites

The last step in meshing this fillet is to project the horizontal and diagonal edges of the mesh to the composite surfaces 17 and 18 respectively.



Figure 92 Fillet Formation

Completed Model

Composite surfaces 17, 18, and 19 where used to form the fillets on one side of the blade. To project onto the fillets on the other side of the blade, the composite surfaces 20, 21, and 22 were also defined. The final results are shown below. The parameter at the beginning of the command file in the next page makes it easy to change the mesh density.



Figure 93 Fillets Using 6 Composite Surfaces

Command File

```
title Fillet

para d 3; c  mesh density

c
c  2D Curves
c

ld  1 lp2 0 0 1 0;  c  line segment
ld  2 lp2 1 0;lad 1 .1 120  c  arc of a circle (fillet)
ld  3 lp2 [1+cos(30)*.1] [.1+sin(30)*0.1] .1 2.05;
                                        c  line segment
ld  4 lp2 .1 2.05;lad [.1-cos(30)*0.1] 2 60;  c  fillet
ld  5 lp2 [.1-cos(30)*0.1] 2.1 -1 2.1;  c  line segment
ld  6 lp2 -1 2.1 -1 2.3;                c  line segment
ld  7 lp2 -1 2.3 1.4 2.3;               c  line segment
ld  8 lp2 1.4 2.3 1.4 2.1;              c  line segment
ld  9 lp2 1.4 2.1 .4 2.1;               c  line segment
ld 10 lp2 .4 2.1;lad .4 2 120;          c  fillet
ld 11 lp2 [.4-cos(30)*.1] [2-sin(30)*.1] 1.3 .05;
                                        c  line segment
ld 12 lp2 1.3 .05;lad [1.3+cos(30)*0.1] .1 60;
                                         c  fillet
ld 13 lp2 [1.3+cos(30)*0.1] 0 2.4 0;  c  line segment
ld 14 lp2 2.4 0 2.4 -.2;  c  line segment
ld 15 lp2 2.4 -.2 0 -.2;  c  line segment
ld 16 lp2 0 -.2 0 0;  c  line segment

c
c  3D Curves
c
c  Each 2D curve is a piece of a cross section of the blade.
c  Each 2D curve is turned into a 3D curve in two ways, one for
c  the beginning cross section of the blade, and one for the
c  trailing cross section of the blade.
c

curd  1 ld2d3d  1 rt y 11 11; c Beginning cross section of blade
curd  2 ld2d3d  2 rt y 11 11;
curd  3 ld2d3d  3 rt y 11 11;
curd  4 ld2d3d  4 rt y 11 11;
curd  5 ld2d3d  5 rt y 11 11;
curd  6 ld2d3d  6 rt y 11 11;
```

```
curd  7 ld2d3d  7 rt y 11 11;
curd  8 ld2d3d  8 rt y 11 11;
curd  9 ld2d3d  9 rt y 11 11;
curd 10 ld2d3d 10 rt y 11 11;
curd 11 ld2d3d 11 rt y 11 11;
curd 12 ld2d3d 12 rt y 11 11;
curd 13 ld2d3d 13 rt y 11 11;
curd 14 ld2d3d 14 rt y 11 11;
curd 15 ld2d3d 15 rt y 11 11;
curd 16 ld2d3d 16 rt y 11 11;
curd 21 ld2d3d  1 rt y 14 14 mx 1 mz 1; c trailing cross section
curd 22 ld2d3d  2 rt y 14 14 mx 1 mz 1; c of blade
curd 23 ld2d3d  3 rt y 14 14 mx 1 mz 1;
curd 24 ld2d3d  4 rt y 14 14 mx 1 mz 1;
curd 25 ld2d3d  5 rt y 14 14 mx 1 mz 1;
curd 26 ld2d3d  6 rt y 14 14 mx 1 mz 1;
curd 27 ld2d3d  7 rt y 14 14 mx 1 mz 1;
curd 28 ld2d3d  8 rt y 14 14 mx 1 mz 1;
curd 29 ld2d3d  9 rt y 14 14 mx 1 mz 1;
curd 30 ld2d3d 10 rt y 14 14 mx 1 mz 1;
curd 31 ld2d3d 11 rt y 14 14 mx 1 mz 1;
curd 32 ld2d3d 12 rt y 14 14 mx 1 mz 1;
curd 33 ld2d3d 13 rt y 14 14 mx 1 mz 1;
curd 34 ld2d3d 14 rt y 14 14 mx 1 mz 1;
curd 35 ld2d3d 15 rt y 14 14 mx 1 mz 1;
curd 36 ld2d3d 16 rt y 14 14 mx 1 mz 1;

c
c  3D surfaces
c
c  Each pair of 3D curves is used to form a ruled surface.
c

sd  1 rule3d  1 21;
sd  2 rule3d  2 22;
sd  3 rule3d  3 23;
sd  4 rule3d  4 24;
sd  5 rule3d  5 25;
sd  6 rule3d  6 26;
sd  7 rule3d  7 27;
sd  8 rule3d  8 28;
sd  9 rule3d  9 29;
sd 10 rule3d 10 30;
sd 11 rule3d 11 31;
sd 12 rule3d 12 32;
```

```
sd 13 rule3d 13 33;
sd 14 rule3d 14 34;
sd 15 rule3d 15 35;
sd 16 rule3d 16 36;
c
c  Composite surfaces
c
c  These surfaces are combined so that a face of the mesh
c  can be projected to the union of the a straight surface
c  and a fillet surface.
c

sd 17 sds 1 2;
sd 18 sds 2 3 4;
sd 19 sds 4 5;
sd 20 sds 9 10;
sd 21 sds 10 11 12;
sd 22 sds 12 13;

c
c  Beginning and trailing surfaces
c

sd 23 cy 0 0 0 0 0 1 12
sd 24 cy 0 0 0 0 0 1 14

c
c  Blade part initialized
c

block 1 [1+%d*5] [1+%d*6] [1+%d*11];
      1 [1+%d*11];
      1 [1+%d] [1+%d*11] [1+%d*12];
      0 1.1 1.3 2.4
      11 14
      -.2 0 2.1 2.3

c
c  Several regions are deleted between hub and shroud on both
c  sides of the blade.
c

dei 1 2 0 3 4;;2 3;

c
c  Regions of the mesh are moved from the initial positioning
```

```
c  to match the twist in the geometry.
c

mbi ;;3 4;x -1
mbi ;-2;;xz 1 1

c
c  The mouse point-and-click is used to move the vertices of the
c  block mesh to key positions along the geometry. Particular
c  attention is needed in positioning the vertices near the
c  fillets.
c

pb 1 1 1 1 1 1 xyz 0.332472 11.9908 0.133370
pb 1 1 2 1 1 2 xyz 0.324061 11.9917 0.328819
pb 1 1 3 1 1 3 xyz -0.671215 11.9801 2.42951
pb 1 1 4 1 1 4 xyz -0.659519 11.9805 2.64011
pb 2 1 1 2 1 1 xyz 1.40077 11.9139 0.104083
pb 2 1 2 2 1 2 xyz 1.40249 11.9134 0.323605
pb 2 1 3 2 1 3 xyz 0.411942 11.9799 2.41842
pb 2 1 4 2 1 4 xyz 0.425266 11.9829 2.62633
pb 3 1 1 3 1 1 xyz 1.59243 11.8848 0.940618e-01
pb 3 1 2 3 1 2 xyz 1.60283 11.8778 0.300943
pb 3 1 3 3 1 3 xyz 0.630181 11.9742 2.40571
pb 3 1 4 3 1 4 xyz 0.626022 11.9782 2.62678
pb 4 1 1 4 1 1 xyz 2.63271 11.7042 0.333894e-01
pb 4 1 2 4 1 2 xyz 2.63577 11.7048 0.234095
pb 4 1 3 4 1 3 xyz 1.69362 11.8749 2.39036
pb 4 1 4 4 1 4 xyz 1.69230 11.8747 2.58992
pb 1 2 1 1 2 1 xyz 0.987701 13.9608 0.786518
pb 1 2 2 1 2 2 xyz 0.986564 13.9607 0.986943
pb 2 2 1 2 2 1 xyz 2.04778 13.8437 0.748686
pb 2 2 2 2 2 2 xyz 2.04022 13.8444 1.00796
pb 2 2 3 2 2 3 xyz 1.05377 13.9552 3.06991
pb 2 2 4 2 2 4 xyz 1.08439 13.9513 3.28339
pb 3 2 1 3 2 1 xyz 2.23867 13.8157 0.738955
pb 3 2 2 3 2 2 xyz 2.27138 13.8158 0.957148
pb 3 2 3 3 2 3 xyz 1.28922 13.9341 3.02537
pb 3 2 4 3 2 4 xyz 1.27843 13.9349 3.27866
pb 4 2 1 4 2 1 xyz 3.27143 13.6140 0.671352
pb 4 2 2 4 2 2 xyz 3.27267 13.6104 0.868432
pb 4 2 3 4 2 3 xyz 2.34297 13.7950 3.02948
pb 4 2 4 4 2 4 xyz 2.33218 13.8001 3.23173

c
```

```
c  Faces of the mesh are projected onto surfaces.
c  Faces that transition from a straight surface to a fillet
c  are projected onto the appropriate composite surface.
c

sfi -1;; 1 2;sd 16
sfi -1;; 3 4;sd 6
sfi -4;; 1 2;sd 14
sfi -4;; 3 4;sd 8
sfi -2;; 2 3;sd 18
sfi -3;; 2 3;sd 21
sfi ;; -1;sd 15
sfi ;; -4;sd 7
sfi 1 2;; -3;sd 19
sfi 3 4;; -3;sd 20
sfi 1 2;; -2;sd 17
sfi 3 4;; -2;sd 22
sfi ; -1;;sd 23
sfi ; -2;;sd 24

endpart
```

2. Elbow

Introduction

In this example, we thought it more important to give a picture of the final mesh first, so that you may have a better idea of what the geometry looks like. The following input file is heavily commented.



Figure 94 Pipe Elbow

```
title Elbow

c This is a model of an elbow connector. A pipe is joined to a
c base at one end and a flange at the other end. The pipe is
c bent so that the base and flange are at a 45 degree angle.
c another. The pipe is joined to the base and flange by fitting
c into a hole in the middle.  The base is square with
c rounded corners, and has four bolt holes.  The flange
c is circular, with four notches for bolts.

c The zoning is made very coarse so pictures of the model
c will reproduce well. To make it finer, just increase the
c numbers of nodes in the two block commands and the one
c cylinder command.

c Global Geometry

c We use a Cartesian coordinate system where the base is a square
c in the x-y plane, with a little thickness in the z direction.
c The pipe goes out of the base in the +z direction (i.e.,
c the direction of the unit vector (0,0,1), and curves
c around in the -x direction, so that at the flange it ends out
c pointing in the direction (0,1,-1). There the flange has
c flat surfaces normal to the pipe.
c Thus the whole model is symmetrical in y (i.e., unchanged by
c reflection in the xz plane).

c Define parameters for center of pipe at near side of flange:

para xcf [ -4.88 + 5.38/sqrt(2) ] ;     $ x coordinate

c y-coordinate is 0

para zcf [ 4.38/sqrt(2) ] ;             $ z coordinate

c The next two surfaces are used to control node distribution.
c They are two planes forming an "X" normal to the planar
c surfaces of the flange.

sd 1 plane [%xcf] 0 [%zcf] [1/sqrt(2)] 1 [1/sqrt(2)]
sd 2 plane [%xcf] 0 [%zcf] [1/sqrt(2)] -1 [1/sqrt(2)]

c Define the torus that forms the outside of the pipe (except
c for the parts of the pipe that fits the cylindrical holes
```

```
c of the base & flange).

sd 3 ts -4.88 0 0 0 1 0 4.88 0 1.19

c Define the planar surfaces of base

sd 5 plane 0 0 -0.5   0 0 1
sd 6 plane 0 0 0      0 0 1

c Two cylinders for the pipe holes in the middle of the base
c and the flange:

sd 7 cy 0 0 0 0 0 1 1.19      $ hole in the middle of the base
sd 8 cy [%xcf] 0 [%zcf] -1 0 1 1.19    $ pipe hole in the flange


c  Part 1:  Base
c We shall take advantage of the base's symmetry. First define
c a block for a quarter of the base. Later we will replicate
c this block to make the whole base.
c indices 1..11, reduced indices 1..5 in i (x) direction
c indices 1..11, reduced indices 1..5 in j (y) direction
c indices 1..3,  reduced indices 1..2 in k (z) direction

block
     1     3     5   7 11 ;     1     3     5   7 11 ; 1    3 ;
-1.875 -1.45 -1.17 -.7  0. -1.875 -1.45 -1.17 -.7  0.  -0.5 0.

dei 4 5; 4 5; ;     $ Delete corner of block for the big hole in
                    $ the middle of the base.
dei 2 3; 2 3; ;     $ Delete some of the middle of the block for
                    $ the 4 little bolt holes.

c Fit the bolt hole to a cylinder:

sfi -2 -3; -2 -3; ;
cy -1.31 -1.31 0 0 0 1 .19
sfi -4 5; -4 5;   ; sd 7 $ Make the pipe hole cylindrical.

c Round the outer corner by fitting to another cylinder:

sfi -1 2; -1 2;   ; cy -1.31 -1.31 0 0 0 1 .56

c Require the top and bottom to be flat. Actually, it would be
c flat anyway because the original block (above) was flat on
c all sides, but it is a good practice to explicitly specify
```

```
c the surfaces we care about.

sfi 1 2 3 4 ; 1 2 3 4 ; -1 ; sd 5
sfi 1 2 3 4 ; 1 2 3 4 ; -2 ; sd 6

c Move the initial positions of some of the vertices.
c These vertices are in the interior of the final base plate
c (some are on the outer surface of this quarter-base)
c with the i- or j- locations of a bolt hole. Without the
c following adjustments, the effect of the projection to the
c pipe hole would be too confined to the regions near the
c pipe hole. With these adjustments the nodes get better
c distributed over the whole base.

mb 2 5 1 3 5 2 x -.26
mb 2 4 1 3 4 2 x -.18
mb 5 2 1 5 3 2 y -.26
mb 4 2 1 4 3 2 y -.18

c Another way we can improve the quality of the mesh is to apply
c a few relaxation iterations to the faces that are near and far
c in the z direction. Note that we do this piece-by-piece, as it
c doesn’t make sense to relax the nodes in a hole.
c (This example was done before an improvement was made to
c TrueGrid’s relax command. Now TrueGrid ignores holes
c automatically, so only one relax command is needed here.)

relaxi 1 4; 3 5; -1 0 -2; 5 .1
relaxi 3 5; 1 4; -1 0 -2; 5 .1
relaxi 1 2; 1 4; -1 0 -2; 5 .1
relaxi 1 4; 1 2; -1 0 -2; 5 .1

c Now that we have a block that does a good job of modeling
c a quarter of the base, we replicate it to make the whole base
c part. The way to do this is to define three coordinate
c transformations(numbered 1, 2, 3) as reflections; and then
c apply them to the block.
c (Transformation # 0 means to keep the original block as well.)

lct 3 ryz; rzx; ryz rzx;
lrep 0 1 2 3;

c Assign material number 1 to this part.

mate 1
```

```
endpart

c  Part 2:  Pipe
c The pipe provides the transition from the square base
c (defined by block, in Cartesian coordinates)
c to the circular flange (defined by cylinder, in cylindrical
c coordinates). We shall begin with the block command
c (for Cartesian coordinates),cut notches in the corners,
c and paste them together so that we can make it
c nicely fit the toroidal shape of a bent pipe.
c At the base end, we use the same coordinates as the base
c and don't do much with the pipe, so it fits the base
c rather easily. At the flange end, we have to go to some effort
c to give the nodes a good distribution and make it fit nicely
c into the flange.
c indices 1..13, reduced indices 1..4 in i (x) direction
c indices 1..13, reduced indices 1..4 in j (y) direction
c indices 1..17, reduced indices 1..4 in k (z) direction

block
  1  3 11 13 ; 1  3 11 13 ;    1 3 15   17 ;
 -1 -1  1  1  -1 -1  1  1   -0.5 0  2.5 3.0

dei 2 3; 2 3; ;               $ Delete inside of pipe
dei 1 2 0 3 4; 1 2 0 3 4; ;   $ Delete outer corners

c (We don't want outer corners because of how the square
c will be distorted into a circle.)

c Fit outside and inside of pipe to torus (this both converts
c the square cross-section to a circle and bends the pipe 45
c degrees):

sfi -1 -4 ; -1 -4 ; 2 3 ; sd 3     $ torus for main part of pipe
                                   $ (outside)
sfi -1 -4 ; -1 -4 ; 1 2 ; sd 7     $ cylinder for pipe in base
                                   $ (outside)
sfi -1 -4 ; -1 -4 ; 3 4 ; sd 8     $ cylinder for pipe in flange
                                   $ (outside)
sfi -2 -3; -2 -3; ;
ts -4.88 0 0 0 1 0 4.88 0  .81     $ inside

c Make cross-section of pipe near end match plane of near end
c of flange (a side effect is to makes nodes reasonably
c distributed in z direction):
```

```
sfi ; ; -3 ; plane  -4.88 0 0 -1 0 1

c make end of pipe match plane of far end of flange:

sfi ; ; -4 ; plane [-4.88-sqrt(2)/2] 0 0 -1 0 1

c Do the same at the base end
c (not as big a deal, the pipe is already almost right there):

sfi ; ; -1 ; sd 5
sfi ; ; -2 ; sd 6

c The remaining problem is that the above two projections
c distributed the nodes rather poorly in an angular direction
c at the flange end. This is because projecting the
c flange end onto planes at a different angle will spread some
c nodes out, compress others together. So now we work on just
eight
c nodes of the pipe: four at the near plane of the flange and
c four at the far plane. We move them so that they will be
c equally spaced around the circles on the inside of the flange.
c TrueGrid will automatically move all the other points where
c we want them, once it gets this little bit of guidance.
c A bonus: now that the cylinder nodes are well distributed,
c if we also give the flange nodes a good distribution, it will
c be easy to merge them together with the pipe nodes.

sfi -2 ; -1 0 -2 ; -3 0 -4 ; sd 2
sfi -1 ; -2 ; -3 0 -4 ; sd 2

c ... later (1,2,4) will be merged with (2,1,4) etc

sfi -2 ; -3 0 -4 ; -3 0 -4 ; sd 1
sfi -1 ; -3 ; -3 0 -4 ; sd 1

c ... later (1,3,4) will be merged with (2,4,4) etc

sfi -3 ; -1 0 -2 ; -3 0 -4 ; sd 1
sfi -4 ; -2 ; -3 0 -4 ; sd 1

c ... later (4,2,4) will be merged with (3,1,4) etc

sfi -3 ; -3 0 -4 ; -3 0 -4 ; sd 2
sfi -4 ; -3 ; -3 0 -4 ; sd 2
```

```
c ... later (4,3,4) will be merged with (3,4,4) etc
c Assign material number 2 to this part

mate 2
endpart


c  Define and apply global coordinate transformation.
c The reason for this is to let us define the flange in
c a convenient coordinate system. Then this transformation
c will move it into place.

c First define the transformation as follows:
c Rotate by 45 degrees (the amount the pipe is bent by) and
c move from the origin to the center of the pipe at the near
c plane of the flange.

gct 1                          $ Transformation number 1 ...
ry -45 mx [%xcf] mz [%zcf];    $ ... is a rotation and translation
lev 1 grep 1;;                 $ Define a level to be the
                               $ application of this transform.
pslv 1                         $ Apply the level.


c  Part 3:  Flange
c We start by making a quarter of the flange. Later we shall
c use its symmetry to make the whole thing.
c Thanks to our coordinate transformation (level 1),
c we can define the flange in terms of the most convenient
c coordinates: with origin at center of flange and
c the z axis the same as the axis of the flange.
c indices 1..8, reduced indices 1..4 in i (r) direction
c indices 1..9, reduced indices 1..4 in j (theta) direction
c indices 1..3, reduced indices 1..2 in k (z) direction

cylinder
  1    4   6    8 ;  1  2   8  9 ;   1   3 ;
  1.19 1.5 1.69 1.94 0. 6. 84. 90.  0.5 1.0

c Make the bolt notches (two half-notches at each end
c of the quarter-flange):

dei 2 4 ; 1 2 0 3 4 ; ;

c Shape the notches with cylinders on inner ends, planes
c near outside:
```

```
sfi  -2 3 ; 1 -2 ; ; cy 1.69 0 0 0 0 1 .19
sfi  -2 3 ; -3 4 ; ; cy 0 1.69 0 0 0 1 .19
sfi 3 4 ; -2 ; ;  plane 0 .19 0 0 1 0
sfi 3 4 ; -3 ; ;  plane .19 0 0 1 0 0

c Evenly space nodes on the part's two inside edges - this
c both helps to fit the pipe and improves the quality of the
c inside node distribution:

res 1 1 1 1 4 1 j 1.0
res 1 1 2 1 4 2 j 1.0

c Now that we have a good model of a quarter of the flange,
c we make three more copies of it, to get a model of the
c whole flange.
c Define three local coordinate transformations (numbered 1, 2,
3)
c as rotations about the center of the flange.
c Then apply them to replicate the quarter-flange
c (transformation 0 means to keep the original copy).

lct 3 rz 90; rz 180; rz 270;
lrep  0 1 2 3;

c Assign material number 3 to this part

mate 3
endpart

c  Finish Up and Merge the Parts into One Model

pplv      $ End applying global coordinate transform.(level 1)
          $ ... Thus the transformation affects only the flange.
merge     $ Go to the Merge phase of TrueGrid.

c Combine our three parts into one model. We do this
c by merging into one point all pairs of points that are
c practically the same (within a tolerance of .01).
c stp merges only points on surfaces of parts.
c In this case, tp would do the same thing.

stp .01

c ...TrueGrid reports 460 nodes merged.
c For demonstration purposes, the following gives an exact
```

```
c calculation of the right number of nodes to merge.
c But for most practical problems, we wouldn't bother
c to figure it out so carefully.
c 84: part 1 was made from 4 copies of another part, so the faces
c where the four copies touch each other have extra nodes:
c 7 in i or j direction, 3 in k direction, 4x7x3=84
c (TrueGrid will report as a 1 & 1 merge)
c 136: part 2 had 4 notches whose sides were pasted together;
c side length=3, 2 extra nodes, pipe length 17, 4x2x17=136
c (TrueGrid will report some as a 2 & 2 merge, others -
c up to 24 - as a 1 & 2 merge)
c 96: after the above merge within part 2, there are common
c nodes where the pipe fits into the base.
c Circumference is 32 nodes, depth 3 nodes, 3x32=96
c (TrueGrid reports these and others as a 1 & 2 merge).
c 96: pipe fits into flange, circumference 32 nodes,
c depth 3 nodes 3x32=96.
c (TrueGrid reports these and others as a 2 & 3 merge)
c 48: flange was made from 4 copies of another part,
c so the faces where the four copies touch each other have
c extra nodes: 4 in i or j direction, 3 in k direction,
c 4x4x3=48 (TrueGrid will report some as a 3 & 3 merge, the rest
c up to 12 - as a 2 & 3 merge)
c 84 + 136 + 96 + 96 + 48 = 460
```

3. Wing And Fuselage

Introduction

This is a model of a cylindrical fuselage where it intersects a wing. The purpose of this model is to demonstrate the various methods of distributing the mesh along surfaces. Several interpolation methods will be presented for comparison. The wing shape is modeled by a simple function. This is only an approximation to a fuselage and wing. For a more realistic model, replace these surfaces with more realistic ones such as NURBS surfaces. This mesh is a C-grid, with the minimum j-index following the shape of the wing.



Figure 95 Surfaces for Defining Approximate Wing and Fuselage

```
title wing
para rad 2;                              c  radius of the fuselage
sd 1 function -1 1 30 62
 (v-30)*cos(3*v)*cos(v)/30; cos(3*v)*sin(v); u;rz 120;

                         c bottom of the wing along fuselage

sd 2 function -1 1 62 90
 (v-30)*cos(3*v)*cos(v)/30; cos(3*v)*sin(v); u;rz 120;

                        c  top of the wing along fuselage

sd 3 cy 0 0 [-%rad] 1 0 0 %rad
sd 4 plan -1 0 0 1 0 0                   c left plane intersecting
                                         c fuselage
sd 5 plan 2 0 0 1 0 0                    c right plane intersecting
                                         c fuselage
sd 6 plan 0 0 [-%rad*.5] 0 0 1           c  back plane

c  Initialize the surface part.
c  Only the z-coordinate is initialized.

block 1 11 31 51 71 91 101;1 21;-1;;;0;

c  Position the vertices interactively, using the PBS command.
c  The "tsave" file converts the PBS commands to PB commands.
c  Only the x and y-coordinates are initialized

c  Nodes along the minimum j-index along the wing.

pb 1 1 1 1 1 1 xy -1        0
pb 2 1 1 2 1 1 xy  0        0
pb 3 1 1 3 1 1 xy  0.804906  0.917935e-01
pb 4 1 1 4 1 1 xy  1.00906   0.478386e-02
pb 5 1 1 5 1 1 xy  0.797057  0.153867e-02
pb 6 1 1 6 1 1 xy  0        0
pb 7 1 1 7 1 1 xy -1        0

c  Nodes along the maximum j-index along the outer boundary.

pb 1 2 1 1 2 1 xy -1        1
pb 2 2 1 2 2 1 xy  0        1
pb 3 2 1 3 2 1 xy  2        1
pb 4 2 1 4 2 1 xy  2        0
pb 5 2 1 5 2 1 xy  2       -1
```

```
pb 6 2 1 6 2 1 xy  0        -1
pb 7 2 1 7 2 1 xy -1        -1

c  Nodes are distributed along the edges, clustering
c the nodes near the critical spots in the geometry.

res 1 1 1 7 2 1 j 1.3        c  Cluster towards the wing in the
                             c j-direction.
res 1 1 1 2 1 1 i [1/1.4]    c  Cluster in the i-direction near
                             c the tip.
res 6 1 1 7 1 1 i 1.4
drs 2 1 1 4 1 1 i 1.1 1.3    c  Cluster in the i-direction along
                             c the wing.
drs 4 1 1 6 1 1 i 1.3 1.1
res 3 2 1 4 2 1 i [1/1.3]    c  Cluster right edge to match the
                             c left edge.
res 4 2 1 5 2 1 i 1.3

sfi 2 4;-1;;sd 2            c  Project onto the upper edge
                            c  of the wing.
sfi 4 6;-1;;sd 1            c  Project onto the lower edge
                            c  of the wing.
sfi ;;-1;sd 3               c  Project the entire mesh onto
                            c  the fuselage.
sfi -1 0 -7;;-1;sd 4        c  Project the left edge onto
                            c  a plane.
sfi 3 4;-2;-1;sd 5          c  Project the right edge onto
                            c  a plane
sfi 1 3 0 5 7;-2;-1;sd 6    c  Project top and bottom onto
                            c  back plane.
```

If the mesh is displayed at this point, **True*Grid*®** will automatically interpolate each rectangular block of the mesh using the default modified bi-linear interpolation method.



Figure 96 Default Bi-Linear Interpolation per Block

Transfinite interpolation can be used to interpolate the entire part. This is an algebraic method. Notice that the sharp corners from the right edge are replicated far into the interior. I f you zoom in on this example near the leading edge of the foil, you will notice that the mesh lines around the foil closely resemble the foil. This feature is hard to get because the foil forms a concave boundary.



Figure 97 tfi ;;-1;

The Thomas-Middlecoff elliptic method can produce a smooth mesh with lines that follow the shape of the boundaries. It also builds elements with angles close to 90 degrees. This method is the best for boundary layer problems, although it is more expensive than the other methods. Notice that some of the elements are larger than the elements produced by the transfinite interpolation method.



Figure 98 tmei ;;-1; 100 0 1

Sometimes a hybrid method can produce the best results. **True*Grid*®** always performs interpolation before relaxation, regardless of whether you issued your interpolation commands before or after your relaxation commands. Therefore **True*Grid*®** uses the results of the interpolation as the initial conditions for the relaxation. In this example, transfinite interpolation is executed first, and then Thomas-Middlecoff relaxation with 25 iterations.



Figure 99 tfi ;;-1; tmei ;;-1; 25 0 1

The equipotential relaxation method is one of the oldest methods of producing a smooth mesh. It is mentioned here to compare it to the other methods. This method is used for structures, and usually does not have dense clustering of nodes along the boundary. It tries to produce equal elements and it quickly forgets the shape of the boundaries.



Figure 100 relaxi ;;-1; 100 0 1

4. Fluid In A Simple Impeller

Introduction

This is a model of the fluid in a simple impeller. Because of its symmetry, it is sufficient to model just a 40 degree segment. The following picture shows the mesh. Note how the mesh lines are clustered in the middle of the region at the leading edge of the blades. This feature of the mesh was needed to solve the fluid flow near the blades in the fluid simulation code TASCflow.



Figure 101 40 Degree Segment Of Fluid

```
title simple impeller

para z1 .392893          c  center of fillet of blade on the hub
     x1 .685786          c  at angle 0
     z2 [.333955+.25]    c  center of fillet of blade on shroud
     x2 .878704          c  at angle 0
     x3 [%x1*cos(40)]    c  center of fillet of blade on the hub
     y3 [%x1*sin(40)]    c  at angle 40
     x4 [%x2*cos(40)]    c  center of fillet of blade on shroud
     y4 [%x2*sin(40)]    c  at angle 40
     theta .25           c  half the angular gap at the lead edge
     rho1 [2*%x1*sin(%theta/2)]   c radius of the fillet at hub
     rho2 [2*%x2*sin(%theta/2)] ; c rad. of the fillet at shroud

ld 1 lep 2 1 2.1 1.1 196 270 0 ;   c  elliptic arc for the hub
ld 2 lep 2 1 2.1 1.1 206 270 0 ;   c  ellipt. arc for hub blade
ld 3 lep 1.9 1 2.1 1.1 212 270 0; c  ellipt. arc for shroud blade
ld 4 lp2 .367949 .6 .367949 1;     c  line segm. for cyl. for hub
ld 5 lp2 2.1 .1 2.1 .35 ;          c  line segm. for outlet cyl.
ld 6 lep 2 1 2.1 1.1 196 206 0 ;   c  ellipt. arc for hub
                                   c  periodic boundary
ld 7 lep 1.9 1 2.1 1.1 200 212 0 ; c  elliptic arc for shroud
                                   c  periodic boundary
ld 8 lep 1.9 1 2.1 1.1 200 270 0 ; c  elliptic arc for the shroud
ld 9 lp2 .54361 .776424 .54361 1;  c  line segment for cylinder
                                   c  for the shroud

curd 1 ld2d3d 6 cy y   5   0 ;     c  press. hub period. boundary
curd 2 ld2d3d 6 cy y   45  40 ;    c  suct. hub periodic boundary
curd 3 ld2d3d 2 cy y   %theta  41; c  press. side of blade at hub
curd 4 ld2d3d 2 cy y [40-%theta] 79; c suct. side of blade at hub
curd 5 ld2d3d 7 cy y        5   0 mz .25;
                         c  pressure shroud periodic boundary
curd 6 ld2d3d 7 cy y       45  40 mz .25;
                         c  suction shroud periodic boundary
curd 7 ld2d3d 3 cy y   %theta  41 mz .25;
                              c  pressure side of blade at shroud
curd 8 ld2d3d 3 cy y [40-%theta] 79 mz .25;
                              c  suction side of blade at shroud

sd 1 crz 1;                             c  hub
sd 2 cn2p %x1 0 %z1 [%x2-%x1] 0 [%z2-%z1]
                              c  leading edge cone at first blade
 %rho1 0 %rho2 [%z2-%z1]
sd 3 crz 4                              c  cylinder for the hub
```

```
sd 4 crz 5                                c  outlet cylinder
sd 5 cr 0 0 .25 0 0 1 8                   c  shroud
sd 6 crz 9                           c  cylinder for the shroud
sd 7 cn2p %x3 %y3 %z1     c  leading edge cone at second blade
 [%x4-%x3] [%y4-%y3] [%z2-%z1] %rho1 0 %rho2 [%z2-%z1]
```

Shown below is the 2D curve number 1 defined as an elliptic arc. This 2D curve is swept in the angular direction to form base curves for the blades. It is also used to construct the hub surface.



Figure 103 2D Curve #1 - elliptic arc



Figure 102 7 Surfaces For The Impeller

Seven surfaces are used to construct the impeller fluid. The figure 102 shows the outlines of these surfaces. Surfaces 1 and 5 form the hub and shroud, respectively. Surface 4 is the outer cylinder interface to the outlet. The two cylinders 3 and 6 form the inlet extensions to the hub and shroud, respectively. The similar cones 2 and 7 form the leading edges of the two blades being modeled.

Eight 3D curves are used to initialize the mesh before projection onto the surfaces. Curves 3 and 7 form the hub and shroud edges on the pressure side of the blade, respectively. Curves 4 and 8 form the hub and shroud edges of the suction side of the blade, respectively. Curves 1, 5, 2, and 6 are the periodic extensions of these curves, respectively.



Figure 104 3D Curves

```
cylinder 1 5 22 25 67;1 23;1 12;;5 45;;
                                   c  initial cylindrical part

pb 1 1 1 1 2 1 xz  0.3679490 1           c  initialize key nodes
pb 1 1 2 1 2 2 xz  0.5436100 1                c  by extracting
pb 2 1 2 2 2 2 xz  0.5436100 0.7764240        c  coordinates from
pb 2 1 1 2 2 1 xz  0.3679490 0.6              c  3D curves and
pb 3 1 1 3 1 1 xyz 0.6857862 0   0.3928933    c  surfaces using
pb 3 1 2 3 1 2 xyz 0.8787036 0   0.5839555         c  pbs command
pb 3 2 1 3 2 1 xyz 0.6857862 40        0.3928933
pb 3 2 2 3 2 2 xyz 0.8787035 40        0.5839555
pb 4 1 2 4 1 2 xyz 0.8787035 0.25      0.5839555
pb 4 1 1 4 1 1 xyz 0.6857862 0.25      0.3928933
pb 4 2 1 4 2 1 xyz 0.6857862 39.75     0.3928933
pb 4 2 2 4 2 2 xyz 0.8787035 39.75     0.5839555
pb 5 1 2 5 1 2 xyz 2.1       41        0.35
```

```
pb 5 1 1 5 1 1 xyz 2.1       41        0.1
pb 5 2 1 5 2 1 xyz 2.1       79        0.1
pb 5 2 2 5 2 2 xyz 2.0983422 78.948029 0.35

cur 4 1 1 5 1 1 3                       c  initialize edges of
cur 4 2 1 5 2 1 4                       c  mesh to be along
cur 4 2 2 5 2 2 8                       c  3D curves
cur 4 1 2 5 1 2 7
cur 2 1 1 3 1 1 1
cur 2 2 1 3 2 1 2
cur 2 1 2 3 1 2 5
cur 2 2 2 3 2 2 6

sfi 2 5; ;-1;sd 1                       c  project faces of the
sfi 3 4;-1; ;sd 2                       c  mesh onto surfaces
sfi 3 4;-2; ;sd 7
sfi 1 2; ;-1;sd 3
sfi 1 2; ;-2;sd 6
sfi -5; ; ;sd 4
sfi 2 5; ;-2;sd 5

drs 1 1 1 5 2 2 k 1.5 1.5               c  clustering of nodes
drs 1 1 1 5 2 2 j 1.4 1.4               c  near the boundary
drs 4 1 1 5 2 2 i 1.15 1.195            c  layers
res 2 1 1 3 2 2 i .825

tmei 2 5;;-1 0 -2;50 0.00001 1          c  smooth the mesh
```

5. Male and Female Sprocket from IGES file

Introduction

The geometry for this male sprocket is in the IGES format. The IGES file contains 58 trimmed surfaces. There are 9 teeth to this sprocket. The 9 teeth are similar and the representative tooth has a symmetry plane. Only half of the representative tooth will be modeled and then replicated to create a full model. The full model is needed to apply non-symmetric boundary conditions .

There is a filleted region consisting of three surfaces. To simplify the projection of the mesh onto this geometry, the three surfaces will be combined to form one surface of projection. This is the standard way to deal with fillets. The geometry will remain unchanged with this operation.



Figure 105 Imported IGES Surfaces



Figure 106 Part 1 before Projections



Figure 107 Part 1 after Projections



Figure 108 Edge before Attachment to Curve



Figure 109 Edge after Attachment to Curve

The edge of the mesh that ideally should follow the center line of the fillet may not project to that position because of its poor initial position. This is easily fixed by constructing a 3D curve along the center line of the fillet and then attaching the center line edge of the mesh to the center line 3D curve. This edge of the mesh is free to move along the filleted surface because the faces of the mesh, including the edge in question, are only constrained to lie upon the one union of surfaces.

The block boundary command (**bb** and **trbb**) will be used to glue four parts together to form one model. Transition elements will be generated to minimize the total number of elements while keeping a high resolution mesh in the area with the largest gradients. A normal offset will be used to separate the male and female components so that initially the contact surfaces will be apart. The gap will be exaggerated for demonstration purposes. This gap can be easily decreased by changing one number.

When building a multiple part model using the Transition Block Boundary command (**trbb**), it is important to plan the order of the parts. This is because the second reference (the **trbb** side) of a transition interface receives the transition elements. In this problem there will be several transition regions and transition regions cannot overlap or intersect at corners. Therefore, the inner part must be generated first. The second, outer part will receive all of the transition elements.

The male parts will be generated first, since we have only the geometry for the male sprocket. Then the Block Boundary interface with a normal offset will be created. This offset interface will be used to define the shape of the female parts.

Offsets are geometrically difficult because the offset surfaces of two connected surfaces may not be connected or may overlap. The offset contact surface for this model has both of these problems. The way to resolve this problem is to create one interface and offset that one interface. Then the corners will be handled properly.

The connected faces of the male sprocket will be combined into one interface by creating a shell part for construction purposes only. Each region of the shell part will be glued to the individual interfaces of the male sprocket part. A new interface will be defined using the whole shell part. This is the standard way to combine several interfaces into one. Then the shell part is ignored by setting the material of the shell part to 0.

A second shell part, for construction purposes only, is needed to create the single offset interface. This is then split up into the individual interfaces that will be needed for the female part. Again, the shell part material is set to zero so that none of the elements are added to the element data base.



Figure 110 Finely meshed parts



Figure 111 Coarser parts with transitions

There are three transition block boundary interfaces between parts 1 and 2. Note that the transition regions of these interfaces do not intersect. If the transitions were in part 1, instead, then the shape of the transition elements would be more complex at the corners where the transition regions would intersect. This is not allowed. The same can be said with respect to parts 5 and 6.



Figure 112 BB interfaces in part 1

Figure 113 BB interfaces in part 2

Block boundaries 7, 8, 9, 10, and 11 are gathered from parts 1 and 2. They are combined in part 3. A new block boundary 12 is defined covering the entire part 3. The inverse of this splits this new block boundary, within part 4, into component offset interfaces 14, 15, 16, 17, and 18 to be used in parts 5 and 6. This creates the normal offset gap.



Figure 114 Composite BB 12



Figure 115 Offset interfaces based on BB 12

The four parts of the mesh are replicated using the **lev** (which stands for level), **pslv** (which stands for push level onto the transformation stack) and **pplv** command (which stands for pop level off of the transformation stack). The **lev** command defines the transformations to be applied to a set of parts to be replicated. The **pslv** and **pplv** mark the scope of this replication. The **pslv** command is placed in the input just before the first part to be replicated. The **pplv** command is placed after the **endpart** of the last part to be replicated. Levels of replications can be nested. This is the reason they are numbered.



Figure 116 Mesh after Replication

Displacement constraints are defined using the **b** command. Most FEM models require some nodes to be constrained. When in the merge phase, the nodes with constraints can be marked in the picture using the **co** command with either the **dx**, **dy**, **dz**, **rx**, **ry**, or **rz** option. It is wise to check all of the constraints and conditions on the mesh before writing the formatted data to disk.



Figure 117 Displacement Constraints



Figure 118 Master Slide Interface 1



Figure 119 Master Slide Interface 2

Every replication creates a set of duplicate nodes at the interface between the duplicate parts. These nodes are merged out in the merge phase. However, nodes in the slave side of a sliding interface will not be merged with nodes on the master side of the same sliding interface. It is best to check the nodes that have been merged with either **labels tol** or **labels fraces**. In these pictures, the master and slave sides of the sliding interfaces are shown using the **co si** command.



Figure 120 Slave Slide Interface 1



Figure 121 Slave Slide Interface 2

The command file follows:

```
useiges sprocket.bin  c Preprocessed binary IGES file is used.
c Original iges file is used for reading of  additional info
iges sprocket.igs 1 1 ;

c Definition of the combined surface in the filleted region:
sd 120 sds 41 105 42;

c Definition of the extracted surface created from the IGES file.
C This surface is the untrimmed version of surface 105.
c It is needed to easily construct a surface curve.
sd 76 igess 18  ; c the sequence number in the IGES file is 18.

c 3D curves are created from the edges of the surfaces:
curd 1 sdedge 42.1    c edge 1 of the surface 42,
curd 2 sdedge 42.3    c edge 3 of the surface 42,
curd 3 sdedge 93.1    c edge 1 of the surface 93,
curd 4 sdedge 93.3    c edge 3 of the surface 93.

c 3D curves by interpolation between two existing 3D curves:
curd 5 intcur 1 -2 .5  ; c curve 1 and reversed  curve 2,
curd 6 intcur 3 -4 .5  ; c curve 3 and reversed  curve 4,

c  3D curve on the axis of symmetry, created as a segment:
curd 7 lp3 0 0 0 0 0 -2.1; ;    c   (0,0,0) and (0,0,-2.1).

c 3D curve on the small arc is created as a contour curve
curd 8 contour 76.3.1 76.3.0;   c of the surface 76.

c Definition of surface 121 - plane by 3 points (pl3)
sd 121 pl3
     rt 0.0000000e+00 0.0000000e+00  0.0000000e+00
     rt 0.0000000e+00 0.0000000e+00 -3.0000000e+00
     rt 6.3799059e-01 3.6834404e-01 -6.0000000e+00

c Definition of surface 122 - plane by 3 points (pl3)
sd 122 pl3
     rt 0.0000000e+00 0.0000000e+00 0.0000000e+00
     rt 0.0000000e+00 0.0000000e+00 -3.0000000e+00
     rt 6.0325468e-01 1.0637008e-01 -1.7881393e-07

c plane by point and normal (plan):
sd 123 plan 0 0 0 0 0 1 c point (0,0,0) and normal (0,0,1)
sd 124 plan 0 0 -2.1 0 0 1 c point (0,0,-2.1) and normal (0,0,1).
```

```
c sliding interface 1 - sliding interface with void
sid 1 sv pnlt .1 fric .01 kfric .02  ; ;
 c penalty pnlt=.1, static friction is fric=.01,
 c kinematic friction is kfric=.02.

c sliding interface 2 - sliding interface with void
sid 2 sv pnlt .1 fric .01 kfric .02  ; ;
 c penalty pnlt=.1, static friction is fric=.01,
 c kinematic friction is kfric=.02

c DYNA3D material 1
dynamats 1 3 e 2.1e11 pr .3 sigy 200e6 etan 3.e11 beta .7;
 c Type Elastic/Plastic (3)
 c Young's Modulus is e=2.1e11, Poisson's Ratio is pr=.3.
 c Yield Stress is sigy=200.e6. Tangent Modulus is etan=3.e11.
 c Hardening parameter is beta=.7.

c replication transformations
lev 1 levct 1 mz -.7;;
c Level set 1, move in the z-direction -.7.

lev 2 levct 1 rz -30 rxz rz 30;grep 0;;
c Level set 2, rotate around z-axis -30 degrees,
c reflect around xz plane and rotate around z-axis 30 degrees.

lev 3 levct 8 rz 40;repe 8;grep 0;;
c Transformation number 3, 8 times, rotate around z-axis 40
c degrees.  Preserve the original part.

pslv 3   c Invoke level 3 transformations.
pslv 2   c Invoke level 2 transformations.

c Part 1 - male sprocket - dense.
cylinder 1 4 13;1 4 7;1 12 23 34;
        .59 .65 .8;16 20 24;0 -.7 -1.4 -2.1;

dei 2 3; 1 2;;   c Deletion of the region.

c Block boundaries 1, 2, and 3 (masters for transition elements).
bb 1 3 1 3 3 4 1 ;
bb 1 1 1 1 3 4 2 ;
bb 1 1 1 2 1 4 3 ;

c Block boundaries 10, 9, and 8 (masters for temporary part).
bb 3 2 1 3 3 4 10;
bb 2 2 1 3 2 4 9;
bb 2 1 1 2 2 4 8;
```

```
c Projection of faces onto combined surface 120.
sfi 2 3; -2; 1 4;sd 120
sfi -2; 1 2; 1 4;sd 120
c Projection of faces onto surface 93.
sfi -3;;;sd 93

curs 2 2 1 2 2 4 8     c Edge of mesh is positioned on curve 8.

c Projections of outer faces (orthogonal to the z-axis)
sfi ;; -1;sd 123    c onto plane surfaces 123, 124.
sfi ;; -4;sd 124

c Linear interpolations are invoked along 2 edges and a face
lini 1 3; -3; -1 0 -4;    c Linear interpolations of the edges.
lini 1 3; -3;;    c Linear interpolations of the face.

c Edge is positioned for better shape of elements
pb 1 2 1 1 2 4 xy 5.90e-01 2.05E+01

c Orientation in cylindrical coordinates for sliding interface.
orpt + .85 16 0  rp 0

si 3 2 1 3 3 4 1 m ;    c sliding interface 1 - master side
si 2 2 1 3 2 4 2 m ;    c sliding interface 2 - master side
si 2 1 1 2 2 4 2 m ;

c Displacements dx,dy and dz are constrained on a face
bi  ; ;-1;dx 1 dy 1 dz 1 ;

mate 1     c Material 1 is assigned to Part 1.

endpart

c Part 2 - Male Sprocket - Coarse.
cylinder 1 4 5 8;1 3 4 5 7;1 12 23 34;
        .5 .59 .62 .75;10 16 20 24 30;0 -.7 -1.4 -2.1;

c Deletion of regions of the mesh.
dei 3 4; 1 3;;
dei 2 4; 2 4; 1 4;

c Slave Side of the transition block boundary number 1
trbb 2 4 1 4 4 4 1;

c Slave Side of the transition block boundary interface 2
trbb 2 2 1 2 4 4 2;
```

```
c Slave Side of the block boundary interface 3
trbb 2 2 1 3 2 4 3;

c Master Side of block boundary interface 11.
bb 4 4 1 4 5 4 11;

c Master Side of block boundary interface 7.
bb 3 1 1 3 2 4 7;

sfi -1;;;sd 88           c Projection onto inner cylinder.
sfi -3; 1 2; 1 4;sd 120     c Projection onto combined surface.
sfi -4; 4 5; 1 4;sd 93      c Projection onto top surface.
sfi 1 4; -5; 1 4;sd 121     c Projection onto plane surface.
sfi 1 3; -1; 1 4;sd 122     c Projection onto plane surface.
sfi ;; -1;sd 123         c Projection onto plane surface.
sfi ;; -4;sd 124         c Projection onto plane surface.

c Orientation in cylindrical coordinates for sliding interface.
orpt + .85 20 0

si 4 1 1 4 5 4 1 m ;  c sliding interface 1 - master side.
si 3 1 1 3 2 4 2 m ;  c sliding interface 2 - master side.

c Displacements are constrained on the face.
bi  ; ;-1;dx 1 dy 1 dz 1 ;

mate 2   c Assignment of material 2 to the whole part.

endpart

c Temporary Parts.
c Part 3 - shell - Initialized in the cylindrical coordinates.
cylinder -1;1 3 6 15 18 20;1 12 23 34;
        .75;10 16 20 20 24 30;0 -.7 -1.4 -2.1;

c Slave sides of the block boundaries for the temporary part:
bb 1 3 1 1 4 4 9;   c  block boundary number 9
bb 1 4 1 1 5 4 10;  c  block boundary number 10
bb 1 5 1 1 6 4 11;  c  block boundary number 11
bb 1 2 1 1 3 4 8;   c  block boundary number 8
bb 1 1 1 1 2 4 7;   c  block boundary number 7

c Block boundary  12 - master for temporary part.
bb 1 1 1 1 6 4 12;

mate 0 ;  c Temporary part is discarded using material 0.
```

```
c Temporary Part
c Part 4 - shell. Part is initialized in Cartesian coordinates.
block -1;1 3 6 15 18 20;1 12 23 34;
.75;
 [.75*sin(10)] [.75*sin(16)] [.75*sin(20)]
 [.75*sin(20)] [.75*sin(24)] [.75*sin(30)];
0 -.7 -1.4 -2.1;

orpt off     c Orientation Point is turned off (Default
orientation is used)

c Block boundary  12 - slave for normal operator.
bb 1 1 1 1 6 4 12 normal -.005;

c Master sides of the block boundaries for the temporary part:
bb 1 5 1 1 6 4 18;   c block boundary number 18
bb 1 4 1 1 5 4 17;   c block boundary number 17
bb 1 3 1 1 4 4 16;   c block boundary number 16
bb 1 2 1 1 3 4 15;   c block boundary number 15
bb 1 1 1 1 2 4 14;   c block boundary number 14

mate 0 ;   c Temporary part is discarded using material 0.

endpart

pslv 1   c Invoke level 1 transformations.

c Part 5 - Female sprocket - dense.
cylinder 1 10 13;1 4 7;1 12 23 34;
        .65 .74 .79;16 20 24;0 -.7 -1.4 -2.1;

dei 1 2; 2 3; ;    c Deletion of regions of the mesh.

c block boundary number 15 with a normal offset.
bb 1 1 1 1 2 4 15;

c block boundary number 16 with a normal offset.
bb 1 2 1 2 2 4 16;

c block boundary interface 17 with a normal offset.
bb 2 2 1 2 3 4 17;

c Master sides of the transition block boundaries:
bb 1 1 1 3 1 4 4;      c Block boundary number 4.
bb 3 1 1 3 3 4 5;      c Block boundary number 5.
```

```
bb 2 3 1 3 3 4 6;      c Block boundary number 6.

sfi ;; -1;sd 123    c Projection onto plane surface.
sfi ;; -4;sd 124    c Projection onto plane surface.

c Orientation in cylindrical coordinates for sliding interface.
orpt + .55 25 0

si 1 1 1 1 2 4 2 s ;  c sliding interface 2 - slave side.
si 1 2 1 2 2 4 2 s ;  c sliding interface 2 - slave side.
si 2 2 1 2 3 4 1 s ;  c sliding interface 1 - slave side.

c Prescribed rotation velocity around z-axis at rot=600,
rotation 0 0 0 0 0 600

mate 3       c Assignment of material 3 to the whole part.

endpart

c Part 6 - Female sprocket - coarse.
cylinder 1 4 5 8;1 3 4 5 7;1 12 23 34;
        .63 .75 .79 .93;10 16 20 24 30;0 -.7 -1.4 -2.1;

dei 1 2; 3 5;;       c Region of the mesh was deleted.

dei 1 3; 2 4;;       c Region of the mesh was deleted.

c Slave Side of the interface block boundary interface 4.
trbb 1 2 1 3 2 4 4;

c Slave Side of the transition block boundary interface 5.
trbb 3 2 1 3 4 4 5;

c Slave Side of the transition block boundary interface 6.
trbb 2 4 1 3 4 4 6;

c Slave sides of the block boundaries with normal offset:
bb 1 1 1 1 2 4 14;   c Block boundary number 14.
bb 2 4 1 2 5 4 18;   c Block boundary number 18.

sfi 1 4; -1; 1 4;sd 122      c Projection onto plane surface.
sfi 2 4; -5; 1 4;sd 121      c Projection onto plane surface.
sfi ;; -1;sd 123      c Projection onto plane surface.
sfi ;; -4;sd 124      c Projection onto plane surface.

c Orientation cylindrical coordinates for sliding interface.
```

```
orpt + .55 20 0

si 1 1 1 1 2 4 2 s ; c sliding interface 2 - slave side.
si 2 4 1 2 5 4 1 s ; c sliding interface 1 - slave side.

c Rotation velocity around z-axis rot=600 is prescribed for part.
rotation 0 0 0 0 0 600

mate 4      c Assignment of material 4 to the whole part.

endpart

pplv    c End of transformation level
pplv    c End of transformation
pplv    c End of transformation

merge

stp .001   c Duplicate nodes are merged with the tolerance .001.
```

6. Fluid About a Ships Hull

In this example, a fluid mesh is constructed about the cross section of the hull of a ship. For some fluid simulations it is important to create a thin nearly orthogonal mesh near the hull. The aspect rations of the elements are not important. This example demonstrates the use of the Block Boundary interface used within a part (intra-part **bb** commands). Both the master and the slave side of the interface are defined in the same part. The option to offset the slave in the normal direction from the master produces the desired boundary layer.



Figure 122 Master and Slave sides



Figure 123 Concave region

There are several things to consider when using this method. Since the slave nodes will be frozen to the position that is normal to the master side, it is not possible to extend this interface from the port side water line to the starboard water line. If this were done, then the extreme port and starboard boundary layer nodes would be normal to the hull instead of at the water line.

For a better mesh, cluster the nodes toward the keel just a little so that there are more nodes in the area of greatest curvature. There is a limit to the boundary layer thickness, based on the curvature of the hull. In this example, if the boundary layer thickness was much greater, the nodes of the slave side would collide in the regions of greatest concavity. This concavity is the difficult part of the mesh. The boundary layer only postpones dealing with the concavity until further into the mesh. It still has to be dealt with. The mesh is relaxed only a little to improve the quality in the concave regions. Anymore relaxation, and the outer region mesh lines would pull away from the boundary layer in the concave regions.



Figure 124 Boundary layer fluid mesh about a cross section of the hull of a ship

```
TITLE  Fluid mesh cross section about a ship

c  4 similar curves define the ships hull cross section
curd 1 csp3 00 -1.6 1 0 -1.4 .1 0 -.4 -.5 0
               -.15 -.6 0 -.05 -.74 0 0 -.75 0;;
curd 2 csp3 00 -1.6 1 0 -1.4 .1 0 -.4 -.5 0
               -.15 -.6 0 -.05 -.74 0 0 -.75 0;ryz;
curd 3 csp3 00 -1.6 1 1 -1.4 .1 1 -.4 -.5 1
```

```
                -.15 -.6 1 -.05 -.74 1 0 -.75 1;;
curd 4 csp3 00 -1.6 1 1 -1.4 .1 1 -.4 -.5 1
                -.15 -.6 1 -.05 -.74 1 0 -.75 1;ryz;

sd 1 plan -1.6 1 0 0 1 0  c  water line
sd 2 rule3d 1 3;  c  starboard side of the hull
sd 3 rule3d 2 4;  c  port side of the hull
sd 4 plan 1.65 1 0 1 0 0  c  boundary layer at port water
sd 5 plan -1.65 1 0 1 0 0  c  boundary layer at starboard water
sd 6 cy 0 1 0 0 0 1 6  c  outer boundary

c  A cylindrical part closely resembles the final shape.
c  This makes it easier to work with - a BLOCK part also work.
cylinder  1 6 31;1 11 46 81 91;1 11;
          1.6 1.65 5 -200 -180 -90 0 20 0 1

c  Position the boundary layer at startboard water line
pb 3 2 1 3 2 2 xy 6 -165
c  Position the boundary layer at port water line
pb 3 4 1 3 4 2 xy 6 -15

c  Cluster the nodes toward the keel
res 1 3 1 1 4 2 j 1.05
res 1 2 1 1 3 2 j [1/1.05]
c  Cluster the boundary layer nodes toward the hull.
c  The elements next to the hull are .005
as 1 1 1 2 5 2 i 0 .005
c Cluster the nodes in outer region toward the boundary layer.
as 2 1 1 3 5 2 i 0 .04

sfi 1 3; -5; 1 2;sd 1  c  Project to water line on starboard side
sfi 1 3; -1; 1 2;sd 1  c  Project to water line on port side
sfi -1; 1 3; 1 2;sd 2  c  Project to starboard side of hull
sfi -1; 3 5; 1 2;sd 3  c  Project to port side of hull
sfi -2; -5; 1 2;sd 4   c  Project to port boundary layer
sfi -2; -1; 1 2;sd 5   c  Project to starboard boundary layer
sfi -3;;;sd 6          c  Project to outer boundary

c  Define the hull face of the fluid as the master side of the
c  intra-part Block Boundary Interface.
bb 1 2 1 1 4 2 1;
c  Attach normally offset boundary layer to master side
c  of the intra-part Block Boundary Interface with a normal
c  offset of .05.
bb 2 2 1 2 4 2 1 normal .05;
```

```
c  Interpolate the faces in the boundary layer region
tfi 2 3; 1 5; -1 0 -2;

c  Relax the mesh in the outer region just a little to improve
c the quality of the mesh in the region of greatest concavity
c  in the boundary layer.
tmei 1 2;;-1 0 -2;10 .00001 1

merge
```

XII. Recipes

1. Nodes.

Parameterization of nodes

You may need to connect one part of the mesh with another part using a beam element. If it is sufficient to use the closest node to a point, then you can use the **ajnp** command to locate that node. This is a parametric method, which will work with any number of nodes in your part.

The position of a node can be parameterized using the **parameter** and **ajnp** commands. The following example shows variable positioning of the closest node to a point based on a variable number of generated beams. There are 2 beams (**Figure 125**) and 15 beams (**Figure 126**) generated between the fixed first and last node. The command file follows after the figures.



Figure 125 Parameterization by ajnp



Figure 126 Parameterization by ajnp

```
merge
para beams_total 2;
 c Define a parameter for a variable number of beams.
 c Figure1:   beams_total = 2.
 c Figure 2:  beams_total = 15.

bm rt1 0 0 0 ; rt2 1 1 0 ; orient 0 1 2 mate 1 cs 1
 nbms %beams_total ;
 c Define a string of beams from the coordinates
 c of the first node (0,0,0) to the last node(1,1,0).
 c The orientation of beam cross section axes is given
 c by the coordinates (0,1,2).

ajnp .5 .5 .5
 c Find the node which is closest to the point (.5,.5,.5).
 c The number of this node is stored in the system
 c parameter (or variable), node.

para node2 %node;
 c Parameter node2 is set to the value of this system
 c parameter.

bm rt1 .5 .5 .5 ; n2 %node2 orient 0 1 2 mate 1 cs 1 nbms 2 ;
 c Define a string of beams from the coordinates of the
 c first node (.5,.5,.5) and the number of the last node,
 c (as determined by ajnp).
 c The orientation of beam cross section axes is given by the
 c coordinates (0,1,2).
 c 2 beams are generated.

labels nodes
 c Nodes are labeled.
```

2. Beams

The recipes for beams are grouped according to simulation code type.

DYNA3D beams

The input for DYNA3D beams consists of the following steps (in any order):

1. Optionally define a beam integration rule using the **bind** command.

2. Define a cross section using the **bsd** command.
> **2.1 Optionally** select thickness or cross section properties.
> **2.2** Be sure that the element type in the material definition agrees with the options in **bsd**.

3. Select a material with the beam element type using the **dynamats** command.
> **3.1 Optionally** select default thickness or cross section properties.
> **3.2 Optionally** select the beam integration rule number defined by the **bind** command, or the beam cross section property number defined by the **bsd** command.

4. Generate beams
> **4.1** Generate beams using the **ibm**, **ibmi**, **jbm**, **jbmi**, **kbm**, **kbmi** in the Part Phase. Use the **mate** or **mt** commands with zero material number in order to cause **True*Grid***® to ignore brick and shell elements present in the model. These command reference both the material and cross section definitions above.
>> **4.1.1 Optionally** select thickness or cross-sectional properties.
>
> **4.2** Generate beams using the **bm** command in the Merge Phase.
>> **4.2.1 Optionally** select thickness or cross-sectional properties. This command references both the material and cross section definitions above.

Definition of cross-sectional properties using the **bm**, **ibm**, **ibmi**, **jbm**, **jbmi**, **kbm**, **kbmi** overrides the ones specified using the **bsd** command. Definition of cross-sectional properties using the **bsd** command overrides the ones specified using the **dynamats** command.

Example

The beam integration rule is defined with the **bind** command. Two materials are specified using the **dynamats** command. The first one is for the Hughes-Liu beam formulation and it specifies thicknesses. The second material (material number 3) uses the Belytschko-Schwer beam formulation and utilizes cross-sectional properties (areas and moments of inertia). Beam cross section 1 is created with the **bsd** command. Beams 1 and 2 are then created with the **ibm** command, material 3 and re-defined cross-sectional properties. Beams 3 and 4 are created with the **jbm** command, material 1 and cross section 1 (**Figure 127**). Beams 5, 6, 7 and 8 are created with the **kbm** command, material 1 and cross section 1. Brick elements are ignored by setting the material number to 0. Finally, beams 9 and 10 are created with the **bm** command using material 1 and cross section 1 and redefining thicknesses.



Figure 127 beams for DYNA3D

```
bind 1 -.33 .7 .3 .5 0. .4 .2 -.1 .3;
     c Define integration rule 1.
     c The s and t (beam) coordinates of 3 integration points
     c and their weights w are:
     c    -.33 .7 .3          .5 0. .4       .2 -.1 .3

dynamats 1 1
  mhead beamat1
  rho 7850 beam elfom hl quad -1
  sthi .26 tthi .29 e 2.1e11 pr .3;
     c Define material 1 for dyna3d beams.
     c Name the material beamat1.
     c The density, rho, is 7850.
     c The type of beam is Hughes-Liu (beam elfom hl).
     c Integration rule:  -1 refers to rule 1
     c Cross section definition:
     c The thickness in s direction is .26 and
     c  the thickness in the t direction is .29.
     c Young's modulus is 2.1e11 and Poisson's Ratio is 0.3

dynamats 3 1
  mhead beamat3
  rho 7850 beam elfom bs shear 1.2 carea .04
  iss .005 itt .006 irr .0055 sarea .05
  e 2.1e11 pr .3 ;
     c Define material 3 for LS-DYNA beams.
     c Name the material beamat3.
     c The density, rho, is 7850.
     c The beam type is Belytschko-Schwer (beam elfom bs).
     c Cross section definition:
     c  Area, Moments of Inertia and Shear Area:
     c   Carea .04 Iss=.005 Itt=.006 Irr=.0055 Sarea=.05.
     c Young's modulus is 2.1e11 and Poisson's Ratio is 0.3.

bsd 1 sthi .26 tthi .29 ; ;
     c Cross section definition 1.
     c The thickness in the s direction is .26 and
     c the thickness in the t direction is .29

block 1 2;1 2;1 2;0 5; 0 5;0 5;
  c definition of structured block mesh

ibm 1 1 2 2 2 2 2 1 3 j csarea .04 sharea .0455
 inertia .022 .027 .02 0 ;
     c Define beams in the i-direction from the block part.
     c Region 1 1 2 2 2 2 is used to create 2 columns of beams
     c   in the j-direction and 1 column in the k-direction.
     c Material number 3 is assigned to the created beams.
     c Local beam cross-sectional axes point in the j-direction.
```

```
     c Cross-sectional area csarea is .04 and
     c  the shear area sharea is .0455.
     c Inertial moments are: Iss = .022, Itt=.027, Irr=.02.
     c A cross section 0 is not assigned.

jbm 1 1 2 2 2 2 2 1 1 i 1 ;
  c Define beams in the j-direction from the block part.
  c Region 1 1 2 2 2 2 is used to create 2 columns of beams in
  c   the i-direction and 1 column of beams in the k-direction.
  c Material 1 is assigned to the created beams.
  c Local beam cross-sectional axes point in the i-direction.
  c Cross section 1 is assigned to created beams 3 and 4.

kbm 1 1 1 2 2 2 2 2 1 i 1 ;
  c Define beams in the k-direction from the block part.
  c Region 1 1 1 2 2 2 is used to create 2 columns of beams in
  c   the i-direction and 2 column of beams in the j-direction.
  c Material 1 is assigned to the created beams.
  c Local beam cross-sectional axes point in the i-direction.
  c Cross section 1 is assigned to created beams 5,6,7, and 8.

mate 0
  c Brick elements are eliminated from the model by
  c assigning a global default material number of 0.

merge

bm n1 1 n2 4 n3 2 mate 1 cs 1 nbms 2 sthi .2 tthi .3  ;
  c Beams 9 and 10 are created from existing nodes 1 and 4.
  c Node 2 is used to orient the local cross-sectional
  c coordinate axes.
  c Material 1 and cross section 1 are assigned to the
  c created beams.
  c There are 2 beams.
  c The thickness in the s direction is .2 and the thickness
  c in the t direction is .3.

labels 1d
  c label beams
```

LS-DYNA Beams

There are 7 types of beam elements supported by **True*Grid*®** for LS-DYNA. Only certain types are allowed for each material type. Each type has a different set of cross-sectional properties. Check the LS-DYNA manual or use the dialogue boxes in **True*Grid*®** in order to avoid mismatches in the options.

The input for LS-DYNA beams consists of the following steps (in any order):

1. Optionally define a beam integration rule using the **bind** command.

2. Optionally define a cross section using the **bsd** command. Do not define a **bind** and **bsd** by the same identifying number.

> **2.1 Optionally** select thickness or cross section properties. Be sure that the element type in the material definition agrees with the options in **bsd** or use the dialogue boxes.

3. Select a beam material element type using the **lsdymats** command.

> **3.1 Optionally** select thickness or cross section properties.
> **3.2 Optionally** select the beam integration rule number defined by the **bind** command, or the beam cross section property number defined by the **bsd** command.

4. Generate beams

> **4.1** Generate beams using the **ibm**, **ibmi**, **jbm**, **jbmi**, **kbm**, or **kbmi** commands in the Part Phase. Use **mate** or **mt** in order to set the material number to zero so that the model's brick and shell elements will be ignored. These command reference both the material and cross section definitions above.
>
> > **4.1.1 Optionally** select thickness or cross section properties.
>
> **4.2** Generate beams using the **bm** command in the Merge Phase.
>
> > **4.2.1 Optionally** select thickness or cross section properties. This command references both the material and cross section definitions above.

The definition of cross-sectional properties using the **bm**, **ibm**, **ibmi**, **jbm**, **jbmi**, **kbm**, or **kbmi** commands overrides the ones specified using the **bsd** command. If cross-sectional properties or thicknesses are not supplied in the commands above, then LS-DYNA will use the properties defined in the material model (**lsdymats** command).

Example

Two materials are specified using the **lsdymats** command. The first one is for the Belytschko-Schwer beam formulation. It uses cross-sectional properties (areas and moments of inertia). The second material is for the Hughes-Liu beam formulation. The cylinder part is then created. It is shaped by projections. Beam cross sections 1 and 2 are created with the **bsd** command for the T and W cross sections of Hughes-Liu beams, respectively. Beam cross section 3 is specified for the Belytschko-Schwer beam by cross-sectional values. Vertical beams along the exterior diameter are created with the **kbmi** command using material 2 and cross section 1. Vertical beams along the interior diameter are created with the **kbmi** command using material 2 and cross section 2. The angular beams and radial beams of the cap are created with the **jbmi** and **ibmi** commands, respectively, using material 1 and cross section 3 (**Figure 128**). Brick elements are then ignored by setting the material number to 0. Beams with material 1 are displayed in **Figure 128** and beams with material 2 are displayed in **Figure 130**.



Figure 128 Beams for LSDYNA3D



Figure 129 Beams with Material 1



Figure 130 Beams with Material 2

```
lsdymats 1 1 struct rho 5000 beam elfom bs carea .05
  iss .008 itt .006 irr .007 sarea .053 e 2.e11 pr .33 ;
  c Define material model 1.
  c The material type is elastic (1 struct).
  c The density, rho, is 5000.
  c Beam type is Belytschko-Schwer (beam elfom bs).
  c Cross section definition:
  c  Cross section Area, Moments of Inertia and Shear Area:
  c  Carea=.05 Iss=.008 Itt=.006 Irr=.007 Sarea=.053.
  c Young’s modulus is 2.e11 and Poisson’s Ratio is 0.3.

lsdymats 2 1 struct beam elfom hl e 2.e11 pr .33 ;
  c Define material model 2.
  c Material type is elastic (1 struct).
  c Beam type is Hughes-Liu (beam elfom hl).
  c Young’s modulus is 2.e11 and Poisson’s Ratio is 0.3.

cylinder 1 3;1 12;1 3;3 5;0 360;3.5 5
  c Define the cylinder structured block mesh.

sd 1 sp 0 0 0 8
sd 2 cy 0 0 0 0 0 5 5;;
sd 3 cy 0 0 0 0 0 5 3;
  c Surface definitions 1,2,3 (sphere, cylinder, cylinder)

mbi ;; -2; z 1.58901
```

```
  c Move the k=2 face closer to the sphere.

sfi ;; -2;sd 1
  c Project the k=2 face onto surface 1 (sphere).

sfi -2;;;sd 2
  c Project the i=2 face onto surface 2 (external cylinder).

sfi -1;;;sd 3
  c Project the i=1 face onto surface 3 (internal cylinder).

bsd 1 lsd 4 1 .3 1 .2 ; ;
  c Define cross section 1 for the Hughes-Liu beam.
  c Cross section type is a T (type 4).
  c Flange width is w=1, flange depth is d=1,
  c  flange thickness is tf=.3 and web thickness is tw=.2.

bsd 2 lsd 1 .5 .2 1 .2 ; ;
  c Define cross section 2 for the Hughes-Liu beam.
  c Cross section type is W (type 1).
  c Flange width is w=1, flange depth is d=1,
  c  flange thickness is tf=.3 and web thickness is tw=.2.

bsd 3 carea .33 iss .02 itt .04 irr .03 sarea .4 ; ;
  c Define cross section 3 for the Belytschko-Schwer beam.
  c The cross-sectional area is carea=.33,
  c the inertial moments are:  Iss=.02, Itt=.4, Irr=.03 and
  c the shear area is sarea=.4.

kbmi -2; ; ;1 12 2 i 1 ;
  c Vertical beams - external diameter.
  c Define beams in the k-direction from the block part.
  c The face i=2 is used to create 1 column of beams in
  c the i-direction and 12 columns of beams in the j-direction.
  c Material 2 is assigned to the beams.
  c Local beam cross-sectional axes point in the i-direction.
  c Cross section 1 is assigned to the created beams.

kbmi -1; ; ;1 12 2 i 2 ;
  c Vertical beams - internal diameter.
  c Define beams in the k-direction from the block part.
  c The face i=1 is used to create 1 column of beams in the
  c i-direction and 12 columns of beams in the j-direction.
  c Material 2 is assigned to the beams.
  c Local beam cross-sectional axes point in the i-direction.
  c Cross section 2 is assigned to the created beams.

jbmi ; ;-2;3 1 1 k 3 ;
  c Beams of cap - angular.
```

```
  c Define beams in the j-direction from the block part.
  c The face k=2 is used to create 3 columns of beams in the
  c i-direction and 1 column of beams in the k-direction.
  c Material 1 is assigned to the beams.
  c Local beam cross-sectional axes point in the k-direction.
  c Cross section 3 is assigned to created beams from 23
  c through 55.

ibmi ;;-2;12 1 1 k 3 ;
  c Beams of cap - radial.
  c Define beams in the i-direction from the block part.
  c The face k=2 is used to create 12 columns of beams in the
  c i-direction and 1 column of beams in the k-direction.
  c Material 1 is assigned to the beams.
  c Local beam cross-sectional axes point in the k-direction.
  c Cross section 3 is assigned to created beams from 1
  c through 22.

mate 0;
  c Brick elements are eliminated from the model by assigning
  c material type 0.

merge

delem lbm 23 24;
  c Delete the 2 radial beams generated by the ibmi command
  c redundantly (the same position as beams 1 and 2).

dm 1
  c Display material 1.
labels 1d
  c Label the beams.

dm 2
  c Display material 1.
labels 1d
  c Label these beams.
```

Example

There are 2 materials defined using the **lsdymats** command. The first is a cable material and the second is an elastic material. Both materials also contain cross-sectional property definitions. The horizontal and vertical beams are defined with the **bm** command by defining the nodes at the beginning and the ending of each string of beams (**Figure 133**). They use the elastic material. The cable beams are created with the **bm** command along curves 1, 2, and 3, respectively (**Figure 132**).



Figure 131 Cable Beams



Figure 132 Cable Beams



Figure 133 Horizontal and Vertical Beams

```
lsdymats 1 71 beam elfom dis1 quad 1
     lump .3 cabarea .75 e 2.e11 ;
  c Define material 1. Type 71 is cable.
  c Type of beam is discrete cable (beam elfom dis1 quad1)
  c Lump inertia is lump=.3.
  c Cable cross-sectional area is cabarea=.75.

  c Young's Modulus is e=2.e11.

lsdymats 2 1 struct beam elfom bs
 carea .3 iss .2 itt .2 irr .2 sarea .4 e 2.e11 pr .3 ;
  c Define material 2. Type 1 struct is elastic structural.
  c Beam type is Belytschko-Schwer (beam elfom bs).
  c Beam cross-sectional area is carea=.3,
  c moments of inertia are Iss=.2 , Itt=.2, Irr=.2 and
  c shear area is sarea=.4.
  c Young's Modulus is e=2.e11 and Poisson's Ratio is pr=.3.

merge

curd 1 3dfunc -80 -50 u ;
     30*(cosh(.6*(u+80)/30)-1)/(cosh(.6)-1) ; 0 ; ;
  c Define curve 1 - Parametric Curve

curd 2 cpcd 1 ryz ;
  c Define curve 2 - Parametric Curve
```

```
curd 3 3dfunc -50 50 u ;
     30*(cosh(.6*u/50)-1)/(cosh(.6)-1) ; 0 ; ;
  c Define curve 3 - Parametric Curve

bm rt1 -80 0 0 ; rt2 80 0 0 ; v 0 0 1 mate 2 nbms 32 ;
  c Define horizontal beams by start and end coordinates.
  c Beam cross section is oriented by the vector (v 0 0 1).
  c Make material 2 the global default (mate 2) and
  c  construct 32 beams (nbms 32)

bm rt1 -50 -30 0 ; rt2 -50 30 0 ; v 0 0 1 mate 2 nbms 12 ;
  c Define vertical beams by start and end coordinates.
  c Beam cross section is oriented by the vector (v 0 0 1).
  c Make material 2 the global default (mate 2) and
  c  construct 12 beams (nbms 12)

bm rt1 50 -30 0 ; rt2 50 30 0 ; v 0 0 1 mate 2 nbms 12 ;
  c Define vertical beams by start and end coordinates.
  c Beam cross section is oriented by vector (v 0 0 1).
  c Make material 2 the global default (mate 2) and
  c  construct 12 beams (nbms 12)

bm n1 1 n2 35 v 0 0 1 mate 1  nbms 10 cur 1 ;
  c Define cable beams using existing
  c  start and end node numbers (n1 1 n2 35)
  c Beam cross section is oriented by vector (v 0 0 1).
  c Make material 2 the global default (mate 2) and
  c  construct 10 beams (nbms 10)
  c The beams are generated along curve 1 (cur 1).

bm n1 35 n2 18 v 0 0 1 mate 1  nbms 20 cur 3 ;
  c Define cable beams using existing
  c  start and end node numbers (n1 35 n2 18)
  c Beam cross section is oriented by vector (v 0 0 1).
  c Make material 1 the global default (mate 1) and
  c  construct 20 beams (nbms 20)
  c The beams are generated along curve 3 (cur 3).

bm n1 18 n2 48 v 0 0 1 mate 1  nbms 20 cur 3 ;
  c Define cable beams using existing
  c  start and end node numbers (n1 18 n2 48)
  c Beam cross section is oriented by vector (v 0 0 1).
  c Make material 1 the global default (mate 1) and
  c  construct 20 beams (nbms 20)
  c The beams are generated along curve 3 (cur 3).

bm n1 48 n2 2 v 0 0 1 mate 1  nbms 10 cur 2 ;
  c Define cable beams by existing
  c  start and end node numbers (n1 48 n2 2)
```

```
  c Beam cross section is oriented by vector (v 0 0 1).
  c Make material 1 the global default (mate 1) and
  c  construct 20 beams (nbms 20)
  c The beams are generated along curve 2 (cur 2).

dm 1            c Display material 1.
labels 1d       c Label beams.

dm 2            c Display material 1.
labels 1d       c Label beams.
```

Spot Weld beams for LS-DYNA

The **lsdymats** command is used to define material 1. A 2D curve, consisting of segments and fillets, is constructed in order to represent the profile of part 1 and is then turned into a 3D curve. This same 2D curve is then offset in order to create a second 3D curve and a ruled surface is created between these two 3D curves. We then use the *contour* option of the **curd** command to define 3D curves which follow the shape of this ruled surface. Finally, we create and shape part 1 by attaching its edges to the contour curves of the ruled surface (**Figure 134**). In order to combine this part with part 2, we first use the **si** command to specify a region of part 1 (the region closest to part 2) as the master side of a sliding interface. Then we use the **bb** command to define the master side of a block boundary in Part 1 (in order to specify proper positioning of the nodes of Part 2).

Having completed part 1, we again use the **block** command to define the block structure of part 2. We use the **si** and **bb** commands, respectively, to define the slave sides of the sliding and block boundary interfaces. We use the *normal* option to the **bb** command in order to position Part 2 a specified distance from Part 1. The spot weld beam elements are defined by the **bm** command from existing nodes. They connect shell elements of the master and slave sides of the sliding interface (**Figure 135**).



Figure 134 Shells Spot Welded



Figure 135 Spot Weld Beam

```
lsdymats 1 100 beam elfom spw e 2.e11 pr .3 es 150.e6 etan 4.e11;
  c Define material 1.
  c Material type is the spot weld (100 beam elfom spw).
  c Young’s Modulus is e=2.e11, Poisson’s Ratio is pr=.3
  c Yield stress is es=150.e6 and
  c  the hardening modulus is etan=4.e11.

merge

ld 1 lp2 0 0;
  lfil 0 35 -10 -60 5 ;
  lp2 35 -10;
  lfil -60 50 -20  180 5;
  lp2 50 -20;
  lfil 0 65 -10 240 5;
  lp2 65 -10;
  lfil 240 100 0 0 5;
  lp2 100 0;
  c Define 2d curve number 1. The curve is formed by the
  c  line segments (lp2) and fillets(lfil).

curd 1 ld2d3d 1 rt y 0 0 ;
  c Define 3d curve number 1.
  c The curve is created from the existing 2d curve number 1.

curd 2 cpcd 1 my 100;
  c Definition of 3d curve number 2.
  c The curve is created from the existing 3d curve number 1 by
  c  transformation (my 100).

sd 1 rule3d 1 2 ;
  c Definition of ruled surface number 1.
  c The curve is created from the existing 2d curve number 1.

curd 3  contour 1.1.3   1.7.3  ;
curd 4  contour 1.1.59  1.7.59  ;
curd 5  contour 1.1.71  1.7.71  ;
curd 6  contour 1.1.120 1.7.120  ;
curd 7  contour 1.1.180 1.7.180  ;
curd 8  contour 1.1.136 1.7.136  ;
curd 9  contour 1.1.198 1.7.198  ;
curd 10 contour 1.1.245 1.7.245  ;
  c Definitions of 3d curves number 3 to 10.
  c Curves are defined from the contours of surface 1.

block 1 5 6 10 11 15 16 20 21 25;1 20;-1;
  0 30 32 40 42 58 60 68 70 100;0 100;0;
  c Structured block mesh definition - Part 1.
```

```
sid 1 lsdsi 7 scoef .1 dcoef .01 ; ;
  c Definition of the type of sliding interface.
  c The type of the sliding interface is
  c Shell Edge tied to Shell Surface (lsdsi 7).
  c Static friction is fric=.1.

si 5 1 1 6 2 1 1 m 4 ;
  c Master side of sliding interface definition 1.

bb 5 1 1 6 2 1 1 ;
  c Master block boundary definition 1.
  c Nodes of master and slave region are matched
  c between Part 1 and Part 2

cure 2 1 1 2 2 1 3
cure 3 1 1 3 2 1 4
cure 4 1 1 4 2 1 5
cure 5 1 1 5 2 1 6
cure 6 1 1 6 2 1 8
cure 7 1 1 7 2 1 7
cure 8 1 1 8 2 1 9
cure 9 1 1 9 2 1 10
  c Edges of Part 1 are gradually placed on the 3d curves.
  c The curve numbers are from 3 to 10.

block 1 10 14 24;1 20;-1;0 42 58 100;0 100;-20.5;
  c Structured block mesh definition - Part 2.

si 2 1 1 3 2 1 1 s ; ;
  c Slave side of sliding interface definition 1.

bb 2 1 1 3 2 1 1 normal -.5 ;
  c Slave block boundary definition 1.
  c The nodes of the slave region are matched to the master
  c between Part 1 and Part 2 with a normal offset of .5.

merge

bsd 1 sthi 1.2 tthi 1.3 ;;
  c Definition of cross section number 1.
  c The thicknesses are sthi=1.2 and tthi=1.3

bm n1 224 n2 704 v 1 0 0 mate 1 cs 1 nbms 1 ;
bm n1 230 n2 710 v 1 0 0 mate 1 cs 1 nbms 1 ;
bm n1 236 n2 716 v 1 0 0 mate 1 cs 1 nbms 1 ;
bm n1 264 n2 744 v 1 0 0 mate 1 cs 1 nbms 1 ;
bm n1 270 n2 750 v 1 0 0 mate 1 cs 1 nbms 1 ;
bm n1 276 n2 756 v 1 0 0 mate 1 cs 1 nbms 1 ;
```

```
  c The spotweld beams are defined by the first nodes
  c  (n1=224,230,236,264,270,276) and by the last nodes
  c  (n2=704,710,716,744,750,756).
  c The orientation of the cross-sectional axes are determined
  c  by a vector v (v 1 0 0).
  c Material 1 and cross section 1 are used (mate 1 cs 1).
  c Number of beams is 1 (nbms 1).
```

```
labels 1d      c Beams are displayed and labeled

co or rst      c Local element axes are displayed
```

ANSYS beams

The input for ANSYS beams consists of the following steps (in any order):

1. Define a cross section using the **bsd** command.

2. Create a material model using the **ansymats** command.

3. Generate beams

3.1 Generate beams using the **ibm**, **ibmi**, **jbm**, **jbmi**, **kbm**, or **kbmi** commands in the Part Phase. Use the **mate** or **mt** commands with zero material number to ignore brick and shell elements present in the model. These command reference both the material and cross section definitions above.

3.2 Generate beams using the **bm** command in the Merge Phase.
This command references both the material and cross section definitions above.

Example

The 3D elastic beam material is specified using the **ansymats** command. Beam cross section 1 is created using the **bsd** command. Beams 29 through 52 are created with the **kbmi** command with the use of material 1 and cross section 1. Beams 1 and 14 are created using the **ibmi** command with the use of material 1 and cross section 1 (**Figure 136** and **Figure 137**). Beams 15 through 28 are created with the **jbmi** command with the use of material 1 and cross section 1. Brick elements are then "removed" by using the **mate** 0 command. Beams 53 and 54 (also 55 and 56) are created using the **bm** command with material 1 and cross section. This **bm** command starts and ends with existing nodes 6 and 24 (also 15 and 27). Beams 57 and 58 are created from existing node 25 and the newly created node with coordinates (0,-5,5). (The node number is 72.) Node 2 is used to orient the local cross-sectional coordinate system. Beams 59 and 60 (also 61 and 62) are created with the **bm** command using material 1 and cross section 1 from existing nodes 72 and 1 (also 72 and 4).



Figure 136 Ansys Beams



Figure 137 Ansys Beams

```
ansymats 1 STIF4 EX 2.1e11;NUXY .33;DENS 7860;KOP61 ;
  c Material definition 1 for 3D elastic beam stif4.
  c Young's modulus is Ex=2.1e11, Poisson's Ratio is Nuxy=.33 and
  c Density=7860.  Option kop61 causes the printout of member
  c forces and moments in the element coordinate system.

bsd 1 BAN4 AREA .05 IXX .006 IYY .0025 IZZ .003 THETA 45
 SHEARY 1.1 SHEARZ 1.1  ; ;
  c Cross section definition 1 for elastic beam.
  c Cross section area is Area=.05,
  c  moments of inertia are Ixx=.006, Iyy=.0025 and Izz .003.
  c Orientation angle of the cross section, theta, is 45 degrees.
  c Shear constants are Sheary=1.1 and Shearz=1.1.


block 1 3 5 7;1 3 5 7;1 3 5;-5 -2 2 5;-5 -2 2 5;1 3 5;
  c definition of the structured block mesh

dei 1 2 0 3 4; 1 2 0 3 4;;
dei 1 2 0 3 4; 2 3; 2 3;
dei 2 3; 1 2 0 3 4; 2 3;
  c Deletion of index progressions.

kbmi 1 3;1 3; ;3 3 1 i 1 ;
  c Definition of beams in the k-direction from the block part.
```

```
  c Index progression 1 3;1 3; ; is used to create 3 columns
  c  of beams in the i-direction and 3 column of beams in the
  c  j-direction.
  c Material 1 is assigned to created beams.
  c Local beam cross-sectional axes point in the i-direction.
  c Cross section 1 is assigned to created beams 29-52.

ibmi 1 3;1 3 ;-2 0 -3;3 2 1 k 1 ;
  c Definition of beams in the i-direction from the block part.
  c Index progression 1 3;1 3;-2 0 3; is used to create 3 columns
  c  of beams in the j-direction and 2 column of beams in the
  c  k-direction.
  c Material 1 is assigned to the created beams.
  c Local beam cross-sectional axes point in the k-direction.
  c Cross section 1 is assigned to created beams 1-14.

jbmi 1 3;1 3 ;-2 0 -3;3 2 1 k 1 ;
  c Definition of beams in the j-direction from the block part.
  c Index progression 1 3;1 3;-2 0 3; is used to create 3 columns
  c  of beams in the i-direction and 2 column of beams in the
  c  k-direction.
  c Material 1 is assigned to the created beams.
  c Local beam cross-sectional axes point in the k-direction.
  c Cross section 1 is assigned to created beams from 15-28.

mate 0
  c Brick elements are eliminated from the model by assigning
  c  material 0 as the default.

merge

bm n1 6 n2 24 n3 9 mate 1 cs 1 nbms 2 ;
  c Beams 53 and 54 are created from existing nodes 6 and 24.
  c Node 9 is used for orientation of the local cross-sectional
  c  coordinate system
  c Material 1 and cross section 1 is assigned to the beams.
  c There are 2 beams.
```

```
bm n1 15 n2 27 n3 18 mate 1 cs 1 nbms 2 ;
  c Beams 55 and 56 are created from  existing nodes 15 and 27.
  c Node 18 is used to orient the local cross-sectional
  c coordinate system.
  c Material 1 and cross section 1 is assigned to the beams.
  c There are 2 beams.

bm n1 25 rt2 0.0 -5.0 5.0 ;
   n3 2 mate 1 cs 1 nbms 2 ;
  c Beams 57 and 58 are created from  existing node 25 and
  c new created nodes with coordinates (0,-5,5).
  c (The new node's number is 72.)
  c Node 2 is used to orient the local cross-sectional
  c coordinate system.
  c Material 1 and cross section 1 is assigned to the beams.
  c There are 2 beams.

bm n1 72 n2 1 n3 4 mate 1 cs 1 nbms 2;
  c Beams 59 and 60 are created from  existing nodes 72 and 1.
  c Node 4 is used for orientation of the local
  c cross-sectional coordinate system
  c Material 1 and cross section 1 is assigned to the beams.
  c There are 2 beams.

bm n1 72 n2 4 n3 1 mate 1 cs 1 nbms 2;
  c Beams 61 and 62 are created from  existing nodes 72 and 4.
  c Node 1 is used for orientation of the local
  c cross-sectional coordinate system
  c Material 1 and cross section 1 is assigned
  c to created beams.
  c There are 2 beams.

labels 1d
  c label beams

labels nodes
  c label nodes
```

ABAQUS beams

The input for ABAQUS beams consists of the following steps (in any order):

1. Define a cross section using the **bsd** command.

2. Create a material model using the **abaqmats** command.

3. Generate the beams

3.1 Generate the beams using the **ibm**, **ibmi**, **jbm**, **jbmi**, **kbm**, or **kbmi** commands in the Part Phase. Use the **mate** or **mt** commands with zero material number to eliminate the brick and shell elements present in the model. These command reference both the material and cross section definitions above.

3.2 Generate beams using the **bm** command in the Merge Phase.
This command references both the material and cross section definitions above.

Example

The elastic isotropic material is specified using the **abaqmats** command. The block structured mesh is used as a starting shape for the model. The face j=1 is projected onto a parabolical ruled surface in order to form the arch. Nodes are spaced in a desired manner using the **drs** command for the double spacing ratio on the arches. Vertical beams 89 through 244 are created using the **jbmi** command with the use of material 1 and cross section 1 (**Figure 138** and **Figure 139**). X-direction beams 13 through 88 are created using the **ibmi** command with the use of material 1 and cross section 2 . Z-direction beams 245 through 346 are created using the **kbmi** command with the use of material 1 and cross section 3. Arch beams 1 through 12 and 61 through 72 are created using the **ibmi** command with the use of material 1 and cross section 4. Brick elements are ignored by using the **mate** 0 command. Beam cross section 1, 2, 3, and 4 are specified using the **bsd** command for rectangle cross section type. The beams are labeled and beam's local coordinate axes are displayed in **Figure 138** and **Figure 139**.



Figure 138 Beams for ABAQUS



Figure 139 Beams for ABAQUS

```
abaqmats 1 aqmpmf .1 aqspdf .1 aqdens 3000 aqelas
  aqelis 1.7e10 .15 ; ; ;
  c Definition of material 1
  c Mass proportional damping factor is aqmpmf=.1 and
  c  stiffness proportional damping factor is aqspdf=.1.
  c Density is aqdens=3000.
  c Elastic behavior is turned on by aqelas
  c  with the isotropic option aqelis.
  c Young's modulus=1.7e10 and Poisson's ratio=.15.

block 1   3   5   7 9 11 13 15 17;
        1  3  5  7; 1  3 5 7;
   -40 -30 -20 -10 0 10 20 30 40;
        0 10 20 30;-9 -3 3 9;
  c Definition of the structured block mesh.

curd 1 3dfunc -40 40 u ; -u*u/65+25 ; -9 ; ;
  c Definition of parametric 3D curve 1 - parabola.

curd 2 cpcd 1 mz 18;
  c Definition of parametric 3D curve 2 - parabola.

sd 1 rule3d 1 2 ;
  c Definition of surface 1.
  c Ruled surface is constructed from curves 1 and 2
```

```
sfi ; -1;;sd 1
  c projection onto the parabolic ruled surface

lini ;; -1 0 -4;
  c linear interpolation is invoked across 2 faces

lini 1 9; 1 4; 1 4;
  c linear interpolation is invoked for the whole volume

drs 1 1 1 9 1 4 i .94 .94
  c Nodes in the region 1 1 1 9 1 4 are distributed
  c using the double relative spacing in the i-direction.
  c It causes a more regular spacing of nodes on the arches.

jbmi 2 8; ;-1 0 -4;13 2 1 k 1 ;
  c Vertical beams:
  c Definition of beams in the j-direction from the block part.
  c Index progression 2 8; ;-1 0 -4; is used to create
  c 13 columns of beams in the i-direction and
  c 2 columns of beams in the k-direction.
  c Material 1 is assigned to the beams.
  c Local beam cross-sectional axes point in the k-direction.
  c Cross section 1 is assigned to created beams from 89-244.

ibmi ;-4; ;1 4 1 k 2 ;
  c x-direction beams:
  c Definition of beams in the i-direction from the block part.
  c Index progression  ;-4; ; is used to create
  c 1 column of beams in the j-direction and
  c 4 columns of beams in the k-direction.
  c Material 1 is assigned to the beams.
  c Local beam cross-sectional axes point in the k-direction.
  c Cross section 2 is assigned to the beams from 13-88.
```

```
kbmi ;-4; ;17 1 1 i 3 ;
  c z-direction beams:
  c Definition of beams in the k-direction from the block part.
  c Index progression  ;-4; ; is used to create
  c 17 columns of beams in the i-direction and
  c 1 column of beams in the j-direction.
  c Material 1 is assigned to the beams.
  c Local beam cross-sectional axes point in the i-direction.
  c Cross section 3 is assigned to the beams from 245-346.

ibmi 2 8;-1; ;1 2 1 k 4 ;
  c Arches:
  c Definition of beams in the i-direction from the block part.
  c Index progression 2 8;-1; ; is used to create
  c 1 column of beams in the j-direction and
  c 2 columns of beams in the k-direction.
  c Material 1 is assigned to the beams.
  c Local beam cross-sectional axes point in the k-direction.
  c Cross section 4 is assigned to the beams from 1-12 and 61-72.

mate 0
  c Brick elements are eliminated from the model
  c by assigning material 0.

merge

bsd 1 cstype 11 abcs1 1.5 abcs2 1.5 ; ;
bsd 2 cstype 11 abcs1 2 abcs2 1.5 ; ;
bsd 3 cstype 11 abcs1 1 abcs2 1.5 ; ;
bsd 4 cstype 11 abcs1 1 abcs2 1 ; ;
  c Cross section definitions 1 to 4 for elastic beam
  c Cross section type cstype 11 is a rectangular cross section.
  c The width(abcs1) and height(abcs2) are specified.


labels 1d
  c Beams are labeled.

co or rst
  c Local beam axes are displayed
```

3. Shells.

Shell Cylinder with Holes

The mesh of a cylindrical shell is created using the **cylinder** command. Shell regions are highlighted. The highlighted regions are deleted using the **dei** command. The command file follows:

```
c Cylinder part definition.
c  The index in the radial direction (i-dir.) is -1 (shell).
c  The radius of the cylinder is 18 units.
c  Indices in the angular direction (j-dir.) are:
c                              1 5 9 13 17 21 25 29 33 37.
c    Angles are: 0 50 80 120 160 200 240 280 310 360 degrees.
c  Indices in the z direction (k-dir.) are: 1 3 5 7 9.
c    z-coordinates are: -10 0 10 20 30 units.

cylinder -1;1 5 9 13 17 21 25 29 33 37; 1 3 5 7 9;
       18;0 50 80 120 160 200 240 280 310 360;-10 0 10 20 30;
c Deletion of two selected regions
dei -1; 3 4 0 6 7; 2 3 0 3 4;
```



Figure 140 physical mesh



Figure 141 computational mesh



Figure 142 Region for Deletion



Figure 143 Region for Deletion



Figure 144 Physical Mesh after Deletion



Figure 145 Computational Mesh after Deletion

Shell Bumper

A cross-sectional 2D curve of a car bumper is created using the **ld** command. The sweeping curve is also defined using the same **ld** command.



Figure 146 2D Curve to be Swept



Figure 147 Sweeping 2D Curve



Figure 148 Surface type Sweep



Figure 149 Contour Curves

The bumper surface is created using the **sd** command with the *sweep* option, using the cross-sectional 2D curve 1 and the sweeping 2D curve 2.

Eight 3D curves are then derived from this surface using the **curd** command with the *contour* option from the definition points of the surface 1.

Finally, the structured block part is defined using the **block** command.



Figure 150 Initial Physical Mesh



Figure 151 Edge to be Moved



Figure 152 Computational Window

The edge of the mesh is selected in the computational window. The **pb** command then positions this edge (region 2 1 1 2 8 1) at x=20 by moving the edge only in the x-direction. This is done approximately using the mouse with the move points panel, x-direction option selected. The dialogue box for the **pb** command will issue the exact coordinate.



Figure 153 Edge was Moved



Figure 154 Edge to be Moved



Figure 155 Computational Window

The edge at 4 1 1 4 8 1 is placed into position at z=10 using the same methods.

Interior contours of surface 1 are defined using the **curd** command. Then edges are selected in the computational window to be attached to these interior contour curves with the **curs** command. It is important to build these intermediate curves and attach the intermediate edges of the mesh to these curves. The curvature of this bumper is too great for the projection method to do a good job without the intermediate initializations.



Figure 156 Edge was Moved



Figure 157 Before Edge Attachment



Figure 158 Computational Window



Figure 159 After Edge Attachment



Figure 160 Edges Placed on Contour Curves



Figure 161 Mesh before Projection



Figure 162 Computational Window

The entire mesh is selected in the computational window. Then the mesh is projected onto surface 1 using the **sfi** command.

The mesh is reflected across two symmetry planes using the **lct** and **lrep** commands. In the merge phase, the duplicate nodes are merge using the **stp** command.



Figure 163 Mesh after Projection



Figure 164 Mesh after Replication

The command file follows:

```
merge

c Definition of 2D curve number 1. The curve is composed
c of segments and fillets.
ld 1
 lp2 4. -4.;     c Segment (lp2) starting at the point (4,-4).
 lfil 180. 0. -1. 90. 1.;
     c Fillet (lfil), start angle is eta=180 degrees
     c The end point is (0,-1), the end angle is
     c phi=90 degrees and the radius of the fillet is 1.
```

```
 lfil 90 0. -1. 180. .3;
     c Fillet (lfil), start angle is eta=90 degrees.
     c The end point is (0,-1), the end angle is phi=180
     c degrees and the radius of the fillet is .3.
 lfil 0. 1. 0. 90. .3;
     c Fillet (lfil), start angle is eta=0 degrees.
     c The end point is (1,0), the end angle is phi=90
     c degrees and the radius of the fillet is .3.
 lp2 1 0.;     c Segment (lp2) ending at the point (1,0).

lvs 1 1;       c Display 2D curve number 1.


c Definition of the 2D curve number 2. The curve is composed
c of a segment and fillet.
ld 2
 lp2 0. 0.;    c Segment (lp2) starting at the point (0,0).
 lfil 0 30. 10. 90. 10.;
     c Fillet (lfil), start angle, eta=0 degrees.
     c The end point is (30,10), the end angle, phi=90
     c degrees and the radius of the fillet is 10.

c Definition of the surface number 1.
sd 1 swept 2 l 1 0 1 1 ; ;
     c Surface is created by sweeping(swept) a 2D curve along
     c a trajectory which is formed by another 2D curve.
     c The sweeping is done along the curve number 2 to the
     c left (2 l). It starts with curve number 1, at the
     c parametric position 0 (1 0) and it ends with the curve
     c number 1 at the parametric position 1 (1 1).

c Definition of the 3D contour curves from the surface points
c of surface number 1.
curd 1 contour 1.0.1 1.1.1;
     c Notation 1.0.1 means surface 1(1),
     c last point in the i-direction(0) and first point
     c in the j-direction(1).
curd 2 contour 1.0.2  1.1.2 ;
curd 3 contour 1.0.24 1.1.24;
curd 4 contour 1.0.25 1.1.25;
curd 5 contour 1.0.36 1.1.36;
curd 6 contour 1.0.37 1.1.37;
curd 7 contour 1.0.48 1.1.48;
curd 8 contour 1.0.49 1.1.49;

c Structured block mesh definition.
block  1 30 40 50;1 5 7 10 12 20 26 38; -1;
```

```
       0 10 20 30;0 -1 -2 -3 -4 -5 -6 -7;0;

pb  2 1 1 2 8 1 x 20   c Edge is placed into position x=20.
pb  3 1 1 3 8 1 x 30   c Edge is placed into position x=30.
pb  4 1 1 4 8 1 z 10   c Edge is placed into position z=10.

curs 1 8 1 4 8 1 1    c Attachment of the edge to curve number 1
curs 1 7 1 4 7 1 2    c Attachment of the edge to curve number 2
curs 1 6 1 4 6 1 3    c Attachment of the edge to curve number 3
curs 1 5 1 4 5 1 4    c Attachment of the edge to curve number 4
curs 1 4 1 4 4 1 5    c Attachment of the edge to curve number 5
curs 1 3 1 4 3 1 6    c Attachment of the edge to curve number 6
curs 1 2 1 4 2 1 7    c Attachment of the edge to curve number 7
curs 1 1 1 4 1 1 8    c Attachment of the edge to curve number 8

c Projection of the entire mesh onto surface number 1.
sfi  ;; -1; sd 1

c Definition of 3 local transformations for the replication.
lct 3 rzx ; rzx ryz ; rzx ryz rzx ;
     c First transformation is reflection about the zx plane.
     c Second transformation is reflection about the zx and yz
     c   planes, successively.
     c Third transformation is reflection about the zx, yz,
     c   and xz planes, successively.

c Part 1 is replicated in the local coordinates using
c local transformations.
lrep 0 1 2 3;
     c The transformation 0 is the identity (include the
     c   original part).

merge

stp.001

disp          c Mesh is displayed in hidden line mode.
```

Shell I-Beam

3D curve number 1 is defined using the **curd** command. This curve is composed of two segments and one arc. 3D curve number 2 is derived from transformed copies (rotations) of 3D curve number 1. Curves 3 and 4 are also derived from curve 1 by transformations (rotation, translation, scaling). The curves were displayed using the **dcd** and **acd** commands. The blue scale in the picture was turned on using the **grid on** command. It is actually a 3D grid but can appear as a 2D grid by setting the angle of perspective to 0 with the **angle** command.

The structured block mesh that follows is defined using the **block** command.



Figure 165 3D curve 1



Figure 166 3D curve 2



Figure 167 3D curves 2 and 3



Figure 168 3D curves 2,3, and 4



Figure 169 Structured Block Mesh



Figure 170 Computational Window



Figure 171 Regions to Delete



Figure 172 Computational Window

This block structured shell mesh is defined with two negative end partitions in the k-direction for the flanges and the middle negative partition in the j-direction for the web. Two regions are highlighted in the computational window. These regions of the mesh are deleted using the **dei** command or by clicking on the delete button in the environment window.



Figure 173 Regions were Deleted



Figure 174 Edge and Curve to Attach



Figure 175 Computational Window

An edge of the mesh is highlighted in the computational window and then attached to curve 3 using the **curs** command. This can be done using the **curs** dialogue or the *attach* button in the environment window.



Figure 176 Edge and Curve were Attached



Figure 177 Edge and Curve to Attach



Figure 178 Computational Window

Another edge of the mesh is attached to curve 3 using the same methods.



Figure 179 Edge and Curve were Attached

All 4 edges of hole 1 are attached to curve 3.



Figure 180 Edges of Hole 1 and Curve 3 were Attached

All 4 edges of hole 2 are attached to curve 4.



Figure 181 Edges of Hole 2 and Curve 4 were Attached



Figure 182 Edge to be Moved



Figure 183 Computational Window

An edge of the mesh is selected in the computational window and moved in the x-direction in order to improve the quality of the elements around the corner of hole 2 using the **mbi** command.



Figure 184 Edge was Moved



Figure 185 Region to Relax



Figure 186 Computational Window

This shows the mesh after all four edges surrounding the hole were repositioned.

The quality of the mesh region is improved by the **equipotential relaxation method** using the **relaxi** command.



Figure 187 Region was Relaxed

The command file follows:

```
merge

c Definition of 3D curve number 1
c The curve is composed of 2 segments and 1 arc.
curd 1
 lp3 0 0 0;mx .4 ;
   c The segment starts at the point (0,0,0)
   c  and it is moved 0.4 units in the x direction.
 arc3 seqnc cy .2 0 0 cy .2 45 0 cy .2 90 0 my .2 mx .2 ;
   c The arc is given by the sequence (seqnc) of 3 points in
   c  the cylindrical (cy) coordinates. The coordinates of
   c  the 3 points (r,eta,z) are : (.2,0,0), (.2,45,0), and
   c  (.2,90,0). The arc is moved in the x and y direction
   c  for 0.2 units.
 lp3 0 .2 0;my .2 ;   c The segment ends at the point (0,0.2,0)
          c and it is moved in the x direction for 0.2 units.

c Definition of 3D curve number 2.
c The curve is composed of the rotated copies (cpcds)
c  of curve number 1.
curd 2
 cpcds 1;ry 0 ;
   c Copy of curve 1 is rotated 0 degrees around the y axis.
 cpcds 1;ry 180 ;
   c Copy of curve 1 is rotated 180 degrees around the y axis
 cpcds 1;ry 180 rx 180 ;
   c Copy of curve 1 is rotated 180 degrees around the y and x
   c  axes, consecutively.
 cpcds 1;rx 180 ;
   c Copy of curve 1 is rotated 0 degrees around the x axis.

c Definition of 3D curve number 3.
c The curve is composed of the rotated and moved copy (cpcds)
c  of the curve number 2.
curd 3 cpcds 2;rx 90 mx 1.5 ;

c Definition of 3D curve number 4.
c The curve is composed of the scaled, rotated and moved
c  copy (cpcds) of the curve number 2.
curd 4 cpcds 2;csca 1.3 mx 3.5 rx 90 ;

dacd      c Display all curves.

c Definition of the structured block part - shells.
c Shell at the index j=-3. Two shells at the index k=-1,-13.
```

```
block 1 5 9 13 21 25;1 -3 5;-1 5 9 -13;
          0 1 2 3 4 5;-.6 0 .6;-1 -.3 .3 1;

dei 2 3 0 4 5; -2; 2 3;  c Deletion of 2 regions.

curs 2 2 2 3 2 2 3       c Attachment of the edge to curve 3.
curs 2 2 2 2 2 3 3       c Attachment of the edge to curve 3.
curs 2 2 3 3 2 3 3       c Attachment of the edge to curve 3.
curs 3 2 2 3 2 3 3       c Attachment of the edge to curve 3.


curs 4 2 3 5 2 3 4       c Attachment of the edge to curve 4.
curs 5 2 2 5 2 3 4       c Attachment of the edge to curve 4.
curs 4 2 2 5 2 2 4       c Attachment of the edge to curve 4.
curs 4 2 2 4 2 3 4       c Attachment of the edge to curve 4.

c The edge is moved in the x-direction for -0.4 units.
mbi -4; 1 3; -4; x -0.4
c The edge is moved in the x-direction for  0.4 units.
mbi -5; 1 3; -4; x 0.4
c The edge is moved in the x-direction for -0.4 units.
mbi -4; 1 3; -1; x -0.4
c The edge is moved in the x-direction for  0.4 units.
mbi -5; 1 3; -1; x 0.4

c Equipotential Relaxation of the region.
relaxi 1 6; -2; 1 4; 30 0 1
  c Maximum number of iterations is 30.
  c Minimal change is 0 and the relaxation coefficient is 1.

disp           c Display the mesh
```

Solid Cylinder with Hole

This example demonstrates the use of cylinder part were everything is interpolated in cylindrical coordinates. The mesh of a cylinder with a hole is created using the **cylinder** command. A solid region is highlighted. The highlighted region is deleted using the **dei** command . The command file follows:

```
c Cylinder part definition.
c  Indices in the radial direction (i-dir.) are:  1 3 5 7 11.
c    Radii are:  6 9 12 15 18 units
c  Indices in the angular direction (j-dir.) are:
c                                  1 5 9 13 17 21 25 29 33 37.
c  Angles are:  0 50 80 120 160 200 240 280 310 360 degrees.
c  Indices in the z direction (k-dir.) are:  1 3 5 7 9.
c   z-coordinates are:  -10 0 10 20 30 units.

cylinder 1 3 5 7 11;1 5 9 13 17 21 25 29 33 37; 1 3 5 7 9;
6 9 12 15 18;0 50 80 120 160 200 240 280 310 360;-10 0 10 20 30;
dei 4 5;7 8;;      c Deletion of the solid region.
```



Figure 188 physical mesh



Figure 189 computational mesh



Figure 190 Region for Deletion



Figure 191 Region for Deletion



Figure 192 Physical Mesh after Deletion



Figure 193 Computational Mesh after Deletion

Full Solid Cylinder

This mesh of full cylinder demonstrates the use of the **butterfly mesh technique**, **spacing of nodes within a region** and **linear interpolation**. It is used to create a high quality mesh with nearly orthogonal elements.

We take advantage of the rotational symmetry and initially mesh one quarter of the cylinder. The structured block part is defined using the **block** command. The cylindrical part is not appropriate since the interior region would fall along the axis of symmetry. There are three partitions in the i and j-directions. The x and y-coordinates of these partitions are x=0,1,1 and y=0,1,1. (Note that this means that two pairs of faces of the mesh lie on top of one another and the volume between them is collapsed.). When the outer faces are projected to the cylinder, this condition will no longer exist. Every block will have a positive volume. This technique is used to minimize the number of steps needed to create a butterfly mesh. A corner block is deleted, leaving two disconnected edges which must end up at the same location on the cylinder, after projections, in order to fill the gap left by the deleted block. Since these two edges start out in the same location, they will end up projected to the same position on the cylinder, as required. This technique can be confusing to the beginner. It is important to understand this for it will lead to a greater understand in the use of **True*Grid*®**.

The cylindrical surface is defined using the **sd** command. The solid region of the mesh is highlighted in the computational mesh . The highlighted solid region 2 3;2 3;; is deleted using the **dei** command. This region was visible only in the computational mesh because it has two zero dimensions in the physical mesh.

Two faces of the mesh are highlighted for the projection. The mesh is projected onto the cylinder using the **sfi** command. Two coincident edges in physical mesh are projected to the same position on the cylinder. The two regions of physical mesh with zero dimensions before projection were expanded to the two solid regions.



Figure 194 Physical Mesh



Figure 195 Computational Mesh



Figure 196 Cylinder Surface and Mesh



Figure 197 Region to delete



Figure 198 Region was deleted



Figure 199 Faces for Projection

The replication transformations are defined using the **lct** command. The mesh is then replicated 4 times by rotation around the z-axis using the **lrep** command.



Figure 200 Faces before Projection



Figure 201 Mesh after Projection

```
block 1 3 7;1 3 7;1 5 9 13;0 1 1;0 1 1;-5 -2 1 4;
 c Structured Block Mesh Definition.
 c The partitions are at indices i=1,3,7 j=1,3,7 and k=1,5,9,13.
 c (The reduced indices are i=1,2,3 j=1,2,3 k=1,2,3,4.)
 c The x and y-coordinates of partitions are x=0,1,1 and y=0,1,1.
 c The x and y coordinates for the reduced indices i=3 and j=3
 c are the same. This ensures the coincident position of 2 edges
 c of the butterfly mesh. The coincident position of the 2 edges
 c is essential for the projection.

sd 1 cy 0 0 0 0 0 1 2.5
 c Surface Definition 1
 c Cylinder(cy), Point on the axis of rotation (0,0,0),
 c Vector of the axis of rotation (0,0,1) and radius (2.5).

c Region is deleted.
dei 2 3;2 3;;

c 2 Faces are projected onto the Cylinder Surface (sd 1)
c 2 Edges of the physical butterfly mesh, which are coincident
c before projection are coincident also after the projection.
sfi 1 -3; 1 -3;;sd 1
lct 4 rz 0; rz 90; rz 180; rz 270;
 c Definition of the local transformation.
 c The 4 consecutive rotations are defined
 c (for 0, 90, 180 and 270 degrees around the local z-axis).

lrep 1 2 3 4;
 c The part is rotated for 0, 90, 180 and 270 degrees.

merge

stp 0.001 c Duplicate nodes are merged with tolerance .001 units
```

Half Solid Sphere

This mesh of a half sphere demonstrates the use of the **butterfly mesh technique**. It is used to create a high quality mesh with nearly orthogonal elements.

We take advantage of the rotational symmetry and initially mesh one quarter of the hemisphere. The structured block part is defined using the **block** command. There are three partitions in the i, j, and k-directions. The x, y, and z-coordinates of these partitions are x=0,1,1, y=0,1,1, and z=0,1,1. (Note that this means that three pairs of faces of the mesh lie on top of one another and the volume between them is collapsed.). When the outer faces are projected to the sphere, this condition will no longer exist. Every block will have a positive volume. This technique is used to minimize the number of steps needed to create a butterfly mesh. Corner blocks are deleted, leaving three disconnected vertices and three pairs of disconnected edges which must end up at the same location on the sphere, respectively, after projections, in order to fill the gap left by the deleted blocks. Since these three vertices and three pairs of edges start out in the same location, respectively, they will end up projected to the same position on the sphere, as required. This technique can be confusing to the beginner. It is important to understand this for it will lead to a greater understand in the use of **True*Grid***®.

This part starts out with 8 blocks. Only four of these blocks are needed. Three deletion commands are used to remove the four blocks that are not needed. In the following pictures, both the physical and computational window are shown before and after each of the deletions.



Figure 202 Initial Structured Block Part



Figure 203 Computational Window



Figure 204 Physical region to be delete



Figure 205 Comp. mesh before deletion



Figure 206 Physical mesh after deletion



Figure 207 Comp. mesh after deletion



Figure 208 Physical region to be deleted



Figure 209 Comp. mesh before deletion



Figure 210 Physical mesh after deletion



Figure 211 Comp. mesh after deletion



Figure 212 Physical region to be deleted



Figure 213 Comp. mesh before deletion



Figure 214 Physical mesh after Deletion



Figure 215 Comp. mesh after deletion



Figure 216 Faces to be Projected



Figure 217 Computational Window

Three faces of the mesh are highlighted for projection. The mesh is projected onto the cylinder using the **sfi** command. Three pairs of coincident edges in the physical mesh are projected to the same position on the sphere. The three regions of the physical mesh with zero dimensions before projection were expanded to the three solid regions.



Figure 218 Faces were Projected

The octant of the sphere is replicated three times using the **gct** and **grep** commands to form the half solid sphere.



Figure 219 Mesh after Replication

The command file follows:

```
c definition of the sphere surface.
sd 1 sp 0 0 0 2
  c The center of the sphere has coordinates (0,0,0) and
  c  the radius of the sphere is 2.

c definition of the structured block part
block 1 5 9; 1 5 9; 1 5 9; 0 1 1;0 1 1;0 1 1;
  c The indices are :
  c the i-direction:1,5,9;
  c the j-direction:1,5,9 and the k-direction:1,5,9.
  c The coordinates for the second and third reduced indices are
  c identical in the x, y, and z directions (0 1 1;0 1 1;0 1 1;).

c deletions for the butterfly mesh technique.
dei 2 3;; 2 3;
dei 2 3; 2 3;;
dei ; 2 3; 2 3;

c Projection of the 3 faces of the butterfly mesh
c  onto the sphere surface.
sfi 1 -3; 1 -3; 1 -3;sd 1

c Definition of global transformations
gct 3 ry 90; ry 180; ry 270;
     c rotations around y axis for 90,180 and 270 degrees

c Global replications of the part
grep 0 1 2 3;
 c 0 means no transformation. 1, 2, and 3 means transformations
 c previously defined using the gct command.

merge
```

Solid Pipe

In this example, the techniques of building a butterfly mesh for a cylinder are extended to a curved pipe. In a previous example, the Full Solid Cylinder (page 189), the block part was initialized so that the outer blocks were collapsed onto the inner block. This made it trivial to project the outer faces to the cylinder. This works well because of the symmetry of the cylinder. For a more general problem where the center line of the butterfly is a curve instead of a line, this method does not work.

Projection is still the best method for building the butterfly structure in this more general case. The edges, and consequently the faces, of the block mesh will be initialized using 3D curves. If the edges are properly shaped with these 3D curves, then the faces of the mesh will, initially, be close enough to the surface of projection so that the projection produces a good mesh.

In other discussions, it is stated that the projection method can only deal with a modest amount of curvature between block partitions. When 3D curves, which approximate the surface of projection, are used to initialize the mesh before projections, then this limitation on curvature no longer applies. This will be evident in this example.

The first step in the construction is to create the geometry. A 3D spline curve forms the center line of the pipe. The pipe surface, with varying radius, is formed along this curve. Two additional surfaces are used to form the ends of the pipe.

Four curves along the pipe surface will be used to initialize the outer edges of the mesh. In this example, they are extracted from the pipe surface. These are know as contour curves. If more than one surface is needed to form the outer surfaces, this method of extracting 4 contour curves may not work, since a simple contour like these can only be extracted from one surface. Spline curves can be used, but they may only have the control points in common with the surfaces. This is acceptable, since it is not essential that the 4 curves be precisely on the outer surfaces. The projection will pull any stray edge nodes, which were initialized to a curve, onto the surface of projection. There are two critical features needed in the construction of these 4 curves. First, they should be smooth. This is why the spline curves work so well. Second, the 4 curves need to be spaced about 90 degrees apart from each other along the circumference of the cross section of the pipe. It is also important that they do not twist around the pipe as they traverse the length of the pipe.

This produces a good exterior mesh, but the interior block going down the middle will be formed from straight line edges. Additional curves are needed for the interior edges to form a good interior mesh. The 4 interior curves are formed by interpolating the 4 outer curves. Then the interior edges are attached to these interior curves.



Figure 220 Pipe Surface and Spline



Figure 221 4 Contour Curves

A portion of the spline curve is visible inside the pipe surface built around that curve. Four exterior curves are extracted. Label the surface points. Select a start and end point with the same first or second index in the point's identifier. By locating the maximum number of contours, one can divide by four to get the proper interval for 90 degree distribution of the curves. Interpolate curves that are 180 degrees apart to form the interior curves. Usually an interpolation parameter of .3 produces good results.



Figure 222 All Curves



Figure 223 All Surfaces



Figure 224 Initial physical mesh



Figure 225 Initial computational mesh

The initial part consists of 9 blocks. The x-direction is associated with the axial curve of the pipe surface. There are two reasons to get the block part close to the geometry in this case. First, if it is not close, you can get confused as you are attaching and projecting portions of the mesh. If you are not concerned about this, then you should attach edges first, and then project the faces.

The second reason for initializing in a reasonable way is that the **cure** command determines which ends of the mesh should be attached to which ends of the curves. If the initial position is bad, it will make the wrong choice.

One technique of initializing the part is to use the computational window to select an edge end node. The appropriate end point of a curve can be easily selected with the mouse by first selecting the curve. In the *pick* panel, select *projection*. Then click the mouse near the end point of the appropriate curve. Then *attach* the end node to the curve end point



Figure 227 Selected regions for deletion



Figure 226 Selected regions

The next step is to remove the four corner blocks that are not needed. Use the computational window to select these blocks. Click-and-drag for each of the four intervals in the j- and k-index bars. Then click on the *delete* button. Alternatively, use the **de** or **dei** command in the mesh menu.



Figure 229 Physical mesh after deletion



Figure 228 Computational mesh



Figure 230 Face selection for projection



Figure 231 Computational face selection

The remaining four outer faces can be projected to the pipe surface. This can be done in one step by selecting the faces as should in the computational window. Select the surface and click on the *project* button. Alternatively, use the **sf** or **sfi** command in the mesh menu. The corners that were deleted are still evident in the pictures. The next step is to use the 3D curves to close the gaps.



Figure 232 Outer faces projected to pipe



Figure 233 Edge to be attached to a curve



Figure 234 Computational selection of an edge

One edge is selected in the computational window. Then select the appropriate 3D curve. If you click on the *attach* button, then the edge will be attached to the curve, but the end nodes will not necessarily be placed at the end of the 3D curve. This is not critical, since the end faces will also be projected to the end planes. This can also be done using the **curs** command in the mesh menu. Alternatively, you can use the **cure** command in the mesh menu to guarantee that the end nodes are attached to the ends of the curve.

If the part has been poorly initialized and if the **curs** command or *attach* button was used instead of the **cure** command, then it is possible that the nodes of the two edges that form a seam will not be aligned. Simply move the end node of one edge to the same position as the end node of the other edge to fix the poor initialization.



Figure 235 Result of edge attachment



Figure 236 Edge to be attached to curve



Figure 237 Computational selection of an edge

Similar steps are taken to attach the opposite edge to the same 3D curve. Select the edge in the computational window, select the 3D curve, and *attach*. Each pair of edges must be attached to the appropriate exterior 3D curve for each corner block that was deleted.



Figure 238 Attach second edge to curve



Figure 239 After all exterior edges are attached



Figure 240 Inner block is poorly shaped



Figure 241 Attach inner edge to curve

As noted above, the interior block is still linear. This is fixed by attaching the four interior edges to the appropriate 3D curves. One edge is selected, the appropriate curve is selected, and the edge is attached. Again, the easiest method of attachment in this case is the **cure** command. All four edges are attached in this manner. The result of all four attachments is shown below.



Figure 242 After inner edge is attached



Figure 243 After all inner edges are attached

```
c Center spline curve for the pipe surface
curd 1 csp3 00 0 0 0 .5 .2 0 1 0 0;;
sd 1 pipe 1 .3 0 .4 1;;

c Bounding planes
sd 2 plan 0 0 0 .028175 .01688 0;;
sd 3 plan 1 0 0 .028175 -.01688 0;;

c Contour curves extracted from the pipe surface
curd 2 contour 1.1.13 1.0.13;
curd 3 contour 1.1.94 1.0.94;
curd 4 contour 1.1.40 1.0.40;
curd 5 contour 1.1.67 1.0.67;

c Curves of interpolation from contour curves
curd 6 intcur 3 4 .3;
curd 7 intcur 3 4 .7;
curd 8 intcur 2 5 .3;
curd 9 intcur 2 5 .7;

c Initial block part
block 1 21;1 5 9 13;1 5 9 13;0 1;-.15 -.05 .1 .25;-.2 -.1 .1 .2;

c Delete four solid regions forming butterfly mesh
dei ;1 2 0 3 4;1 2 0 3 4;

c Project exterior faces of the mesh to pipe surface
sfi ;-1 -4;-1 -4;sd 1
c Project ends to bounding planes
sfi -1;;;sd 2
sfi -2;;;sd 3

c Attach exterior edges of the mesh to contour curves
cure 1 3 4 2 3 4 2
cure 1 4 3 2 4 3 2
cure 1 1 3 2 1 3 3
cure 1 2 4 2 2 4 3
cure 1 3 1 2 3 1 4
cure 1 4 2 2 4 2 4
cure 1 1 2 2 1 2 5
cure 1 2 1 2 2 1 5
c Attach interior edges of mesh to interior curves
cure 1 2 3 2 2 3 6
cure 1 3 2 2 3 2 7
cure 1 3 3 2 3 3 8
cure 1 2 2 2 2 2 9
```

Solid Wedge with Transition Elements

Four parts are created, each on top of the other, to form the shape of a wedge. There is a transition interface between each pair of parts so that the number of elements is reduced at each layer. Three-to-one transitions are used at all three interfaces, reducing 27 elements in the first part to 1 element in the last part with the this last part being a wedge element.



Figure 244 Surfaces



Figure 245 Part 1 - Physical Window



Figure 246 Computational Window

This is the first part with 27 elements in the i-direction. The maximum j-face is used as the master side of Block Boundary interface (**bb** command).



Figure 248 Master Side of Block Boundary 1



Figure 247 Computational Window



Figure 249 Before Projection



Figure 251 Computational Window

Then 4 faces of the block are projected onto surfaces 1, 2, 3, and 4 using the **sfi** command. These pictures show the before and after views for the projection to surface 1. Similar steps are take for the other three surfaces.



Figure 250 After Projection



Figure 254 Part 2 - Physical Window

Part 2 is created using the **block** command. The coordinates of the bottom face nodes are copied from Part 1 using the **trbb** command. The other 3 faces are projected onto surfaces using the **sfi** command.



Figure 252 Slave Side of Block Boundary 1



Figure 253 Computational Window

The transition elements are created between Part 1 and Part 2, after the **merge** command is issued.



Figure 255 Parts 1 and 2

Another 2 parts are defined, **bb**, **trbb** and **sfi** commands are issued .



Figure 256 Parts 1, 2, 3, and 4

Command file follows:

```
c Definitions of Surfaces 1 to 7;
c Surfaces are defined as planes (plan) determined by a point
c and a normal:
sd 1 plan 5 8 0 1 -.5 0    c point (5,8,0) and normal (1,-.5,0)
sd 2 plan 5 8 0 1 .5 0     c point (5,8,0) and normal (1,.5,0)
sd 3 plan 5 0 0 0 1 0      c point (5,0,0) and normal (0,1,0)
sd 4 plan 5 2 0 0 1 0      c point (5,2,0) and normal (0,1,0)
sd 5 plan 5 4 0 0 1 0      c point (5,4,0) and normal (0,1,0)
sd 6 plan 5 6 0 0 1 0      c point (5,6,0) and normal (0,1,0)
sd 7 plan 5 8 0 0 1 0      c point (5,8,0) and normal (0,1,0).

c Definition of Part 1. It has 27 elements in the i-direction
c (divisible by 3).
block 1 28;1 4;1 5;0 10;0 2;0 5;

c Definition of the master side of the block boundary 1.
bb 1 2 1 2 2 2 1;

sfi -1;;;sd 1    c Projection of face onto surface 1.
sfi -2;;;sd 2    c Projection of face onto surface 2.
sfi ;-1;;sd 3    c Projection of face onto surface 3.
sfi ;-2;;sd 4    c Projection of face onto surface 4.

c Definition of Part 2. It has 9 elements in the i-direction
c (divisible by 3).
block 1 10;1 4;1 5;0 10;2 4;0 5;

c Definition of the slave side of transition block boundary 1.
trbb 1 1 1 2 1 2 1;

c Definition of the master side of the block boundary 2.
bb 1 2 1 2 2 2 2;

sfi -1;;;sd 1    c Projection of face onto surface 1.
sfi -2;;;sd 2    c Projection of face onto surface 2.
sfi ;-2;;sd 5    c Projection of face onto surface 5.

c Definition of Part 3. It has 3 elements in the i-direction
block 1  4;1 4;1 5;0 10;4 6;0 5;

c Definition of the slave side of the transition
trbb 1 1 1 2 1 2 2;      c block boundary 2.

c Definition of the master side of the block boundary 3.
bb 1 2 1 2 2 2 3;
```

```
sfi -1;;;sd 1    c Projection of face onto surface 1.
sfi -2;;;sd 2    c Projection of face onto surface 2.
sfi ;-2;;sd 6    c Projection of face onto surface 6.

c Definition of Part 3. It has 3 elements in the i-direction.
block 1  2;1 3;1 5;0 10;6 8;0 5;

c Definition of the slave side of the transition
trbb 1 1 1 2 1 2 3;      c block boundary 3.

sfi -1;;;sd 1    c Projection of face onto surface 1.
sfi -2;;;sd 2    c Projection of face onto surface 2.
sfi ;-2;;sd 7    c Projection of face onto surface 7.

merge

tvv       c Display all parts in the Fill Mode.
```

Cylinder and Block

The mesh of a cylinder drop test is created from one block and two cylinder parts.

Duplicate nodes are merged between the parts using the **stp** command and the parts are labeled using the **labels parts** command.

Parts are displayed in the exploded view using the **pexp** command.



Figure 257 Structured Block Mesh



Figure 258 Computational Window



Figure 259 Structured Cylinder Part



Figure 260 Computational Window



Figure 261 Structured Cylinder Mesh



Figure 262 Computational Window



Figure 263 Parts with Merged Nodes



Figure 264 Parts in Exploded View

The command file follows:

```
c Definition of the structured block part -> Part 1
block 1 2;1 2;1 2;0 5 -7.5 -4.5 0 -7.5
  c Indices in the i-direction are: 1 and 2.
  c Indices in the j-direction are: 1 and 2.
  c Indices in the k-direction are: 1 and 2.
  c Coordinates of the indices in the x-direction are: 0 and 5.
  c Coordinates of the indices in the y-direction are:
  c -7.5 and -4.5.
  c Coordinates of the indices in the z-direction are: 0 and -7.5

c Definition of the structured cylinder part -> Part 2
cylinder 1 4;1 13 40;
     1 5 9 27; 4.25 4.5 -90 -34.61 90 0 -.75 -1.5 -6
  c Indices in the i-direction are: 1 and 4.
  c Indices in the j-direction are: 1, 13, and 40.
  c Indices in the k-direction are: 1, 5 , 9, and 27.
  c Coordinates of the indices in the radial-direction are:
  c   4.25 and 4.5.
  c Coordinates of the indices in the angular-direction are:
  c  -90,-34.61, 90
  c Coordinates of the indices in the z-direction are:
  c  0, -.75, -1.5, and -6

c Definition of the structured cylinder part -> Part 3
```

```
cylinder 1 5; 1 13 40;1 5; 4.25 3.25 -90 -34.61 90 -5  -6
 c Indices in the i-direction are: 1 and 5.
 c Indices in the j-direction are: 1, 13, and 40.
 c Indices in the k-direction are: 1 and 5.
 c Coordinates of the indices in the radial-direction are:
 c   4.25 and 3.25.
 c Coordinates of the indices in the angular-direction are:
 c  -90,-34.61, 90
 c Coordinates of the indices in the z-direction are:
 c   -5 and -6.

merge

c Merging nodes between the parts with tolerance .001 units.
stp .001;

labels parts      c Parts are labeled.

iniexp ;          c All explode offsets are initialized to zero.

c Parts 2 and 3 are moved by a vector (0,1,0)
c from the initial position.
pexp 0 1 0 2 3; ; ;

c Part 3 is moved by a vector (-3,0,1)
c from the previous position.
pexp -3 0 1 3; ; ;
```

Fork

This example is of a fork connector. It is formed by two parts. The first part is a pair of prongs extending from an arm. The second part is a pin that passes through two holes in the pongs. A sliding interface is placed between these two parts. Some nodes are constrained and others have loads applied to them. All of these conditions can be viewed in the merge phase using the **co** command. Nodes must also be merged because of the butterfly mesh for the pin. None of the nodes on opposite sides of the sliding interface will be merged.



Figure 265 Surfaces

This picture shows the six surfaces defined using the **sd** command with options **cy** (cylinder) and **plane** (plane). Note that there are two cylinders for the pin. One is for the larger radius master side of the sliding interface. The other is for the smaller slave side of the sliding interface. The bottom two pictures show the physical and computation views of the first part generated with the **block** command.



Figure 266 Part 1 - Structured Block Part



Figure 267 Computational Window



Figure 268 Region to Delete



Figure 269 Computational Window

A volumetric region is deleted using the **dei** command. This creates the gap between the prongs.



Figure 270 Mesh after Deletion



Figure 271 Regions to Delete



Figure 272 Computational Window

More regions are deleted using the **dei** command. This creates the holes which the pin will pass through. At this stage, they are square.



Figure 273 Mesh after Deletion



Figure 274 Mesh before Projection



Figure 275 Computational Window

The interior facial regions are projected onto surface 1 (the cylinder) using the **sfi** command.



Figure 276 Mesh after Projection



Figure 277 Mesh before Projection



Figure 278 Computational Window

Next, the interior corners are rounded by projecting two rows of faces onto surface 2 (a cylinder) using the **sfi** command.



Figure 279 Mesh after Projection



Figure 280 Mesh before Projection



Figure 281 Computational Window

The central interior faces are projected onto surface 4 (plane) using the **sfi** command.



Figure 282 Mesh after Projection



Figure 283 Edge to be Moved



Figure 284 Computational Window

The inside edge of the fork is pulled back in the x-direction using the **pb** command.



Figure 285 Edge was Moved



Figure 286 Nodes of Region to be Spaced



Figure 287 Computational Window

The nodes in the main body of the fork are re-distributed in the z-direction (k-direction in the Computational Window) using the **res** command.



Figure 288 Nodes of Region were Spaced



Figure 290 Face to be Relaxed



Figure 291 Computational Window

Now, a slight equipotential relaxation is used to re-distribute the nodes on the outer face. This is an elliptic smoothing method The relaxation is slight so as to not disturb the previous nodal redistribution by too much.



Figure 289 Face after it is relaxed



Figure 292 Face to be smoothed



Figure 293 Computational Window

Lastly, the remaining long exterior faces are slightly relaxed in the same fashion. The face is smoothed using the equipotential relaxation method invoked with the **relaxi** command.



Figure 294 Equipotential Relaxation

Part 2 is defined using the **block** command. The butterfly technique is used to create a mesh of the cylinder from the block structured mesh using the **dei** and **sfi** commands. See also 189.



Figure 295 Part 2 - Structured Block Mesh



Figure 296 Parts 1 and 2 with small gap.



Figure 297 Concentrated Load 1



Figure 298 Concentrated Load 2

The **merge** phase is used to assemble the entire model. It is then merged with the **stp** command. Loads and sliding interfaces should be checked at this stage using the **co** command. After everything has been checked, then **write** the output with the write command.



Figure 299 Master Side of Sliding Interface 1



Figure 300 Slave Side of Sliding Interface 1

The command file follows:

```
c Load curves for conditions dependent on time
lcd 1 0 0 1 [4000./49] 10 1;
lcd 2 0 0 1 [220./96] 10 1;

nike3d       c Output option for NIKE3D is selected.

c Global properties of sliding interface number 1.
sid 1 sl fric .3;   c with friction .3.

c Definition of cylinders
sd 1 cy 0 .875 0 1 0 0 .375
sd 2 cy .347 0 -.875 0 1 0 .125.
sd 3 cy 1.153 0 -.875 0 1 0 .125
sd 6 cy 0 .875 0 1 0 0 .37

c Planes defined by a point and a normal
sd 4 plane .311 0 0 1 0 0
sd 5 plane 1.189 0 0 1 0 0

c Definition of the structured block part (Part 1).
block
  1 5 7 13 15 19;
  1 5 7 10 13 15 19;
  1 7 9 15 17 20 23 25 29 31;
  0. .222 .347 1.153 1.278 1.5
  0 .4 .575 .875 1.175 1.35 1.75
  -3 -1 -.875 -.475 -.3 0 .3 .475 .875 1.

dei 2 5;;2 10;      c Deletion of the region.
dei ;3 5;5 7;       c Deletion of the region.

c Projections.
sfi ; -3 -5; -5 -7;sd 1     c Projection onto cylinder.
sfi -2 3; ; -2 3;sd 2       c Projection onto cylinder.
sfi 4 -5; ; -2 3;sd 3       c Projection onto cylinder.
sfi -2; 2 6; 4 8;sd 4       c Projection onto plane.
sfi -5; 2 6; 4 8;sd 5       c Projection onto plane.

c Placement of the edges in the x-direction.
pb 2 0 10 2 7 10 x .188
pb 5 0 10 5 7 10 x 1.312

c Relative spacing of the nodes in the k-direction.
res 1 1 1 6 7 2 k .7

c Relaxations of the mesh.
```

```
relaxi -1; ; ; 20 .001 1
relaxi -6; ; ; 20 .001 1
relaxi ; -1; 1 3; 20 .001 1
relaxi ; -7; 1 3; 20 .001 1

c nodal forces to the face
fci -5; 2 6; 4 8; 2 1. 1 0 0

c displacement constraint
bi ; ; -1; dx 1 dy 1 dz 1;

c Assignment of the master side (1 m) of the sliding interface
sii 1 2 0 5 6; -3 -5; -5 -7; 1 m

mate 1       c Assignment of material 1 to the part.

c Definition of the structured block part (Part 2).
block
  1 5 15 19;
  1 7 13 19;
  1 7 13 19;
  0. .222  1.278 1.5;
  .775 .775 .975 .975;
  -.1 -.1 .1 .1;

dei ;1 2 0 3 4;1 2 0 3 4;    c Deletion for the butterfly mesh.

c Projection of the butterfly mesh onto cylinder
sfi ; -1 -4; -1 -4; sd 6

c Assignment of the slave side (1 s) of the sliding interface
sii 1 2 0 3 4; -1 -4; -1 -4; 1 s

c Assignment of the concentrated nodal loads
fci 2 3;; -1; 1 1. 0 0 1

mate 2       c Assignment of material 2 to the part.

merge

co size 3    c Size of the arrows is set.

co fc 1;  c Concentrated load 1 is displayed.
co fc 2;  c Concentrated load 2 is displayed.
co si 1 m;     c Master side of sliding interface 1 is displayed.
co si 1 s;     c Slave  side of sliding interface 1 is displayed.

stp .001
```

5. Shells and Solids.

Shell and Solid for Contact

The mesh for the simulation of contact between two bodies is created from one block (shell) and one cylinder. The model is that of a thick-walled cylinder, consisting of brick elements, contacting a torus which is modeled with shell elements.



Figure 301 2D curve 1



Figure 302 2D curve 2

Two 2D curves are defined at first using the **ld** command with option **lep**. Then the surface is created using the **sd** command with the **swept** option.



Figure 303 Surface by Swept option





Figure 304 Initial Position of the Mesh

Figure 305 Computational Window

The shell mesh is created using the **block** command. The mesh is then positioned closer to the surface using the **tri** command.





Figure 306 Mesh after Repositioning

Figure 307 Computational Window





Figure 309 Mesh after Patch **Figure 308** Computational Window

Edges of the mesh are matched with the edges of the surface using the **patch** command. This command also projects the face to the surface of the patch command. The patch command is best used for simple meshes like this one. The face mesh is interpolated using the **tfi** command for a better effect.



Figure 310 Mesh after Transfinite Interpolation





Figure 311 Structured Cylinder Mesh **Figure 312** Computational Window

Finally, the mesh is replicated three times by rotation through 90 degrees using the **lct** and **lrep** commands. The cylinder is created using the **cylinder** command.





Figure 313 Mesh after Replication by Rotation **Figure 314** Complete Mesh

```
c 2D curve number 1, which will be swept as a cross section.
ld 1 lep 6 7 5 5 30 140 0 ;

c 2D curve number 2, which will be the sweeping trajectory.
ld 2 lep 5 5 0 0 0 90 0 ;

c surface number 1 by sweeping of the 2D curve 1
sd 1 swept 2 r 1 0 1 1;;      c on 2D curve 2.

c Shell part
block 1 3 5 7 9;-1;1 3 5 7 9; 0 2 4 6 8; 9 ; 0 2 4 6 8;

c Mesh is positioned by the set of transformations
tri 1 5; ;1 5;mx -4 mz -4 ry 45 mx 7 mz 7  ;

c Attachment of the mesh onto the surface 1.
patch 1 1 1 5 1 5 1

c Transfinite interpolation of the mesh.
tfi 1 5; ; ;

c 3 global transformations around y-axis
gct 3 ry 90; ry 180; ry 270 ;

c Global replication 3 times
grep 0 1 2 3;

c Cylinder Part -> tube made from hexahedral elements.
cylinder 1 6; 1 3 5 7 9 11 13 15 17 19 21 23 25; 1 10;
         2 4;0 30 60 90 120 150 180 210 240 270 300 330 360;0 20;

c local translation (my) of 20 units.
lct 1 my 20 ;

c Translation of the part by local transformation number 1.
lrep 1;

merge
```

XIII. Pre-Processing Example

This chapter contains complete examples, including material specifications and boundary conditions. We are starting with just one example, which can specify either DYNA or LS-DYNA as the output formats. This particular example was contributed by Mr. Len Schwer and he gets all the credit as well as XYZ Scientific Applications, Inc.'s gratitude. You can also find all of the listed files in the distributed Preprocessing sub-directory of the Examples directory.

1. Dynamic Brazilian Tests

The problem to be solved is the dynamic loading of a cylindrical concrete specimen mounted between two long steel rods, a so called Brazilian test (cylindrical specimen) in a split Hopkins bar (two long rods). The main reference for this work is:

"Three-dimensional finite-element simulation of the dynamic Brazilian tests on concrete cylinders", Gonzalo Ruiz, Michael Ortiz, Anna Pandolfi, International Journal for Numerical Methods in Engineering, Volume: 48, Issue: 7, Date: 10 July 2000, Pages: 963-994



Figure 315 Concrete model

http://www3.interscience.wiley.com/cgi-bin/abstract/72503043/START

L. Schwer's modeling consists of these TG files:

Brazilian.run.tg main TG file with analysis options
material.defs common material parameter definitions for use by either LS-DYNA or DYNA3D
materials.dyna DYNA3D material model definitions
materials.lsdyna LS-DYNA material model definitions; includes an example of inputs for a User Supplied Material Model (mat 48)
interface.defs DYNA3D or LS-DYNA contact interface definitions
brazilian.geom TG commands need to construct the mesh and associated boundary conditions

The complete content of each of these files now follows:

Brazilian.Run.tg

```
c ----------------------------------------------------------------
c
c           Model of Dynamic Brazilian Tests on Concrete
c
c                     L Schwer  September 2000
c
c ----------------------------------------------------------------
c
title Model of Dynamic Brazilian Tests on Concrete
c
c ----------------------------------------------------------------
c
c Change Notes:
c
c 19 Sep - revise system of units
c
c ----------------------------------------------------------------
c
c  units are
c                    length -> millimeters  = 10^-3 m
c                    time   -> milliseconds = 10^-3 s
c                    mass   -> gram         = 10^-3 kg
c                    Force  -> m*a = g*(mm/ms^2)
c                                = kg-m/s^2 = N
c                  pressure -> F/A = N/mm^2 = 10^6 N/m^2 = MPa
c
c
c ----------------------------------------------------------------
c
c  Main Reference:
c       "Three-dimensional finite-element simulation of the
c          dynamic Brazilian tests on concrete cylinders"
c        Gonzalo Ruiz, Michael Ortiz, Anna Pandolfi
c      International Journal for Numerical Methods in Engineering
c      Volume: 48, Issue: 7, Date: 10 July 2000, Pages: 963-994
c http://www3.interscience.wiley.com/cgi-bin/abstract/72503043/START
c
c ----------------------------------------------------------------
c
c   * * *  model analysis & configuration options/parameters  * * *
c
  parameter
c    lsdyna =1 generate ls-dyna output or =0 generate dyna3d
         lsdyna 0
c
c    hqsymm =0 full model  =2 half z-symmetry
c                   =4 quarter yz-symmetry
         hqsymm 2
      ;
c                   half or quarter symmetry model
         if (%hqsymm .ne. 0) then
           plane 1 0.0 0.0 0.0  0.0 0.0 1.0  0.001 symm ;  c half z-symmetry
          if (%hqsymm .eq. 4) then
           plane 2 0.0 0.0 0.0  0.0 1.0 0.0  0.001 symm ;  c quarter
yz-symmetry
          endif
         endif                    c symmetry model
```

```
c
  parameter
      endtime  90.00e-2    c simulation duration
      plotime  9.00e-2     c plot interval
      printime 4.00e-3     c print interval
      dtmass   0.0e-4      c mass scaled time step
      endmasp  25.0        c terminate when added mass % exceeds value
c     drtime =0   no dynamic relaxation
      drtime  0.00e-2      c dynamic relaxation loading time
      perdrt  2.50e-2   c percent of drtime for start/stop load curve ramps
      drendt [1.125*%drtime] c total dynamic relaxation time
      ;
c
c  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -
c    concrete material model options
c  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -
c
  parameter
c    concrete =1  elastic material
c             =37 smooth cap model (dyna3d only)
c             =48 smooth cap model (ls-dyna user supplied)
         concrete  1
c
c                  for elastic model select concrete flavor
c    GranRuiz = 0   elastic concrete properties reported by Gran
c             = 1   elastic concrete properties reported by Ruiz
         GranRuiz 1
c
c                  for Smooth Cap Model select concrete flavor
c    wsmr5    = 0   APTEK Round B WSMR-5 concrete model
c             = 1   APTEK Revised WSMR-5 concrete model
         wsmr5  0
c
                    ;
c
c  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -
c -----------------------------------------------------------------
c
c      * * *  load curves  * * *
c
  parameter
             lccase  2    c cases 1 - 6 from Ortiz paper
             duration 100.0e-3
      if (%lccase .eq. 1) then
             rise   66.0e-3
             stress   60.2
             vstriker [3.04*1000/1000]   c 3.04 m/s -> 3.04 mm/msec
      elseif (%lccase .eq. 2) then
             rise   72.0e-3
             stress   72.8
             vstriker [3.67*1000/1000]   c 3.67 m/s -> 3.67 mm/msec
      elseif (%lccase .eq. 3) then
             rise   80.0e-3
             stress   79.4
             vstriker 0.0
      elseif (%lccase .eq. 4) then
             rise   85.0e-3
             stress   122.5
             vstriker 0.0
      elseif (%lccase .eq. 5) then
```

```
             rise   41.0e-3
             stress   184.5
             vstriker 0.0
      elseif (%lccase .eq. 6) then
             rise   48.0e-3
             stress   264.3
             vstriker [13.34*1000/1000]   c 13.34 m/s -> 13.34 mm/msec
      endif
             loadcurve 1
            ;
c
c load curve for axial stress loading
lcd %loadcurve
0.0                    0.0
%rise                %stress
[%rise + %duration]  %stress
[2*%rise + %duration]  0.0
1.0e-0                 0.0 ;
c
c -----------------------------------------------------------------
c -----------------------------------------------------------------
c
c    * * *  include files  * * *
c
  include material.defs     c material definitions
  include interface.defs    c interface definitions
c
  include brazilian.geom    c bar & specimen geometry
c
c
c -----------------------------------------------------------------
c  * * * * *  DYNA3D Options * * * * *
c -----------------------------------------------------------------
c
c %%%%                          %%%%
      if (%lsdyna .eq. 0.0) then
c %%%%                          %%%%
c
 dyna3d
 dynaopts
 term  %endtime
 plti %plotime
 prti %printime
c
 pnlt 0.05              c scale factor for sliding interface penalty
                        c   default is pnlt=0.10
 prtflg 1              c print initial time steps
 defpf 1                c print discrete element forces
 nrest 99990
 nrunr 20000
c rfpf 0                c reaction force print flag
c itrx 500 tolrx 1.0e-8  c dynamic relaxation
c drdb                  c dynamic relaxation plot files
             ;
c
c %%%%                   %%%%
       else
c %%%%                   %%%%
c
c -----------------------------------------------------------------
```

```
c  * * * * *  LS-DYNA Options * * * * *
c -----------------------------------------------------------------
c
lsdyopts
c LSDYNA3D FIXED
 LSDYNA3D KEYWORD
c
npopt 1                     c print suppression during input of nodes &
elements
ipnint 1                    c print initial time step size
ikedit 99000                c problem status report step interval
ianprc 0                    c analysis procedure 0=structural only
c struct 1                     c structured input file = 1 ; else = 0
c
c
endtim %endtime             c termination time
 dt2ms [-%dtmass]           c mass scaled time step
 ms1st 0                    c limit mass scaling to first step =1
 endmas [%endmasp]          c terminate when added mass % exceeds value
c
d3plot dtcycl %plotime ;    c complete plot states
d3thdt dtcycl %printime ;   c time history interval
  nodout      %printime     c ASCII database
  elout       %printime     c ASCII database
  GLSTAT      %printime     c global data
c MATSUM      %printime     c material energies
c RBOUT       %printime     c rigid body data
c SLEOUT      %printime     c interface energy
c NCFORC      %printime     c nodal interface forces
c RCFORC      %printime     c resultant interface forces
c DEFORC      %printime     c discrete elements
c JNTFORC     %printime     c joint forces
c
strflg 1                    c include shell strain in plot database
epsflg 1                    c include effective plastic strain in plot
database
c
slsfac 0.05                 c scale factor for sliding interface penalties
penopt 1                    c contact surface penalty stiff - use min of
master seg
                            c                   slave node
c
  if (%drendt .gt. 0.0) then   c dynamic relaxation ::::::
c          dynamic relaxation options
 drterm        %drendt      c term time for DR (optional)
c nrcyck        1000         c number of iterations between conv checks
 nrcyck        [%drendt/%dtmass/5]  c number of iterations between conv checks
 drtol         1.e-3        c convergence tolerance
 drfctr        0.995        c dynamic relaxation factor
 tssfdr        0.0          c scale for computed time step
 irelal        1            c Auto (Papadrakakis) control
 edttl         1.0e-3       c convergence tolerance with Papadrakakis
 idrflg        1            c DR stress initialization (=1 active)
c
c D3RDFL DTCYCL [%drendt/3] c DR plotfile time increment (non-functioning)
c D3RDFL NPLTC 4            c DR plotfile number of plot states
(non-functioning)
c
  endif                        c dynamic relaxation ::::::
;
```

```
c
c
c %%%%                  %%%%
       endif
c %%%%                  %%%%
c
c -----------------------------------------------------------------
c -----------------------------------------------------------------
c
c
stp .008
merge
```

material.defs

```
{
******************************************************************************
******************************************************************************
*****
*****
*****      This is the general material definitions file for the Dynamic
*****   Brazilian Test Example.  Notice that, at the end of this file, you
*****
*****   also need to uncomment either the DYNA or the LS-DYNA material
*****
*****   file.  Please see the ReadME.txt file for more details.
*****
*****      All the files in this sub-directory (folder) were provided by
*****
*****                        L. Schwer
*****
*****   and we gratefully acknowledge his efforts in preparing this
*****
*****   example.
*****
*****
*****
*****                               XYZ Scientific Applications, Inc.
*****
*****                               October 16, 2000
*****
*****
*****
******************************************************************************
******************************************************************************
}
c -----------------------------------------------------------------
c
c            materials for dyna3d/ls-dyna simulation of
c           Model of Dynamic Brazilian Tests on Concrete
c
c                     L Schwer  September 2000
c
c -----------------------------------------------------------------
c
c -----------------------------------------------------------------
c
c Change Notes:
c
```

```
c 11 Aug - add striker bar material
c
c 25 Aug - add Ruiz elastic concrete properties
c
c ------------------------------------------------------------------
c
c  units are
c                    length -> millimeters  = 10^-3 m
c                    time   -> milliseconds = 10^-3 s
c                    mass   -> gram         = 10^-3 kg
c                    Force  -> m*a = g*(mm/ms^2)
c                                  = kg-m/s^2 = N
c                  pressure -> F/A = N/mm^2 = 10^6 N/m^2 = MPa
c
c ------------------------------------------------------------------
c
c ----------------------------------------------------------------
c                 material parameter ID's
c ----------------------------------------------------------------
c
 parameters
             mate0        0            c null material
             sbarmat   [%mate0 +1]     c striker bar (S-ST)
             ibarmat   [%sbarmat +1]   c incident bar (S-ST)
             tbarmat   [%ibarmat +1]   c tranmission bar (S-ST)
             specmat   [%tbarmat +1]   c specimen (concrete)
             bearmat   [%specmat +1]   c bearing strip  (ST)
c
           ;
c
c conversion factors
   parameter
         MPa [1./145]           c 145 psi -> 1 MPa
         dencon [1.07e4]        c lbf-sec^2/in^4 -> grams/cm^3
        ;
c
c ----------------------------------------------------------------
c                material property definitions
c ----------------------------------------------------------------
c
 parameters
c                  Steel               PH 13-8 Mo Stainless
            stden [%dencon*7.34e-4]  s13den [%dencon*7.23e-4]
            stym  [3.00e7*%MPa]     s13ym  [2.83e7*%MPa]
            stpr   0.30             s13pr   0.278
            styld [30.0e3*%MPa]     s13yld [200.0e3*%MPa]
            sttan [30.0e3*%MPa]     s13tan [8.8e4*%MPa]
c
c      concrete properties (elastic) reported by Gran
c             ... units MPa and grams/cm^3 --> *10^-3 = grams/mm^3
            conden [2.30*1.e-3]
            conym [24.40e3]
            conpr [0.2]
c      concrete properties  (elastic) used by Ruiz
c             ... units MPa and grams/cm^3 --> *10^-3 = grams/mm^3
            conden [2.405*1.e-3]
            conym [37.9e3]
            conpr [0.2]
    ;
c
```

```
c check for  Round B or Revised APTEK WSMR-5 concrete model
c
   if (%wsmr5 .eq. 0) then                                      c -- %wsmr5
 parameter     c  Smooth Cap Model with Damage -
               c  APTEK Round B WSMR-5 concrete model (added 09/00)
      scrho [2.30*1.0e-3]
      shear 16e3    bulk 24.1e3 grun 0.0     shock 0.0    pore 1
      alpha 328     theta 0.03  gamma 317    beta 1.48e-3
      Efit  0.0     Ffit  0.0   kinN 0.0     kinC 0.0
      rzero 5       xzero 0     irock 1      secp 0
      w 0.05        d1 1.8e-4   d2 2.4e-7    nplot 16     maxeps 0.0
      Afit 0.7      Bfit 0.01   Rdam0 0.4
      Cfit 1.0      Dfit 0.0    Tenfail 3.9
      dbeta 0.0     ddelta 0.0  vplas 0.0    vpdam 0.0
      alpha1 0.705  theta1 0.0  gamma1 0.0   beta1 0.0
      alpha2 0.686  theta2 0.0  gamma2 0.0   beta2 0.0
      null  0  c space holder
    ;
c
c
   elseif (%wsmr5 .eq. 0) then                                  c -- %wsmr5
c
c
 parameter     c  Smooth Cap Model with Damage -
               c  APTEK Revised WSMR-5 concrete model (added 09/00)
      scrho [2.30*1.0e-3]
      shear [2.014e+06*%MPa]   bulk [2.250e+06*%MPa]  grun 0.0
                               shock 0.0              pore 1
c
      alpha [2.753E+04*%MPa]   theta 0.0
      gamma [2.673E+04*%MPa]   beta [1.724E-05/%MPa]
      Efit  1.000E+00          Ffit [8.30e-2/sqrt(%MPa)]  kinN 0.0 kinC 0.0
c
      rzero 5.000e+00     xzero [1.450E+04*%MPa] irock 1  secp 0
      w 0.05              d1 [1.724E-06/%MPa]    d2 [1.664E-11/(%MPa*%MPa)]
c
      nplot 16   maxeps 0.0  dbeta 0.0  ddelta 0.0 vplas 0.0    vpdam 0.0
c
      Afit 0.65    Bfit [8.300e-03/sqrt(%MPa)]  Rdam0 [9.633E-00*sqrt(%MPa)]
      Cfit 1.0     Dfit [2.491e+01/sqrt(%MPa)]  Tenfail [2000.0*%MPa]
  alpha1 0.74735 theta1 [2.275e-06/%MPa]  gamma1 0.170  beta1 [0.345e-03/%MPa]
  alpha2 0.660   theta2 [2.758e-06/%MPa]  gamma2 0.160  beta2 [0.345e-03/%MPa]
c
      null  0  c space holder
    ;
c
 endif                                                        c -- %wsmr5
c
c
c            Steel          AL2024-T351    AMS-4911-Ti
c           stden 7.34e-4  al2den 2.0e-4  tiden 2.0e-4
c           stym  3.00e7   al2ym  1.00e7  tiym  1.65e7
c           stpr  0.30     al2pr  0.30    tipr  0.30
c           styld 30.0e3   al2yld 50.0e3  tiyld 128.e3
c           sttan 30.0e3   al2tan 1.10e5  titan 2.20e5
c
c            AL6061-T6       4140 Steel      4130 Rc50 Stl    Ti-6Al-4V
c           al6den 2.0e-4   st4den 7.34e-4  st5den 7.34e-4  c tiden 2.0e-4
c           al6ym  1.00e7   st4ym  3.00e7   st5ym  3.00e7   c tiym  1.65e7
c           al6pr  0.30     st4pr  0.30     st5pr  0.30     c tipr  0.30
```

```
c          al6yld 28.0e3   st4yld 30.0e3   st5yld 220.0e3  c tiyld 128.e3
c          al6tan 30.0e3   st4tan 30.0e3   st5tan 1.50e5   c titan 2.20e5
c
c ---------------------------------------------------------------
c
c %%%%                          %%%%
  if (%lsdyna .eq. 1.0) then
c              LS-DYNA specific
   include materials.lsdyna     c LS-DYNA Material Definitions
c
c ................................................................
c
  else
c              DYNA3D specific
   include materials.dyna       c DYNA Material Definitions
c
c
  endif
c %%%%                           %%%%
c
c
c
```

materials.dyna

```
{
******************************************************************************
******************************************************************************
*****
*****
*****     This is the DYNA-specific material definitions file for the
*****
*****  Dynamic Brazilian Test Example.  This file follows the general
*****
*****  parameters, which are defined in materials.def
*****
*****  Please see the ReadME.txt file for more details.
*****
*****     All the files in this sub-directory (folder) were initially
*****
*****  provided by
*****
*****                     L. Schwer
*****
*****  and we gratefully acknowledge his efforts in preparing this
*****
*****  example.
*****
*****
*****
*****                            XYZ Scientific Applications, Inc.
*****
*****                            October 16, 2000
*****
*****
*****
******************************************************************************
******************************************************************************
}
c
c
```

```
c ------------------------------------------------------------------
c
c            materials for dyna3d simulation of
c           Model of Dynamic Brazilian Tests on Concrete
c
c                     L Schwer  September 2000
c
c ----------------------------------------------------------------
c
c ----------------------------------------------------------------
c
c  * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *
c          ----  solids and rigid solids  ----
c  * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *
c
c  ..................   solids  ..................
c
dynamats %sbarmat 3
head
         striker bar (PH 13-8 Mo Stainless solids)
e    %s13ym
pr   %s13pr
sigy %s13yld
etan %s13tan
rho  %s13den  ;
c
dynamats %ibarmat 3
head
         incident bar (PH 13-8 Mo Stainless solids)
e    %s13ym
pr   %s13pr
sigy %s13yld
etan %s13tan
rho  %s13den  ;
c
c
dynamats %tbarmat 3
head
         transmission bar (PH 13-8 Mo Stainless solids)
e    %s13ym
pr   %s13pr
sigy %s13yld
etan %s13tan
rho  %s13den  ;
c
c
dynamats %bearmat 3
head
         bearing strips (steel solids)
e    %stym
pr   %stpr
sigy %styld
etan %sttan
rho  %stden  ;
c
c  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -
c
 if (%concrete .eq. 1) then          c ... concrete material type
c
dynamats %specmat %concrete
```

```
head
         specimen (concrete elastic solids)
e    %conym
pr   %conpr
rho  %conden  ;
c
c
 elseif (%concrete .eq. 37) then     c ... concrete material type
c
c
DYNAMATS %specmat 0 %concrete
 if (%wsmr5 .eq. 0) then
    head
     specimen (APTEK Round B WSMR-5 concrete model with damage)
 elseif (%wsmr5 .eq. 1) then
    head
     specimen (APTEK Revised WSMR-5 concrete model with damage)
 endif
c
rho  %conden
c
c                                                Card 1
P31 %shear  P32 %bulk   P33 %grun   P34 %shock
P35 %pore   P36 %null   P37 %null   P38 %null
c                                                Card 2
P41 %alpha  P42 %theta  P43 %gamma  P44 %beta
P45 %Efit   P46 %Ffit   P47 %kinN   P48 %kinC
c                                                Card 3
P51 %rzero  P52 %xzero  P53 %irock  P54 %secp
P55 %Afit   P56 %Bfit   P57 %Rdam0  P58 %vpdam
c                                                Card 4
P61 %w      P62 %d1     P63 %d2     P64 %nplot
P65 %maxeps P66 %Cfit   P67 %Dfit   P68 %TenSig
c                                                Card 5
P71 %null   P72 %null   P73 %dbeta  P74 %ddelta
P75 %vplas  P76 %null   P77 %null   P78 %null
c                                                Card 6
P81 alpha1  P82 theta1  P83 gamma1  P84 beta1
P85 alpha1  P86 theta2  P87 gamma2  P88 beta2
               ;
c
c card images from an APTEK DYNA3D input file
c ---------------------------------------------------
c $   1   37 2.300E-01    0    4         0    0         0         0    0
c     1   37 2.240E-04    0    4         0    0         0         0    0
c  RDA Concrete WSMR with Willam Warnke Mpa centimeter msecs units
c   G         K
c 1.389e+04 1.552E+04 0.000E+00 0.000E+00 1.000E+00
c 2.014e+06 2.250e+06 0.000E+00 0.000E+00 1.000E+00
c  alpha      theta     lambda    beta     A dil    B dil
c 1.899E+02 0.055E+00 1.844E+02 2.500E-03 1.000E+00 1.000e+00 0.000e+00
0.000e+00
c  2.753E+04 0.055E+00 2.673E+04 1.724E-05 1.000E+00 8.300e-02 0.000e+00
0.000e+00
c $  R        X_0      roc  flag shencompc    A dam   B dam     r_0    etadamg
c $5.000e+00 1.000E+02 1.000E+00 0.000E+00 0.650e+00 0.100e+00 0.800E+00
0.000e+00
c  5.000e+00 1.450E+04 1.000E+00 0.000E+00 0.650e+00 8.300e-03 9.633E-00
0.000e+00
c $  W       D_1       D_2       plotvar               C(dam)    D(dam)
```

```
sigfailt
c $0.050E+00 2.500E-04 3.500E-07 2.400e+01 0.000e-00 1.000e+00 3.000e+02
3.200e+00
c  0.050E+00 1.724E-06 1.664E-11 2.400e+01 0.000e-00 1.000e+00 2.491e+01
4.640e+02
c        0.0       0.0       0.0       0.0 0.000e+00       0.0
c $Q1alpha1     theta1   lambda1    beta 1Q2 alpha2    theta2   lambda2  beta2
c   0.74735   0.00033   0.170e+00 0.05     0.660E+00 0.0004    0.160     0.05
c -----------------------------------------------------
c
c
  endif                              c ... concrete material type
c
c
```

materials.lsdyna

```
{
******************************************************************************
******************************************************************************
*****
*****
*****     This is the LS-DYNA specific material definitions file for the
*****
*****  the Dynamic Brazilian Test Example.  This file comes after the
*****
*****  general parameters, which are defined in materials.def
*****
*****  Please see the ReadME.txt file for more details.
*****
*****     All the files in this sub-directory (folder) were provided by
*****
*****                       L. Schwer
*****
*****  and we gratefully acknowledge his efforts in preparing this
*****
*****  example.
*****
*****
*****
*****                              XYZ Scientific Applications, Inc.
*****
*****                              October 16, 2000
*****
*****
*****
******************************************************************************
******************************************************************************
}
c -----------------------------------------------------------------
c
c            materials for ls-dyna simulation of
c           Model of Dynamic Brazilian Tests on Concrete
c
c                     L Schwer  September 2000
c
c ---------------------------------------------------------------
c
c ---------------------------------------------------------------
c
c                ls-dyna material definitions
```

```
c
c   dynamats -> LSDYMATS
c   elastic solids - STRUCT brick elfob csb
c   other   solids -        brick elfob csb
c   shells - Shell elfor blev tsti 3 quad 0 shth 1.00
c   head on one line
c   dynaeos -> LSDYEOS
c
c ---------------------------------------------------------------
c
c  * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *
c          ----  solids and rigid solids  ----
c  * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *
c
c  ..................   solids  ..................
c
LSDYMATS %sbarmat 3 brick elfob csb
head     striker bar (PH 13-8 Mo Stainless solids)
e    %s13ym
pr   %s13pr
sigy %s13yld
etan %s13tan
rho  %s13den  ;
c
c
LSDYMATS %ibarmat 3 brick elfob csb
head     incident bar (PH 13-8 Mo Stainless solids)
e    %s13ym
pr   %s13pr
sigy %s13yld
etan %s13tan
rho  %s13den  ;
c
c
LSDYMATS %tbarmat 3 brick elfob csb
head     transmission bar (PH 13-8 Mo Stainless solids)
e    %s13ym
pr   %s13pr
cubedel.pssigy %s13yld
etan %s13tan
rho  %s13den  ;
c
c
LSDYMATS %bearmat 3 brick elfob csb
head     bearing strips (steel solids)
e    %stym
pr   %stpr
sigy %styld
etan %sttan
rho  %stden  ;
c
c  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -
c
 if (%concrete .eq. 1) then        c ... concrete material type
c
LSDYMATS %specmat 1 STRUCT brick elfob csb
head     specimen (concrete elastic solids)
e    %conym
pr   %conpr
rho  %conden  ;
```

```
c
 elseif (%concrete .eq. 48) then   c ... smooth cap with damage
c
c
 parameter      c UMAT 48 Smooth Cap Model with Damage - APTEK WSMR-5 09/00
             mid  1  c material ID
              mt 48  c material model
             lmc 48  c length of material constant array (inputs)
             nhv 26  c number of history variables
          iortho  0  c orthotropic flag =0 no; =1 yes
           ibulk  3  c address of bulk modulus in material constant array
              ig  2  c address of shear modulus in material constant array
           ivect  0  c vectorization flag =0 no; =1 yes
           ifail  0  c failure flag =0 no; =1 yes
          ;
c
c parameter     c UMAT 48 Smooth Cap Model with Damage -
c               c  APTEK Round B WSMR-5 concrete model (added 09/00)
c      rho  %conden
c      shear 16e3    bulk 24.1e3 grun 0.0     shock 0.0    pore 1
c      alpha 328     theta 0.03  gamma 317    beta 1.48e-3
c      Efit  0.0     Ffit  0.0   kinN 0.0     kinC 0.0
c      rzero 5       xzero 0     irock 1      secp 0
c      w 0.05        d1 1.8e-4   d2 2.4e-7    nplot 16     maxeps 0.0
c      Afit 0.7      Bfit 0.01   Rdam0 0.4
c      Cfit 1.0      Dfit 0.0    Tenfail 3.9
c      dbeta 0.0     ddelta 0.0  vplas 0.0    vpdam 0.0
c      alpha1 0.705  theta1 0.0  gamma1 0.0   beta1 0.0
c      alpha2 0.686  theta2 0.0  gamma2 0.0   beta2 0.0
c      null  0  c space holder
c    ;
c
c
c =================================================================
c =================================================================
c
c
c  ------   model 48 is a user supplied material model ------
LSDYMATS %specmat %mt   brick elfob csb
c       i8b    c 8pt brick
        csb    c 1pt constant strain
 if (%wsmr5 .eq. 0) then
    head     specimen (APTEK Round B WSMR-5 concrete model with damage)
 elseif (%wsmr5 .eq. 1) then
    head     specimen (APTEK Revised WSMR-5 concrete model with damage)
 endif
c
rho %scrho
card1 %mid    %scrho  %mt     %lmc    %nhv   %iortho %ibulk  %ig ;
card2 %ivect  %ifail ;
card3 %shear  %bulk   %grun   %shock  %pore   %scrho ;
card4 %alpha  %theta  %gamma  %beta   %Efit   %Ffit  %kinN    %kinC ;
card5 %rzero  %xzero  %irock  %secp   %Afit   %Bfit  %Rdam0   %vpdam ;
card6 %w      %d1     %d2     %nplot  %maxeps %Cfit  %Dfit    %Tenfail;
card7 %null   %null   %dbeta  %ddelta %vplas  ;
card8 %alpha1 %theta1 %gamma1 %beta1  %alpha2 %theta2 %gamma2 %beta2 ;
;
c
c
 endif                              c ... concrete material type
```

interface.defs

```
{
******************************************************************************
******************************************************************************
*****
*****
*****     This is the general interface definitions file for the
*****
*****  Dynamic Brazilian Test Example.  This input file works for both
*****
*****  the DYNA and LS-DYNA versions of this example.
*****
*****  Please see the ReadME.txt file for more details.
*****
*****     All the files in this sub-directory (folder) were initially
*****
*****  provided by
*****
*****                      L. Schwer
*****
*****  and we gratefully acknowledge his efforts in preparing this
*****
*****  example.
*****
*****
*****
*****                             XYZ Scientific Applications, Inc.
*****
*****                             October 16, 2000
*****
*****
*****
******************************************************************************
******************************************************************************
}
c -----------------------------------------------------------------
c
c                    interface definitions
c               for ls-dyna/dyna3d simulation of
c           Model of Dynamic Brazilian Tests on Concrete
c
c                     L Schwer  September 2000
c
c ---------------------------------------------------------------
c
c ---------------------------------------------------------------
c
c
c -----------------------------------------------------------------
c  * * * * *  sliding interface definitions * * * * *
c -----------------------------------------------------------------
c
c Changes:
c
c
c 20 Aug - add sliding-with-void surface for striker-incident bars
c
c 25 Aug - option for striker-incident interface (omit)
c
c -----------------------------------------------------------------
```

```
c
 parameter barbear 1                  c 1 incident/transmission bar
                                      c    to bearing strip
           specbear [%barbear  + 1]   c 2 specimen to bearing strip
           strikeri [%specbear + 1]   c 3 striker to incident bar
           strikeri  0                c 0 omit striker to incident
           last     [%strikeri + 0] ;
c
c %%%%                          %%%%
       if (%lsdyna .ne. 1.0) then
c                                 DYNA3D definitions
c tied interfaces (3)
c
         sid %barbear  tied ;
         sid %specbear tied ;
c
c optional striker inface
  if (%strikeri .ne. 0.0) then
c  omit striker surface if pressure loading
      if (%vstriker .ne. 0.0) then
         sid %strikeri sv ;
      endif
  endif
c
       else
c                                 LS-DYNA definitions
         sid %barbear  LSDSI 2 ;
         sid %specbear LSDSI 2 ;
c
c optional striker inface
  if (%strikeri .ne. 0.0) then
c omit striker surface if pressure loading
      if (%vstriker .ne. 0.0) then
         sid %strikeri LSDSI 3 ;
      endif
   endif
c
       endif
```

brazilian.geom

```
{
*****************************************************************************
*****************************************************************************
*****
*****
*****     This is the geometry definitions file for the Dynamic Brazilian
*****
*****  Test Example.
*****
*****  Please see the ReadME.txt file for more details.
*****
*****     All the files in this sub-directory (folder) were initially
*****
*****  provided by
*****
*****                    L. Schwer
*****
*****  and we gratefully acknowledge his efforts in preparing this
*****
*****  example.
*****
*****
*****
*****                           XYZ Scientific Applications, Inc.
*****
*****                           October 16, 2000
*****
*****
*****
*****************************************************************************
*****************************************************************************
}
c -----------------------------------------------------------------
c
c           Model of Dynamic Brazilian Tests on Concrete
c
c                     L Schwer  September 2000
c
c -----------------------------------------------------------------
c
c -----------------------------------------------------------------
c
c Change Notes:
c
c 21 Jul - add non-reflecting boundary to transmitted bar (nri)
c        - additional length for incident bar
c        - element print block output
c
c 22 Jul - correct element print blocks (epb)
c
c 11 Aug - add striker bar
c
c 13 Aug - lengthen modeled portions of incident & transmission bars
c          to include distance from end to strain gages (40 inches)
c          and a 16 inch run-up section near strain gage locations.
c
c 20 Aug - add sliding-with-voids interface between striker and
c          incident bars (%strikeri)
c
```

```
c 25 Aug - add option for %strikeri interface
c
c ----------------------------------------------------------------
c
c  units are
c                    length -> millimeters  = 10^-3 m
c                    time   -> milliseconds = 10^-3 s
c                    mass   -> gram         = 10^-3 kg
c                    Force  -> m*a = g*(mm/ms^2)
c                                 = kg-m/s^2 = N
c                  pressure -> F/A = N/mm^2 = 10^6 N/m^2 = MPa
c
c ----------------------------------------------------------------
c
 parameter cyfaci [1/2.3] c factor for fitting grids in cylinders
          cyfaco [1.9] ; c factor for fitting grids on cylinders
c
c ----------------------------------------------------------------
c
c  geometry & symmetry
c           striker bar 25.7 inch x 2 inch dia
c           incident bar 12 feet x 2 inch dia
c           transmiter bar 11 feet x 2 inch dia
c           specimen 2 inch x 2 inch dia
c           bearing strip  0.25 inch x 0.25 inch x 2 inch
c           distance to strain gages 40 inch
  parameter
            coninmm   [25.4]        c convert inch to millimeter
            sbarlen   [25.7*%coninmm]
            ibarlen   [12*12*%coninmm]
            tbarlen   [11*12*%coninmm]
            speclen   [2*%coninmm]
            bearlen   [0.25*%coninmm]
            strainlen [40.0*%coninmm]    c distance to strain gages
            runuplen  [16.0*%coninmm]    c run-up to strain gages
            bardia    [2*%coninmm]
c  mesh spacing
         bmseqj     3                      c mseq j
         barney    [(5 - 1)*(%bmseqj + 1)] c number of elements (5 indices)
         bardy     [%bardia/%barney]       c y-distance
         bardx     [2*%bardy]              c x-distance 2:1 aspect
         smseqj     9
       ;
c
c ----------------------------------------------------------------
c
c   ...  incident bar  ...
c
 parameter barlen [%strainlen + %runuplen] ;
c
block 1 2 3 4  ; 1 2 3 4 5 ; 1 2 3 4 5 ;
c x-coors
[-%barlen]   [-%barlen + %runuplen]
[-%barlen + %runuplen + %bardx] 0.0
c y-coors
[-%cyfaci*%bardia/2] [-%cyfaci*%bardia/2] 0.0
[ %cyfaci*%bardia/2] [ %cyfaci*%bardia/2]
c z-coors
[-%cyfaci*%bardia/2] [-%cyfaci*%bardia/2] 0.0
[ %cyfaci*%bardia/2] [ %cyfaci*%bardia/2]
```

```
c
 mseq i [(%runuplen/%bardx) - 1]
        0 [((%barlen - %runuplen - %bardx)/%bardx) - 1]
 mseq j %bmseqj %bmseqj %bmseqj %bmseqj
 mseq k %bmseqj %bmseqj %bmseqj %bmseqj
c
dei ; 1 2 0 4 5 ; 1 2 0 4 5 ;  c four corners
c
c
c half or quarter symmetry model
 if (%hqsymm .ne. 0) then
  dei ; ; 3 5 ;   c half z-symmetry
   if (%hqsymm .eq. 4) then
    dei ; 1 3 ; ;   c quarter yz-symmetry
   endif
 endif                   c symmetry model
c
c
sfi ; -1 0 -5 ; 2 4 ; xcy [%bardia/2] ;
sfi ; 2 4 ; -1 0 -5 ; xcy [%bardia/2] ;
c
c
c apply pressure load to left end of incident bar
  if (%vstriker .eq. 0.0) then
   pri -1 ; ; ;  %loadcurve -1.0
  endif
c
c contact surfaces on right & left ends
   sii -4 ; ; ;  %barbear m
c optional striker inface
  if (%strikeri .ne. 0.0) then
c  omit striker surface if pressure loading
    if (%vstriker .ne. 0.0) then
     sii -1 ; ; ;  %strikeri m
    endif
  endif
c
c element print blocks
   epb 2 3 2  3 4 3        c strain gage location
c
c
 lct 1 mx [-%bardia/2 - %bearlen] ; lrep 1 ;
c
mate %ibarmat
c
endpart
c
c   ...  striker bar  ...
c
block 1 2 3 ; 1 2 3 4 5 ; 1 2 3 4 5 ;
c x-coors
[-%sbarlen] [-%bardx] 0.0
c y-coors
[-%cyfaci*%bardia/2] [-%cyfaci*%bardia/2] 0.0
[ %cyfaci*%bardia/2] [ %cyfaci*%bardia/2]
c z-coors
[-%cyfaci*%bardia/2] [-%cyfaci*%bardia/2] 0.0
[ %cyfaci*%bardia/2] [ %cyfaci*%bardia/2]
c
```

```
 mseq i  [(%sbarlen/%bardx) - 1] 0
 mseq j %bmseqj %bmseqj %bmseqj %bmseqj
 mseq k %bmseqj %bmseqj %bmseqj %bmseqj
c
dei ; 1 2 0 4 5 ; 1 2 0 4 5 ;  c four corners
c
c
c half or quarter symmetry model
 if (%hqsymm .ne. 0) then
  dei ; ; 3 5 ;   c half z-symmetry
   if (%hqsymm .eq. 4) then
    dei ; 1 3 ; ;   c quarter yz-symmetry
   endif
 endif                    c symmetry model
c
c
 sfi ; -1 0 -5 ; 2 4 ; xcy [%bardia/2] ;
 sfi ; 2 4 ; -1 0 -5 ; xcy [%bardia/2] ;
c
c
c apply inital velocity to striker bar
  if (%vstriker .ne. 0.0) then
   velocity %vstriker 0.0 0.0
  endif
c
c optional striker inface
  if (%strikeri .ne. 0.0) then
c  contact surface on right end
    sii -3 ; ; ;  %strikeri s
  endif
c
c element print blocks
   epb 2 3 2  3 4 3        c right end of striker
c
 lct 1 mx [-%barlen - %bardia/2 - %bearlen] ; lrep 1 ;
c
c omit striker if pressure loading is used
  if (%vstriker .ne. 0.0) then
    mate %sbarmat
  else
    mate %mate0
  endif
c
endpart
c
c
c   ...  tranmitted bar  ...
c
c parameter barlen [%tbarlen/4] ;
c
block 1 2 3 4  ; 1 2 3 4 5 ; 1 2 3 4 5 ;
c x-coors
 0.0 [%barlen - %runuplen - %bardx]
 [%barlen - %runuplen] [%barlen]
c y-coors
[-%cyfaci*%bardia/2] [-%cyfaci*%bardia/2] 0.0
[ %cyfaci*%bardia/2] [ %cyfaci*%bardia/2]
c z-coors
[-%cyfaci*%bardia/2] [-%cyfaci*%bardia/2] 0.0
```

```
[ %cyfaci*%bardia/2] [ %cyfaci*%bardia/2]
c
 mseq i [((%barlen - %runuplen - %bardx)/%bardx) - 1]
         0 [(%runuplen/%bardx) - 1]
 mseq j %bmseqj %bmseqj %bmseqj %bmseqj
 mseq k %bmseqj %bmseqj %bmseqj %bmseqj
c
dei ; 1 2 0 4 5 ; 1 2 0 4 5 ;  c four corners
c
c half or quarter symmetry model
 if (%hqsymm .ne. 0) then
  dei ; ; 3 5 ;   c half z-symmetry
   if (%hqsymm .eq. 4) then
    dei ; 1 3 ; ;   c quarter yz-symmetry
   endif
 endif                    c symmetry model
c
sfi ; -1 0 -5 ; 2 4 ; xcy [%bardia/2] ;
sfi ; 2 4 ; -1 0 -5 ; xcy [%bardia/2] ;
c
c contact surface on left end
   sii -1 ; ; ;  %barbear m
c
c non-reflecting surface on right end
   nri -4 ; ; ;
c
c element print blocks
   epb 2 3 2  3 4 3       c strain gage location
c
c
 lct 1 mx [%bardia/2 + %bearlen] ; lrep 1 ;
c
mate %tbarmat
c
endpart
c
c      ...  specimen  ...
c
block 1 2 3 ; 1 2 3 4 5 ; 1 2 3 4 5 ;
c x-coors
 [-%bardia/2] 0.0 [%bardia/2]
c y-coors
[-%cyfaci*%bardia/2] [-%cyfaci*%bardia/2] 0.0
[ %cyfaci*%bardia/2] [ %cyfaci*%bardia/2]
c z-coors
[-%cyfaci*%bardia/2] [-%cyfaci*%bardia/2] 0.0
[ %cyfaci*%bardia/2] [ %cyfaci*%bardia/2]
c
 mseq i [2*%smseqj + 1] [2*%smseqj + 1]
 mseq j %smseqj %smseqj %smseqj %smseqj
 mseq k %smseqj %smseqj %smseqj %smseqj
c
dei ; 1 2 0 4 5 ; 1 2 0 4 5 ;  c four corners
c
c half or quarter symmetry model
 if (%hqsymm .ne. 0) then
  dei 1 2 ; ; ;   c half z-symmetry
   if (%hqsymm .eq. 4) then
    dei ; 1 3 ; ;   c quarter yz-symmetry
```

```
   endif
 endif                    c symmetry model
c
sfi ; -1 0 -5 ; 2 4 ; xcy [%bardia/2] ;
sfi ; 2 4 ; -1 0 -5 ; xcy [%bardia/2] ;
c
c contact surfaces
   sii ; 2 4 ; -1 0 -5 ;  %specbear m
c
c element print block
c  epb 2 3 3 3 4 4        c upper right center core quadrant
c
 lct 1 ry 90 ; lrep 1 ;
c
mate %specmat
c
endpart
c
c   ...  bearing strip  ...
c
block 1 2 3 ; 1 2 3 ; 1 2 3 ;
c x-coors
 [%bardia/2 + %bearlen] [%bardia/2 + %bearlen] [%bardia/2 + %bearlen]
c y-coors
 [-%bearlen/2] 0.0 [ %bearlen/2]
c z-coors
 [-%bardia/2] 0.0 [ %bardia/2]
c
 mseq i 1 1
 mseq j 1 1
 mseq k [2*%bmseqj + 1] [2*%bmseqj + 1]
c
c half or quarter symmetry model
 if (%hqsymm .ne. 0) then
  dei ; ; 2 3 ;   c half z-symmetry
   if (%hqsymm .eq. 4) then
    dei ; 1 2 ; ;   c quarter yz-symmetry
   endif
 endif                    c symmetry model
c
sfi -1 ; ; ; zcy [%bardia/2] ;              c onto specimen
sfi -2 ; ; ; zcy [%bardia/2 + %bearlen/2] ; c mid-plane
sfi ; -1 ; ; xzplan my [-%bearlen/2] ;      c bottom plane
sfi ; -3 ; ; xzplan my [ %bearlen/2] ;      c top plane
c
c contact surfaces
   sii -1 ; ; ;  %specbear s
   sii -3 ; ; ;  %barbear  s
c
c element print block
   epb 2 2 1 3 3 2        c upper right quadrant
c
c
 lct 1 ryz ; lrep 0 1 ;
c
mate %bearmat
c
endpart
c
```

Index